

# 第 5 章 海量数据分析的利器——初识查找引用函数

在 Excel 2010 版中对于内置函数分成 12 大类，除了所谓不需要学习都会的逻辑函数外，最常用的就是查找与引用函数，几乎每个用户在工作中都接触过此类函数。

查找与引用函数的应用有一定的难度，很多应用都涉及到数组公式，甚至还需要应用多维引用技术。这里仅就一些常见的函数及基础的应用做一些讲解，带领读者初识查找引用函数。

## 5.1 认识函数中的大众情人

查找与引用函数的普及和其中的一个函数有关。谈到函数公式，绝大多数的 Excel 用户会感到很头痛，觉得很专业，也不愿意花时间深入学习。但有一个函数例外，这个函数难度中等但普及率相当高，可以说几乎所有接触过函数公式的用户都知道这个函数，很多人了解运用函数公式就是从这个函数开始的。

这个函数就是人见人爱的函数领域中的大众情人——VLOOKUP。

为什么说 VLOOKUP 函数是函数领域的大众情人？原因在于这个函数的确非常管用，尤其在处置大量数据的查找时效率可谓惊人。下面以一个实例来认识一下这个神秘的函数。

### 5.1.1 一个典型的单条件查找示例

某公司月底结账时需要编制进销存报表，进销存报表中 D 列“本期销售”需要从 H 列“销售汇总”中取数，匹配的原则为 A 列“货号”与 G 列“货号”相符，如图 5.1 所示。

	A	B	C	D	E	F	G	H
1	货号	期初结存	本期进货	本期销售	期末结存		货号	销售汇总
2	E091	557	184		741		E049	263
3	A038	524	125		649		C012	287
4	A045	665	158		823		B080	319
5	A057	527	172		699		A057	394
6	B030	724	139		863		E091	315
7	B077	628	194		822		E001	286
8	B080	577	162		739		F087	324
9	C012	841	112		953		D032	382
10	E049	917	185		1102		A038	285
11	C050	980	109		1089		B030	297
12	E001	518	183		701		B077	320
13	C074	502	198		700		C074	280
14	F087	901	191		1092		A045	470
15	D032	931	103		1034		C050	835

图 5.1 进销存报表示例

问题分析：这是一个非常典型的查找问题，D 列需要填列的数据在 H 列已有，仅需要根据 A 列与 G

列货号匹配后返回。在函数应用没有普及的时候，很多人的做法是一个个手工查找，然后把找到的数值录入电脑。这种做法现在早已成为笑谈。

解决方案：在 D2 单元格录入一个函数公式然后向下复制，得到如图 5.2 所示结果。

=VLOOKUP(A2,\$G\$1:\$H\$15,2,0)

D2		=VLOOKUP(A2,\$G\$1:\$H\$15,2,0)						
	A	B	C	D	E	F	G	H
1	货号	期初结存	本期进货	本期销售	期末结存		货号	销售汇总
2	E091	557	184	315	426		E049	263
3	A038	524	125	285	364		C012	287
4	A045	665	158	470	353		B080	319
5	A057	527	172	394	305		A057	394
6	B030	724	139	297	566		E091	315
7	B077	628	194	320	502		E001	286
8	B080	577	162	319	420		F087	324
9	C012	841	112	287	666		D032	382
10	E049	917	185	263	839		A038	285
11	C050	980	109	835	254		B030	297
12	E001	518	183	286	415		B077	320
13	C074	502	198	280	420		C074	280
14	F087	901	191	324	768		A045	470
15	D032	931	103	382	652		C050	835

图 5.2 VLOOKUP 函数查找返回的结果

公式解析：由上述公式可见 VLOOKUP 函数有 4 个参数，VLOOKUP 函数的语法结构如下：

VLOOKUP（要查找的值，查找的区域，查找区域中返回的列，查找模式）

这个函数的含义就是在查找区域中的第 1 列查找，如有找到则返回该值在查找区域中对列的值，4 个参数的含义如表 5.1 所示。

表 5.1 VLOOKUP函数 4 个参数的理解

参数顺序	参数含义	上例中的取值	注意事项
第1参数	要查找的值	A2	在查找区域的第一列中查找
第2参数	查找的区域	\$G\$1:\$H\$15	如是第4参数是精确匹配则区域第1列不需要排序，如是模糊匹配则需要对区域的第1列升序排序
第3参数	查找区域中返回的列	2	返回查找值在查找区域中所在行第2列的值
第4参数	查找模式	0	等价于FALSE，精确匹配，如果为1或者TRUE，模糊匹配

估计很多初学者看了上面这个表格后仍很难理解透彻，刚接触函数时一下子要理解这么多概念的确不容易。我们可以先跳过难以理解的理论部分内容，从 D2 单元格返回的值来看一下函数公式的运算原理。遇到问题时再回过来看对于函数参数的解释。

以 D2 单元格中的公式为例，就是在查找区域\$G\$1:\$H\$15 的第 1 列中查找 A2 单元格的值“E091”，结果在 G6 单元格中查到了结果，这时候返回查找区域\$G\$1:\$H\$15 对应行（第 6 行）第 2 列的值。查找演示如图 5.3 所示。

	A	B	C	D	E	F	G	H	I
1	货号	期初结存	本期进货	本期销售	期末结存		货号	销售汇总	
2	E091	557	184	315	426		E049	263	
3	A038	524	125	285	364		C012	287	
4	A045	665	188	470	353		B080	319	
5	A057	527	172	394	305		A057	394	
6	B030	724	139	297	566		E091	315	
7	B077	628	194	320	502		E001	286	
8	B080	577	162	319	420		F087	324	
9	C012	841	112	287	666		D032	382	
10	E049	917	185	263	839		A038	285	
11	C050	980	109	835	254		B030	297	
12	E001	518	183	286	415		B077	320	
13	C074	502	198	280	420		C074	280	
14	F087	901	191	324	768		A045	470	
15	D032	931	103	382	652		C050	835	

图 5.3 查找结果演示

备注：第 4 参数在保留第 3 个参数和第 4 个参数之间的“，”，后面省略内容和使用 0 值或者 False 是等价的，以下两个公式也能返回类似的结果。

=VLOOKUP(A2,\$G\$1:\$H\$15,2,)

=VLOOKUP(A2,\$G\$1:\$H\$15,2,FALSE)

职场点睛：演示的示例中只有 10 多行，实际工作中的数据源可能有成百上千行，一样只要用一个公式向下复制就能解决问题。学会这个函数后在实际工作中遇到类似问题可以大幅度提高工作效率。

### 5.1.2 VLOOKUP 函数的局限性

大众情人往往并不完美，实际生活中如此，函数领域中亦如此。VLOOKUP 函数在大数据量查找中具有优势，但也有其局限性。使用 VLOOKUP 函数最常见的困惑有如下几种：

- ❑ 查找值必须位于查找区域的第一列；
- ❑ 仅能处理单条件查找的问题；
- ❑ 只能返回满足查找条件的第一个值；
- ❑ 不能用一个公式返回所有满足条件的查找值。

我们没有必要为难 VLOOKUP 函数，退一步海阔天空，思路开阔一些，运用其他函数也能轻松搞定查找问题。本章后续将介绍 MATCH、INDEX、LOOKUP、OFFSET、INDIRECT、ROW、COLUMN 等查找与引用函数。正如一个人的能力是有极限的，一个函数的能力也是有限的，运用函数组合能够解决更多的问题。

本节示例文件：《5.1 认识函数领域的大众情人.xlsx》

## 5.2 突破 VLOOKUP 函数反向查找的限制

接触函数领域的大众情人——VLOOKUP 函数后，既惊艳其在海量数据查找中的高效，也有一些遗憾，比如查找值必须位于被查找区域的第一列。

实际工作中需要查找的值未必就在被查找区域的第一列，这类问题称之为“反向查找”。直接用

VLOOKUP 函数无法解决，需要做一些技术处理。

### 5.2.1 一个反向查找的例子

如图 5.4 所示，学生的学号与姓名是一一匹配的，如已知学号要查找姓名，可以用 VLOOKUP 函数解决，F2 单元格公式如下：

```
=VLOOKUP(E2,A2:B10,2,0)
```

但反过来已知姓名要查找学号，直接用 VLOOKUP 函数就不行了。

	A	B	C	D	E	F
1	学号	姓名	成绩		学号	姓名
2	A001	管莺菲	79		A002	石永绍
3	A002	石永绍	67			
4	A003	越俊	79		姓名	学号
5	A004	崔亨	100		管莺菲	
6	A005	于成中	79			
7	A006	狄影淑	95			
8	A007	糜彩青	64			
9	A008	秋彩伊	79			
10	A009	裘健栋	55			
11						

图 5.4 已知姓名查询学号

#### 1. 问题分析

这里直接用 VLOOKUP 不行，原因在于在查找区域中需要返回的“学号”在“姓名”的左边，也就是不符合需要查找项目“姓名”在查找区域 A2:B10 的第一列的条件。

#### 2. 解决方案 1

知道了问题的所在，解决就容易了。如果在“学号”的左侧有一列“姓名”，这个问题就迎刃而解了。一个笨方法是在 A 列“学号”之前插入一列，复制原 B 列“姓名”的内容，这样就可以直接使用 VLOOKUP 函数了。

这种方法思路是对的，但操作方法不可取。不建议随意更改数据源，用函数完全能达到插入一个辅助列，重构数据源的效果。在 F5 单元格录入如下公式：

```
=VLOOKUP(E5,IF({1,0},B2:B10,A2:A10),2,)
```

#### 3. 公式解析

整个公式外层还是 VLOOKUP 函数，关于这个函数的用法不再赘述。

公式中使用了一个 IF 函数作为 VLOOKUP 函数的第 2 参数，在编辑栏中选取这一段函数“IF({1,0},B2:B10,A2:A10)”，按 F9 键，返回结果为：

{ "管莺菲","A001";"石永绍","A002";"越俊","A003";"崔亨","A004";"于成中","A005";"狄影淑","A006";"糜彩青","A007";"秋彩伊","A008";"裘健栋","A009" }

返回结果中的花括号代表数组，逗号表示同行不同列，分号表示换行。整个返回结果代表一个 9 行 2 列的内存数组。

如果直接用 F9 键看不容易理解，还可以将这个 IF 函数的运算结果放到单元格中。选取 H2: I10 单元格区域，录入公式“=IF({1,0},B2:B10,A2:A10)”，按“Ctrl+Shift+Enter”三键结束，返回结果如图 5.5

所示。

H2		fx		{=IF({1,0},B2:B10,A2:A10)}						
	A	B	C	D	E	F	G	H	I	J
1	学号	姓名	成绩		学号	姓名		IF函数返回的辅助区域		
2	A001	管莹菲	79		A002	石永绍		管莹菲	A001	
3	A002	石永绍	67					石永绍	A002	
4	A003	越俊	79		姓名	学号		越俊	A003	
5	A004	崔亨	100		于成中	A005		崔亨	A004	
6	A005	于成中	79					于成中	A005	
7	A006	狄影淑	95					狄影淑	A006	
8	A007	糜彩青	64					糜彩青	A007	
9	A008	秋彩伊	79					秋彩伊	A008	
10	A009	裘健栋	55					裘健栋	A009	
11	公式：{=IF({1,0},B2:B10,A2:A10)}									

图 5.5 IF 函数返回结果分析

由上图可见，使用 IF 函数，返回了一个内存数组，相当于重构了一个区域，而这个重构的区域正好满足使用 VLOOKUP 查找时查找值位于被查找区域第一列的要求。

#### 4. 函数套路——IF{1,0}解析

关于 IF 函数在第 4 章中有过介绍，这里 IF 函数的用法和常规不同。第一个参数中使用了一个常量数组{1,0}，第二、三参数各使用了一列数据区域。对于返回结果，可以理解为一个 1\*2 的数组与另一个 9\*1 的数组的运算。返回的是一个 9\*2 的数组，具体原理可参阅第 4 章第 6 节中两个不同方向一维数组的运算。

如能理解数组运算的原理，就能明白为什么 IF 函数第一参数不能写成{1;0}，不要小看仅一个符号的差异，逗号表示横向，分号表示纵向，使用{1;0}，整个 IF 函数返回的结果相当于是一个 2\*1 的数组与一个 9\*1 的数组运算，因这两个数组方向相同大小不一样，部分返回值会出错，使用 F9 测试的结果是{"管莹菲";"A002";#N/A;#N/A;#N/A;#N/A;#N/A;#N/A;#N/A}，仅是一个 9 行 1 列的数组，如图 5.6 所示。

K2		fx		{=IF({1;0},B2:B10,A2:A10)}							
	A	B	C	D	E	F	G	K	L	M	N
1	学号	姓名	成绩		学号	姓名		IF函数第1参数的常量数组中使用分号			
2	A001	管莹菲	79		A002	石永绍		管莹菲	管莹菲		
3	A002	石永绍	67					A002	A002		
4	A003	越俊	79		姓名	学号		#N/A	#N/A		
5	A004	崔亨	100		于成中	A005		#N/A	#N/A		
6	A005	于成中	79					#N/A	#N/A		
7	A006	狄影淑	95					#N/A	#N/A		
8	A007	糜彩青	64					#N/A	#N/A		
9	A008	秋彩伊	79					#N/A	#N/A		
10	A009	裘健栋	55					#N/A	#N/A		
11	公式：{=IF({1;0},B2:B10,A2:A10)}										

图 5.6 IF{1,0}中逗号与分号的差异

如果能够理解 IF{1,0}的返回结果，读者可以自行测试以下公式：

```
=VLOOKUP(E5,IF({0,1},A2:A10,B2:B10),2,)
=VLOOKUP(E5,CHOOSE({1,2},B2:B10,A2:A10),2,)
=VLOOKUP(E5,CHOOSE({2,1},A2:A10,B2:B10),2,)
```

这三个公式也能得出正确的结果，CHOOSE 函数构建内存数组的效果同 IF{1,0}，这里不再详细讲

解。

职场点睛：困扰很多用户的反向查找问题就这样解决了，强烈建议用户多尝试一下这个函数套路，学会这一招，今后在工作中遇到类似问题无需每次再构建辅助列，一个公式就能搞定。

5. 解决方案 2

使用 IF{1,0}构建内存数组的方法很经典，但相对理解有一点难度。查找的问题并非一定要通过 VLOOKUP 函数来解决，使用其他函数也能轻松搞定。在 F5 单元格录入如下公式：

```
=INDEX(A2:A10,MATCH(E5,B2:B10,0))
```

6. 公式解析

对于这个公式，里层是一个 MATCH 函数，作用是“定位”，在单元格区域中搜索指定项，然后返回该项在单元格区域中的相对位置（返回值是一个数值）。公式各参数解释如下表。

表 5.2 MATCH函数 3 个参数的理解

参数顺序	参数含义	上例中的取值	注意事项
第1参数	要查找的值	E5	
第2参数	查找的区域	B2:B10	如是第3参数是0则区域第1列不需要排序，如是1需要对区域的第1列升序排序；如是-1则需要对对区域的第1列降序排序
第3参数	查找模式	0	为0时精确匹配，如为1或者-1则是模糊匹配

读者可以在编辑栏中选中公式的 MATCH 部分“MATCH(E5,B2:B10,0)”，按 F9 键观察返回值，返回结果如图 5.7 所示。




图 5.7 INDEX+MATCH 函数运算过程

公式的外层是一个 INDEX 函数，我们可以把 INDEX 函数的第 1 参数理解为一个矩形区域，函数的结果是返回矩形区域中的某个值，具体返回哪一个值则由该函数的第 2、3 参数决定。第 2、3 参数告诉 Excel，返回值在区域中的第几行和第几列。

本例中 INDEX 函数的第一参数 A2:A10 仅有一列，可以省略第 3 参数，根据第二参数即 MATCH 函数计算的结果来决定返回值的行。

读者可在 E1 单元格的数据有效性下拉列表中任意选择学生姓名，观察 F1 单元格公式结果的变化。

7. 函数套路——INDEX+MATCH 解析

INDEX+MATCH 函数组合相当于把原来 VLOOKUP 一个函数的工作分配给了两个函数来完成。虽然没有 VLOOKUP 那么惊艳，但这个函数组合能完成的工作比单一的 VLOOKUP 函数要多很多，至少



VLOOKUP 能做的工作，INDEX+MATCH 函数组合都能完成。

回到上一节中使用 VLOOKUP 精确查找的进销存案例，使用 INDEX+MATCH 函数组合也能完成，公式如下：

`=INDEX($H$1:$H$15,MATCH(A2,$G$1:$G$15,0))`

这个公式的含义是先在 G 列的“货号”区域\$G\$1:\$G\$13 中查找 A2 单元格中的货号，然后在 H 列“销售汇总”区域\$H\$1:\$H\$13 返回货号对应的销售额。

职场点睛：这个函数套路给我们的启示是解决问题不要仅局限在一种方法上，多掌握几个基础的函数对日常工作绝对是有帮助的。

### 5.2.2 IF{1,0}与 INDEX+MATCH 方法比较

两种方法都很经典，对于初学者建议多尝试一下 INDEX+MATCH 函数的组合，思路不要总局限在 VLOOKUP 函数上。

对于 IF{1,0}，使用时要考虑是否需要使用数组公式（即按“Ctrl+Shift+Enter”三键结束）。本节的示例是不需要使用数组公式结束的，因为 IF 函数的第二和第三参数仅是引用了工作表中的单元格区域，并没有计算，如在引用的同时还有计算，就需要使用数组公式结束。

本节示例文件：《5.2 突破 VLOOKUP 函数反向查找的限制.xlsx》

## 5.3 轻松应对多条件查找

本章前两节中讲解的示例都属于单条件查找，实际工作中更多的是需返回满足多个条件的对应值。相对于单条件查找，多条件查找肯定不能简单应用 VLOOKUP 函数。

### 5.3.1 多条件查找的 2 种解决方案

先看一个多条件查找的经典案例，原始数据在 A:C 列，E2 单元格中设置了数据有效性，取值为产品代码，F2 单元格中取值为部门号，需要在 G2 单元格中设置公式，要求 G2 单元格中返回值随 E2、F2 单元格取值变动而变动，如图 5.8 所示。

	A	B	C	D	E	F	G	H
1	产品	部门	销售额		产品	部门	销售额	
2	产品A	部门1	3200		产品A	部门1		
3	产品B	部门1	4000					
4	产品C	部门1	1600					
5	产品D	部门1	4800					
6	产品A	部门2	1300					
7	产品B	部门2	3000					
8	产品C	部门2	1400					
9	产品D	部门2	4800					
10								

图 5.8 一个多条件查找的示例

#### 1. 问题分析

这里需要返回满足两个条件时对应的销售额，如果把这两个条件连接起来变成一个条件，就能回到上两节讲解的单条件查找的问题中去了。如是多个条件方法也一样。

## 2. 解决方案 1

使用 IF{1,0} 构建内存数组，然后使用 VLOOKUP 函数，在 G2 单元格中输入如下公式，并按“Ctrl+Shift+Enter”三键结束：

```
=VLOOKUP(E2&F2,IF({1,0},A2:A9&B2:B9,C2:C9),2,)
```

## 3. 公式解析

公式中使用了连接符&，将两个查找条件 E2 和 F2 单元格连接起来，在 IF 函数构建的查找区域中，也使用连接符&将两个数据区域连接。将 IF 函数的返回结果放入单元格区域中，如图 5.9 所示。

I2 {=IF({1,0},A2:A9&B2:B9,C2:C9)}											
A	B	C	D	E	F	G	H	I	J	K	
1	产品	部门	销售额		产品	部门	销售额		IF {1,0} 的返回结果		
2	产品A	部门1	3200		产品A	部门2	1300		产品A部门1	3200	
3	产品B	部门1	4000						产品B部门1	4000	
4	产品C	部门1	1600						产品C部门1	1600	
5	产品D	部门1	4800						产品D部门1	4800	
6	产品A	部门2	1300						产品A部门2	1300	
7	产品B	部门2	3000						产品B部门2	3000	
8	产品C	部门2	1400						产品C部门2	1400	
9	产品D	部门2	4800						产品D部门2	4800	
10									公式: {=IF({1,0},A2:A9&B2:B9,C2:C9)}		

图 5.9 IF{1,0}中使用连接符&的返回结果

使用连接符&把多条件查找的问题转换成了单条件查找，连接两个条件是关键的一步。

## 4. 解决方案 2

连接了 2 个条件，将多条件查找变成了单条件查找，后续的问题就简单了，INDEX+MATCH 函数组合也可以解决，在 G2 单元格中录入如下公式，并按三键结束。

```
=INDEX(C2:C9,MATCH(E2&F2,A2:A9&B2:B9,))
```

## 5. 公式解析

这个公式还是典型的 INDEX+MATCH 组合，公式没有悬念，需要注意的是应按三键结束，给 Excel 传递一个信息，这里有多重运算。

### 5.3.2 特殊的多条件查找

多条件查找还有两种特殊的情况，展现的外观不同于之前介绍的例子，但实质上就是多条件查找。

#### 1. 双向查找

查找的条件分别散布在行和列上，这种问题一般称其为“双向查找”。沿用上例的数据，需要查找的条件分别在第 1 行和第 E 列，要求在 F2 单元格中写一个公式，向下向右复制完成。



	A	B	C	D	E	F	G	H
1	产品	部门	销售额			部门1	部门2	
2	产品A	部门1	3200		产品A			
3	产品B	部门1	4000		产品B			
4	产品C	部门1	1600		产品C			
5	产品D	部门1	4800		产品D			
6	产品A	部门2	1300					
7	产品B	部门2	3000					
8	产品C	部门2	1400					
9	产品D	部门2	4800					
10								

图 5.10 双向查找的示例

### 1) 问题分析

条件不在同一行中，实质上还是多条件，使用连接符&连接 2 个条件，将多条件变成单条件是最常用的思路。

### 2) 解决方案

在 F2 单元格中录入如下公式，按三键结束，并向下向右复制完成。

```
=VLOOKUP($E2&F$1,IF({1,0},$A$2:$A$9&$B$2:$B$9,$C$2:$C$9),2,)
```

### 3) 公式解析

公式本身没有悬念，关键要注意 VLOOKUP 函数第一个参数中单元格的引用类型。对于 E2 单元格，公式复制后引用的 E 列是不变的，引用的行会改变，所以对于 E2 单元格，需要使用行相对列绝对的混合引用方式\$E2；对于引用的 F1 单元格判断依此类推。

示例文件中给出了另两种思路的公式如下，注意使用 INDEX+MATCH 函数组合需要按三键结束。

```
=INDEX($C$2:$C$9,MATCH($I2&J$1,$A$2:$A$9&$B$2:$B$9,))
```

```
=LOOKUP(1,0/((($A$2:$A$9=$M2)*($B$2:$B$9=$N$1)),$C$2:$C$9)
```

职场点睛：这个案例的启示是要学会分析问题，将复杂的问题简单化并归纳到已知的套路中去。函数问题如此思考，实际工作也是一样的。

## 2. 多列查找

还有一种特殊的多条件查找，需要用一个公式解决多个查找问题，如图 5.11 所示，在 B9 单元格录入一个公式，向右复制后能在 A9 单元格姓名变动的情况下返回对应的性别、年龄和职务。

	A	B	C	D	E	F
1	工号	姓名	性别	年龄	职务	
2	A01	张三	男	45	总经理	
3	A02	李四	女	27	秘书	
4	A03	王五	女	25	前台	
5	A04	赵六	女	32	客服	
6	A05	钱七	女	37	客服	
7						
8	姓名	性别	年龄	职务		
9	张三					
10						

图 5.11 多列查找的示例

### 1) 问题分析

使用 VLOOKUP 函数一个单元格设置一个公式很简单，但不符合批量操作的要求。要批量操作就要找到这些操作之间的内在规律。先写出单独解决 B9、C9、D9 单元格的公式如下：

```
=VLOOKUP(A9,B2:E6,2,)  
=VLOOKUP(A9,B2:E6,3,)  
=VLOOKUP(A9,B2:E6,4,)
```

可以发现这三个单元格中公式的差异仅是 VLOOKUP 函数的第 3 个参数不一样，分别是数字 2、3、

4。如果有一个函数在 B9 单元格中返回 2，公式向右复制时能变成 3、4，这个问题就解决了。

Excel 中就有这样一个函数——COLUMN。

### 2) 解决方案 1

在 B9 单元格中录入如下公式，并向右复制完成。

```
=VLOOKUP($A$9,$B$2:$E$6,COLUMN(),)
```

### 3) 公式解析

这个公式中的关键是 COLUMN 函数。COLUMN 函数返回参数指定单元格引用的列号，在省略参数的情况下返回 COLUMN 函数所在单元格的列号。

	A	B	C	D	E
8	姓名	性别	年龄	职务	
9	张三	男	45	总经理	
10					
11	COLUMN	2	3	4	
12	公式	=COLUMN()			
13					

图 5.12 COLUMN 函数的返回值

如图 5.12 所示，列出了在省略参数的情况下函数公式“=COLUMN()”在 B、C、D 列的返回值分别是 2、3、4，也就是第 9 行中 VLOOKUP 函数第 3 个参数的取值。

如果这个公式不是放在 B:D 列，写法就要改变一下，COLUMN 函数要加上参数，公式如下：

```
=VLOOKUP($A$9,$B$2:$E$6,COLUMN(B1),)
```

F9		=VLOOKUP(\$A\$9,\$B\$2:\$E\$6,COLUMN(B1),)									
	A	B	C	D	E	F	G	H	I	J	
8	姓名	性别	年龄	职务		性别	年龄	职务			
9	张三	男	45	总经理		男	45	总经理			
10		=VLOOKUP(\$A\$9,\$B\$2:\$E\$6,COLUMN(),)				=VLOOKUP(\$A\$9,\$B\$2:\$E\$6,COLUMN(B1),)					
11	COLUMN	2	3	4		2	3	4			
12	公式	=COLUMN()				=COLUMN(B1)					
13											
14											
15											

COLUMN函数能否省略参数要看返回单元格位置

图 5.13 COLUMN 函数两种写法比较

图 5.13 列示了两种不同的公式写法，COLUMN 函数能否省略参数要看函数所在单元格的位置，这点与 ROW 函数的用法类似。

### 4) 解决方案 2

既然这种多列查找还是一种多条件查找，自然可以试试多条件查找的几种经典解决方案，比如 INDEX+MATCH 函数组合。在 B9 单元中录入如下公式，并向右复制完成。

=INDEX(\$C\$2:\$E\$6,MATCH(\$A\$9,\$B\$2:\$B\$6,1),MATCH(B8,\$C\$1:\$E\$1,1))

#### 5) 公式解析

这里使用了两次 MATCH 函数,分别定位两个条件(行方向的姓名,列方向的要查找项目)在 INDEX 函数第一个参数中的位置。

#### 6) 解决方案 3

有兴趣的读者还可以试试 IF{1,0}的解决方案,公式如下,需按三键结束,这里不再讲解。

=VLOOKUP(\$A9&B8,IF({1,0},\$B2:\$B6&C1,C2:C6),2,)

本节示例文件:《5.3 轻松应对多条件查找.xlsx》

## 5.4 查找满足条件的最后一个值

本章前 3 节介绍的都是精确查找,满足条件的返回值都只有一个。如满足条件的值有多个,将会是什么情况呢?在精确查找模式下,不管是单条件还是多条件查找,不论是用 VLOOKUP 函数、VLOOKUP{1,0}或者 INDEX+MATCH 函数组合,返回结果都仅是满足条件的第一个值。

### 5.4.1 LOOKUP 函数应用于模糊查找

可能查找需要返回的值并非第一个,这样就需要使用其他函数组合来完成,本节介绍使用 LOOKUP 函数返回满足条件的最后一个值,请看下面的例子。

如图 5.14 所示,A 列为升序排列的成交时间,B 列为货号,C 列存放单笔成交单价,需提取当天各个货号的最后一笔成交单价,即黄色标注部分。

#### 1. 问题分析

这个问题如用 VLOOKUP 函数,只能返回各个货号的第一笔成交单价,如图 5.14 中红色字体标注。

VLOOKUP 公式: =VLOOKUP(E2,\$B\$2:\$C\$10,2,0)

F2		fx		=VLOOKUP(E2,\$B\$2:\$C\$10,2,)			
	A	B	C	D	E	F	G
1	成交时间	货号	单价		货号	最后一笔单价	
2	9:00:00	A	88		A	88	
3	10:00:00	B	91		B	91	
4	11:00:00	A	92		C	98	
5	12:00:00	C	98				
6	13:00:00	B	94				
7	14:00:00	A	87				
8	15:00:00	C	96				
9	15:30:00	B	89				
10	16:00:00	C	84				

图 5.14 VLOOKUP 函数返回了第一笔成交单价

#### 2. 解决方案

在查找货号的情况下,因相同的货号不止一个,精确查找无法解决。换一种思路,可以试一下模糊查找,在 F2 单元格输入如下公式,向下复制,如图 5.15 所示。

=LOOKUP(1,0/(\$B\$2:\$B\$10=E2),\$C\$2:\$C\$10)

F2		=LOOKUP(1,0/(\$B\$2:\$B\$10=E2),\$C\$2:\$C\$10)					
	A	B	C	D	E	F	G
1	成交时间	货号	单价		货号	最后一笔单价	
2	9:00:00	A	88		A	87	
3	10:00:00	B	91		B	89	
4	11:00:00	A	92		C	84	
5	12:00:00	C	98				
6	13:00:00	B	94				
7	14:00:00	A	87				
8	15:00:00	C	96				
9	15:30:00	B	89				
10	16:00:00	C	84				

图 5.15 LOOKUP 函数查找最后一笔成交单价

### 3. 公式解析

这个公式使用了 LOOKUP 函数，LOOKUP 函数不同于 VLOOKUP 函数的地方是，其本身就是一个模糊查找的函数，不像 VLOOKUP 函数通过第 4 参数来控制查找模式。

在讲解 VLOOKUP 函数时曾经提到，在模糊查找模式下，数据区域的第一列需要按升序排列，使用 LOOKUP 函数时自然对被查找区域数据也有类似的要求。但是否一定要对查找区域先排序呢？那样不是很麻烦吗？答案是否定的，这里使用了 LOOKUP(1,0)的函数套路。

#### 4. 函数套路——LOOKUP (1, 0) 解析

如果直接使用模糊查找，数据区域第一列需要按升序排列。如将查找值与被查找区域的第一列都转化为如 1 或 0 之类的数值（例如查找 1，查找区域中都是 0），就可以规避预先排序的问题。这个公式可以归纳为以下套路：

=LOOKUP (1, 0 / (条件区域=条件), 返回值区域)

第 2 个参数使用了“0 / (条件区域=条件)”，其中“条件区域=条件”判断结果非 TRUE 即 FALSE，用 0 除以判断的结果 TRUE 或者 FALSE，得到的值为 0 或者 #DIV/0!。

选取 LOOKUP 函数第 2 参数部分，即：

0/(\$B\$2:\$B\$10=E2)

按 F9 后得到结果为{0;#DIV/0!;0;#DIV/0!;#DIV/0!;0;#DIV/0!;#DIV/0!;#DIV/0!;#DIV/0!}。把这一段的计算结果放入单元格，如图 5.16 所示，可以发现在 B 列的货号中，对应 E2 单元取值“A”的行返回值为 0，其他行返回值为“#DIV/0!”。这一段公式的实质是构建了一个辅助列，作为 LOOKUP 函数的第 2 个参数。

F2		=LOOKUP(1,0/(\$B\$2:\$B\$10=E2),\$C\$2:\$C\$10)						
	A	B	C	D	E	F	G	H
1	成交时间	货号	单价		货号	最后一笔单价		辅助列
2	9:00:00	A	88		A	87		0
3	10:00:00	B	91		B	89		#DIV/0!
4	11:00:00	A	92		C	84		0
5	12:00:00	C	98					#DIV/0!
6	13:00:00	B	94					#DIV/0!
7	14:00:00	A	87					0
8	15:00:00	C	96					#DIV/0!
9	15:30:00	B	89					#DIV/0!
10	16:00:00	C	84					#DIV/0!

图 5.16 公式第 2 参数运算结果

公式分解到这一步，就是一个 LOOKUP 函数了，这个 LOOKUP 函数使用向量形式有 3 个参数，在第 2 个参数区域 (0/(\$B\$2:\$B\$10=E2)) 中查找第 1 个参数 (1)，然后返回第 3 个参数区域 (\$C\$2:\$C\$10) 中相同位置的值。

第 2 参数中只有 0 和错误值，错误值不参与查找，其他都是 0 值，肯定找不到“1”，这时模糊查找发挥作用了。Excel 默认辅助列中的值都已按升序排列，如找到 0 值后会认为下面还有更大的值，继续查找，直到定位到最后一个 0 值为止。这样查找就能返回 E2 单元格中货号 A 对应的最后一笔成交价 87。

对于货号 B 和 C 的查找，读者可以按照上述方法自行测试一下，加深对于 LOOKUP(1,0)函数套路的理解。

#### 5.4.2 用模糊查找做精确查找的工作

LOOKUP 是用于模糊查找的函数，但应用这个函数可以完成精确查找的工作，不论是单条件还是多条件，也不管是不是所谓的反向查找都可以完成，但其前提是要用 LOOKUP (1,0) 的函数套路。

在满足条件的值只有一个的情况下，使用 LOOKUP (1,0) 函数套路能达到使用 VLOOKUP{1,0}与 INDEX+MATCH 函数组合同等的效果。

沿用第 3 节多列查找的例子，在 B9 单元格中录入如下公式，向右复制完成，不需要按三键结束。

=LOOKUP(1,0/(\$B2:\$B6&C1=\$A9&B8),C2:C6)

LOOKUP 函数的第 2 参数 “0/(\$B2:\$B6&C1=\$A9&B8)” 相当于构建了一个仅有 0 和 #DIV/0! 的内存数组（辅助列），将运算结果放在单元格中再次理解一下，如图 5.17 所示。

	A	B	C	D	E	F	G	H	I
1	工号	姓名	性别	年龄	职务		辅助列		
2	A01	张三	男	45	总经理		#DIV/0!	返回0所在行的数据	
3	A02	李四	女	27	秘书		#DIV/0!		
4	A03	王五	女	25	前台		0		
5	A04	赵六	女	32	客服		#DIV/0!		
6	A05	钱七	女	37	客服		#DIV/0!		
7							{=0/(\$B2:\$B6&C1=\$A9&B8)}		
8	姓名	性别	年龄	职务					
9	王五	女	25	前台					
10	公式：	=LOOKUP(1,0/(\$B2:\$B6&C1=\$A9&B8),C2:C6)					查找1，只能找到0		

图 5.17 LOOKUP 函数运行结果分析

在辅助列中查找 1，只能找到一个 0（其他都是错误值），因满足条件的仅有一个值，模糊查找变成了精确查找。

关于多条件查找的其他示例读者也可以自行测试，在满足查找条件的返回值仅有一个的情况下，使用 LOOKUP（1,0）函数套路可以起到精确查找的效果。

本节示例文件《5.4 查找满足条件的最后一个值.xlsx》

## 5.5 巧妙利用函数的模糊查找

上一节中讲到了 LOOKUP 函数的模糊查找，介绍了 LOOKUP（1,0）函数套路查找满足条件的最后一个值。其实不仅是 LOOKUP 函数，VLOOKUP 和 MATCH 函数也可运用于模糊查找，MATCH 函数还有 2 种模糊查找模式。

### 5.5.1 典型的区间查找案例

模糊查找最典型的应用是区间查找，下面举两个例子。

#### 1. 个人所得税计算

很多用户计算个人所得税时使用的是 IF 函数，因个人所得税是 7 级超额累进税率，用 IF 函数公式写起来很麻烦，而且有错也不容易查核。这里介绍使用 VLOOKUP 函数的模糊查找功能轻松搞定这个问题。

之前介绍过，VLOOKUP 函数第 4 个参数如使用 1 或者 TRUE，是模糊查找模式。模糊查找就是在查找区域的第 1 列中没有找到查找值时，并不简单地返回“#N/A”，而是根据模糊查找的原理返回比查找值小的最大值。

函数的帮助文件很难理解，我们还是通过实例来认识理解函数。

实例中 B 列为税前薪资，C 列为扣除税前费用 3,500 后的应纳税额，在 D 列计算个人所得税，不考虑缴纳社保等其他事项的影响，如图 5.18 所示。



	A	B	C	D	E	F
1	姓名	薪资	应纳税额	个调税		扣除费用
2	员工A	25000	21500			3500
3	员工B	7000	3500			
4	员工C	22000	18500			
5	员工D	21000	17500			
6	员工E	14000	10500			
7	员工F	29000	25500			
8	员工G	10000	6500			
9	员工H	5000	1500			
10	员工I	29000	25500			

图 5.18 个人所得税计算表

### 1) 问题分析

个调税 = 应纳税额 \* 税率 - 速算扣除数

需要先求出税率和速算扣除数。这还是一个查找问题，只是查找返回值要根据查找值（应纳税额）的分段区间，需要应用模糊查找。

### 2) 解决方案 1

直接输入公式比较麻烦，我们可以先找一个税率表，加工之后作为辅助区域，定义为名称后使用。如图 5.19 所示，这是一个网上下载的最新税率表，其中 B 列的“分段”数字是自行添加的，读者可以观察一下 B 列“分段”数字与 A 列“含税级距”之间的关系。

将表中 B2:D8 区域定义为名称“税率”，用于简化公式录入。

税率		fx		0	
	A	B	C	D	E
1	含税级距	分段	税率(%)	速算扣除数	
2	不超过1500元的	-	3	-	
3	超过1500元至4,500元的部分	1,500	10	105	
4	超过4,500元至9,000元的部分	4,500	20	555	
5	超过9,000元至35,000元的部分	9,000	25	1,005	
6	超过35,000元至55,000元的部分	35,000	30	2,755	
7	超过55,000元至80,000元的部分	55,000	35	5,505	
8	超过80,000元的部分	80,000	45	13,505	
9					

图 5.19 个人所得税税率表

做好准备工作后，可以在“个人所得税计算表”的 D2 单元格输入如下公式并向下复制。

=C2\*VLOOKUP(C2,税率,2,1)/100-VLOOKUP(C2,税率,3,1)

### 3) 公式解析

这个公式使用了 2 个 VLOOKUP 函数，第 1 个函数取得“税率”，第 2 个函数返回“速算扣除数”。如对于计算结果不理解的话，可以用 F9 键来分析一下公式的分段运算结果。

对于 D2 单元格中的公式，在编辑栏中选取“VLOOKUP(C2,税率,2,1)”，按 F9 键，返回结果是 25。选择后一段 VLOOKUP 函数，按 F9 键，可以发现运算结果为 1,005。由此可见，对于员工 A，应纳税额 21,500 对应的税率为 25%，速算扣除数是 1,005。

进入“个人所得税税率表”检验一下，结果无疑是正确的。

明明在“个人所得税税率表”的 B 列没有要查找的值 21,500，为什么不仅未报错还返回了正确的结果？这就是 VLOOKUP 函数的模糊查找在起作用。

对于查找值 21,500，如在查找区域的第 1 列中没有找到，则会返回比 21,500 小的最大值，在这个例子里比 21,500 小的最大值是 9,000，返回的就是 9,000 所在行对应的税率和速算扣除数。如图 5.20。

	A	B	C	D	E	F
1	含税级距	分段	税率(%)	速算扣除数		
2	不超过1500元的	-	3	-		
3	超过1500元至4,500元的部分	1,500	10	105		
4	超过4,500元至9,000元的部分	4,500	20	555		
5	超过9,000元至35,000元的部分	9,000	25	1,005		
6	超过35,000元至55,000元的部分	35,000	30	2,755		
7	超过55,000元至80,000元的部分	55,000	35	5,505		
8	超过80,000元的部分	80,000	45	13,505		
9						
10						
11	B列已按升序排列，查找21,500时，如找不到则返回比其小的最大值9,000。					
12						

图 5.20 模糊查找的运算过程

对于其他几位员工个税调税返回的值，读者可以自行运用上面的方法加以验证，加深对模糊查找原理的理解。

职场点睛：学会了 VLOOKUP 函数的模糊查找，应对个税调税之类问题就不再需要使用很长的 IF 函数多层嵌套了。对于税率表可以定义为名称，公式显得更加灵活。

#### 4) 解决方案 2

这里使用 LOOKUP 函数也能实现，公式如下：

**=C2\*LOOKUP(C2,税率表!\$B\$2:\$B\$8,税率表!\$C\$2:\$C\$8)/100-LOOKUP(C2,税率)**

#### 5) 公式解析

观察上述公式可以发现，2 段 LOOKUP 函数分别使用了不同数量的参数。第一个 LOOKUP 函数是 3 个参数，第二个 LOOKUP 函数是 2 个参数。

这里分别运用了 LOOKUP 函数的向量形式和数组形式。

**向量形式：LOOKUP(C2,税率表!\$B\$2:\$B\$8,税率表!\$C\$2:\$C\$8)**

使用向量形式是 3 个参数，在第 2 个参数区域（税率表!\$B\$2:\$B\$8）中查找第 1 个参数（C2），然后返回第 3 个参数区域（税率表!\$C\$2:\$C\$8）中相同位置的值。

**数组形式：LOOKUP(C2,税率)**

使用数组形式是在数组（已定义名称“税率”）的第一行或第一列中（这里已定义名称“税率”的列数多于行数，在第一列中查找）查找指定的值，然后返回数组的最后一行或最后一列中相同位置的值。

在分段模糊查找时可以用 LOOKUP 代替 VLOOKUP，且比 VLOOKUP 更加灵活。

## 2. 学生成绩分级

如图 5.21 所示，引用第 4 章中学生成绩分级的例子，在 F 列增加一个成绩“分段”值，比较 F 列“分段”值与 G 列“分数”的取值，可以发现“分段”的取值就是“分数”取值的上限，这个规律在上例图 5.19 中也能得到验证。

这里使用 VLOOKUP 函数和 LOOKUP 函数都能解决，公式如下：

**=VLOOKUP(B2,\$F\$2:\$H\$5,3,1)**

=LOOKUP(B2,\$F\$2:\$H\$5)								
	A	B	C	D	E	F	G	H
1	学生	成绩	IF公式	VLOOKUP公式	LOOKUP公式	分段	分数	等级
2	A	91	优秀	优秀	优秀	0 [0,60)		不及格
3	B	61	及格	及格	及格	60 [60,80)		及格
4	C	58	不及格	不及格	不及格	80 [80,90)		良好
5	D	65	及格	及格	及格	90 [90,100]		优秀
6	E	84	良好	良好	良好			
7								
8	IF公式	=IF(B2>=90,"优秀",IF(B2>=80,"良好",IF(B2>=60,"及格","不及格")))						
9	VLOOKUP公式	=VLOOKUP(B2,\$F\$2:\$H\$5,3,1)						
10	LOOKUP公式	=LOOKUP(B2,\$F\$2:\$H\$5)						

图 5.21 三个函数公式比较

读者可以比较一下三个公式，用辅助区域结合查找函数比用 IF 公式会方便很多，而 LOOKUP 函数比 VLOOKUP 运用更加灵活。

## 5.5.2 模糊查找规律的总结

通过上面两个例子的学习，相信读者对于函数的模糊查找应该有了一定的认识，这里对模糊查找的规律加以总结。

- ❑ 使用模糊查找机制时，VLOOKUP 函数查找区域（LOOKUP 数组形式）的第一列或者 LOOKUP 向量形式的第 2 个参数需要按升序排列；
- ❑ 在模糊查找时，如不能返回对于查找值的精确匹配，则返回比查找值小的最大值所在行对应的值；
- ❑ 如果查找值比查找区域第 1 列中的任何值都小，则返回错误值“#N/A”；
- ❑ 如果查找区域的第一列未按升序排列，则查找返回的结果可能正确，可能错误，也可能直接返回错误值“#N/A”。

备注：模糊查找是函数中的一大难点，我们日常工作中使用的多是精确查找。建议初学者除本节的举例外，对模糊查找需要慎用。

本节示例文件《5.5 巧妙利用函数的模糊查找.xlsx》

## 5.6 处理跨工作表查找

数据管理的理想状态是所有数据都在一张工作表中，这也仅是一种理想状态。之前举例的查找数据都是在一张工作表中，如涉及在多个工作表中查询，就需要使用其他查找与引用函数。

### 5.6.1 一个跨工作表查询的实例

某企业在全国各地有若干家分公司，统计各家分公司的销售量，每个月一张工作表，表结构一致，需要在“查询”表中根据月份和公司名称查询销售量，“查询”表的 A2 和 B2 单元格设置了数据有效性下拉列表，要求 C2 单元格数字随 A2 和 B2 单元格取值变化而变化，如图 5.22 所示。

	A	B		A	B	C
1	分公司	数量	1	月份	分公司	数量
2	北京分公司	259	2	一月	北京分公司	375
3	上海分公司	375	3			
4	广州分公司	463	4			
5	深圳分公司	542	5			
6	沈阳分公司	548	6			
7	重庆分公司	375	7			
8	武汉分公司	373	8			
9	厦门分公司	996	9			

图 5.22 跨工作表查询

### 1. 问题分析

如果仅查询“一月”表中的数据，只需使用如下公式即可：

```
=VLOOKUP(B2,一月!A2:B9,2,)
```

观察这个公式，不同于之前所有查询数据都在同一个工作表中的例子，在 VLOOKUP 函数的第 2 参数中引用的区域“A:B”之前加上了这个区域所在工作表名称“一月”及“!”，即“一月!A:B”。如果把“一月”当成一个变量，用单元格引用 A2 来代替，是否就可以实现 VLOOKUP 函数第 2 个参数的查找区域随着变量月份（A2 单元格）的变动而变动？

答案是否定的，使用公式“=VLOOKUP(B2,A2&"!A2:B9",2,)" 结果会报错“#VALUE!”，对第 2 参数“A2&"!A2:B9”使用 F9，返回结果是“"一月!A2:B9"”，仅比之前的公式多了一对半角的双引号，但就是不能返回正确的结果，如图 5.23 所示。

VLOOKUP    x ✓ f    =VLOOKUP(B2,"一月!A2:B9",2,)						
	A	B	C	D	E	F
1	月份	分公司	数量			
2	一月	北京分公司	375			
3						
4			375			
5			"!A2:B9",2,)			

图 5.23 错误的引用公式

### 2. 解决方案

并非思路不对，关键在于函数的写法，在引用区域中使用变量，需要请出引用函数 INDIRECT。在“查询”表的 C2 单元格中输入如下公式：

```
=VLOOKUP(B2,INDIRECT(A2&"!A2:B9"),2,)
```

### 3. 公式解析

这个公式外层是最常用的 VLOOKUP 函数，用了精确查找；内层使用了 INDIRECT 函数作为 VLOOKUP 函数的第 2 参数。

如果使用 F9 键查看 INDIRECT 函数中参数的运算结果，发现和之前错误的公式中一样，都是 “一月!A2:B9”，如图 5.24。

VLOOKUP    x ✓ f_x    =VLOOKUP(B2,INDIRECT("一月!A2:B9"),2,)							
	A	B	C	D	E	F	G
1	月份	分公司	数量				
2	一月	北京分公司	A2:B9",2,)				

图 5.24 INDIRECT 函数中参数的运行结果

差异仅在外面多了一个 INDIRECT 函数，结果完全不一样，我们有必要来认识一下这个神秘的函数。  
Excel 帮助文件中对于 INDIRECT 函数的说明中有一句话很值得玩味 “如果需要更改公式中对单元格的引用，而不更改公式本身，请使用函数 INDIRECT”。改变对单元格的引用而不更改公式本身，可以理解对于公式中引用了变量（单元格），实现了引用内容与变量（单元格）的联动。  
这个函数只有 2 个参数，但理解比较困难，详见表 5.3 所示。

表 5.3 INDIRECT 函数 2 个参数的理解

参数顺序	参数含义	上例中的取值	注意事项
第1参数	对单元格的引用	A2&"!A2:B9"	引用的单元格与文本之间用连接符&，文本外使用一对引号
第2参数	引用样式	省略	TRUE或省略为A1样式的引用，FALSE为R1C1样式的引用

不要把这个函数想得太神秘，其效果就是引用，在使用的时候可以先想好直接引用时是如何使用的，然后把引用内容放到 INDIRECT 函数中。需要注意的是如有引用的单元格，需和引用的文本用连接符&连接，引用的文本本身外面要加上一对半角英文状态下的双引号。  
这个例子已可以使用 INDIRECT 的 R1C1 引用方式，公式如下：  
=VLOOKUP(B2,INDIRECT(A2&"!R2C1:R9C2",FALSE),2,)  
比较这两个公式，使用 R1C1 引用时，INDIRECT 函数不能省略第 2 参数。这里 “R2C1:R9C2” 相当于 A1 引用下的 “A2:B9”。

5.6.2 引用工作表名的规则

如果把工作表名改成 “1 月”、“2 月”、“3 月”之类的，即工作表名字中间有个空格，这时候仍沿用之前的公式就会报错，正确的公式如下：  
=VLOOKUP(B2,INDIRECT("'"&A2&"!A2:B9"),2,)  
差异仅在引用的工作表外面加上了一对半角英文状态下的单引号 “'”。如果不能确定何时需要这一对单引号，可以在使用 INDIRECT 函数时一直加上，多加 Excel 是不会报错的。当然也可以在单元格中测试一下跨工作表的引用，如引用 “3 月”工作表的 A:B 列，如图 5.25 所示。

D5		fx = '3月'!A:B			
	A	B	C	D	E
1	月份	分公司	数量		
2	1月	上海分公司	575		
3					
4					
5				#VALUE!	
6				= '3月'!A:B	

图 5.25 在单元格中测试跨工作表的引用

如果在单元格中引用跨工作表区域时在工作表名外面有这对单引号，则使用 INDIRECT 函数时需要加上，反之则不需要，这是最简单的判断方法。

职场点睛：上述案例中如将三张数据表中的数据先汇总在一张工作表中，前面加上“月份”字段，问题就简单了，变成了常规的多条件查找，可见在实际工作中规范数据源表非常重要。

本节示例文件：《5.6 处理跨工作表查找.xlsx》

## 5.7 查找返回指定区域

之前章节中接触的示例都是查找返回单个值，不论是第一个还是最后一个。如果需要返回的值有多个？如果需要返回的是某个区域？有些需求使用查找函数已经不能解决了。

### 5.7.1 查找返回横向区域

请看如下示例，已知各个业务员 1-6 月份的销售额，A10 单元格中为业务员姓名，B10 单元格中为需要计算销售额的起止月份（设置了单元格自定义格式，实际上是数字 2 到 6），需要在 C10 单元格中使用一个公式计算某个业务员若干月份的销售额合计，如图 5.26 所示。

	A	B	C	D	E	F	G
1		1月	2月	3月	4月	5月	6月
2	业务员A	193	748	715	903	529	390
3	业务员B	230	259	236	105	781	827
4	业务员C	397	604	470	904	767	308
5	业务员D	657	302	649	273	922	316
6	业务员E	780	337	261	576	199	235
7	业务员F	172	752	639	909	441	561
8							
9	业务员	月份	销售额				
10	业务员E	1-3月	1378				

图 5.26 计算业务员若干月销售额合计

#### 1) 问题分析

初看像是一个查找问题，使用查找函数如 VLOOKUP 虽然可以很方便地查找指定业务员某个月份的



销售额，但如要计算多个月份的销售额，就需要在 VLOOKUP 函数的第 3 参数中使用数组，涉及到构建数组的技巧，略有点麻烦。而使用引用函数 OFFSET 仅是一个最基础的应用。

2) 解决方案

可以使用引用函数返回满足条件的数据区域，再外套 SUM 函数求和，C10 单元格公式如下：

```
=SUM(OFFSET(A1,MATCH(A10,A2:A7,1),1,B10))
```

3) 公式解析

公式中使用了 OFFSET 函数，然后外套 SUM 函数求和。OFFSET 函数有 5 个参数，第 2 个参数使用了 MATCH 函数，查找引用的业务员姓名 (A10 单元) 在 A2: A7 单元格区域中的位置。整个 OFFSET 函数的含义是以 A1 单元格为起点，以 MATCH 函数计算结果向下偏移，再向右偏移一个单元格，然后返回一个 1 行若干列 (B10 单元格的取值) 的区域。整个过程如图 5.27 所示。




图 5.27 查询返回区域演示结果

OFFSET 函数有 5 个参数，是 Excel 中参数最多的函数之一，也是超级难解的函数之一。本章仅讲解最基础的应用，先从函数的 5 个参数开始理解。

表 5.4 OFFSET函数 5 个参数的理解

参数顺序	参数含义	上例中的取值	注意事项
第1参数	起点，可以是单元格或区域	A1	使用区域时以左上角单元格为偏移的起点
第2参数	行方向偏移	MATCH(A10,A2:A7,)	正数（代表在起始引用的下方）或负数（代表在起始引用的上方）
第3参数	列方向偏移	1	正数（代表在起始引用的右边）或负数（代表在起始引用的左边）
第4参数	返回引用区域的行数	省略	可以省略，负数代表向上方取数
第5参数	返回引用区域的列数	B10	可以省略，负数代表向左方取数

备注：Excel 的帮助文件中在提到 OFFSET 函数的第 4-5 参数时有错，这两个参数可以使用负数。初次接触 OFFSET 函数会比较难理解，这里将以几个示意图来帮助理解。

1. 省略第 4、5 参数

图 5.28 中使用的公式为 “=OFFSET(A1,5,4)” ，OFFSET 函数以第 1 参数 A1 单元格为起点，按第 2 参数向下偏移，按第 3 参数向右偏移，在省略了第 4、5 参数的情况下，返回区域同第 1 参数的大小。

	A	B	C	D	E	F	G	H	I	J
1	11	12	13	14	15	16	17	18	19	
2	21	22	23	24	25	26	27	28	29	
3	31	32	33	34	35	36	37	38	39	
4	41	42	43	44	45	46	47	48	49	
5	51	52	53	54	55	56	57	58	59	
6	61	62	63	64	65	66	67	68	69	
7	71	72	73	74	75	76	77	78	79	
8	81	82	83	84	85	86	87	88	89	
9	91	92	93	94	95	96	97	98	99	
10	101	102	103	104	105	106	107	108	109	
11										
12	65									
13	=OFFSET(A1,5,4)									

第4、5参数省略，返回区域同第1参数大小

图 5.28 OFFSET 函数演示 1

## 2. 省略第 5 参数

图 5.29 中使用的公式为“=OFFSET(A1,5,4,3)”，这个公式有点类似于“查找返回区域”的案例，OFFSET 函数以第 1 参数 A1 单元格为起点，按第 2 参数向下偏移，按第 3 参数向右偏移，在第 5 参数省略的情况下，以第 4 参数为返回区域的高度，以第 1 参数的宽度为返回区域的宽度，返回 3\*1 的区域。

	A	B	C	D	E	F	G	H	I	J
1	11	12	13	14	15	16	17	18	19	
2	21	22	23	24	25	26	27	28	29	
3	31	32	33	34	35	36	37	38	39	
4	41	42	43	44	45	46	47	48	49	
5	51	52	53	54	55	56	57	58	59	
6	61	62	63	64	65	66	67	68	69	
7	71	72	73	74	75	76	77	78	79	
8	81	82	83	84	85	86	87	88	89	
9	91	92	93	94	95	96	97	98	99	
10	101	102	103	104	105	106	107	108	109	
11										
12	#VALUE!									
13	=OFFSET(A1,5,4,3)									

返回区域高度为第4参数的3，宽度因第5参数省略，视同第1参数的宽度1。

图 5.29 OFFSET 函数演示 2

图 5.29 中 A12 单元格返回值出错，原因在于函数返回的是一个单元格区域而不是一个单元格，把单元格区域放在单元格中，大小不匹配所以报错。使用 F9 键的结果为{65;75;85}，这个运算结果可以作其他函数的参数，如再外套 SUM 函数等。

## 3. 省略第 4 参数

图 5.30 中使用的公式为“=OFFSET(A1:C3,5,4,4)”，OFFSET 函数第 1 个参数使用了区域 A1: C3，但仍以区域左上角单元格 A1 为起点，按第 2 参数向下偏移，按第 3 参数向右偏移，在第 4 参数省略时以第 1 参数的高度作为返回区域的高度，返回区域的宽度使用第 5 参数。

	A	B	C	D	E	F	G	H	I	J
1	11	12	13	14	15	16	17	18	19	
2	21	22	23	24	25	26	27	28	29	
3	31	32	33	34	35	36	37	38	39	
4	41	42	43	44	45	46	47	48	49	
5	51	52	53	54	55	56	57	58	59	
6	61	62	63	64	65	66	67	68	69	
7	71	72	73	74	75	76	77	78	79	
8	81	82	83	84	85	86	87	88	89	
9	91	92	93	94	95	96	97	98	99	
10	101	102	103	104	105	106	107	108	109	
11										
12	#VALUE!									
13	=OFFSET(A1:C3,5,4,,4)									

第1参数虽是一个区域，但以左上角单元格为起点，第4参数省略时使用第1参数的高度。

图 5.30 OFFSET 函数演示 3

通过以上三个演示，可以得出关于 OFFSET 函数的以下几个规律：

- ❑ OFFSET 函数有 5 个参数，其中第 4、5 参数是可以省略的；
- ❑ OFFSET 函数第 1 参数可以是一个单元格区域，但仅以该区域左上角单元格为偏移的起点；
- ❑ 在省略第 4、5 参数的情况下，以第 1 参数的大小作为返回区域的大小；
- ❑ OFFSET 函数返回的结果可以是一个单元格，也可以是一个单元格区域。

## 5.7.2 谈谈查找函数和引用函数

本章中重点讲述了 6 个函数，其中 VLOOKUP、LOOKUP、MATCH 是查找函数，INDEX、INDIRECT 和 OFFSET 属于引用函数，同属查找与引用函数大类，其实并不完全一样。

区分查找与引用函数，可以看一下如下示例，如图 5.31，计算业务员 D 在 6 月份的销售额。

	A	B	C	D	E	F	G
1		1月	2月	3月	4月	5月	6月
2	业务员A	193	748	715	903	529	390
3	业务员B	230	259	236	105	781	827
4	业务员C	397	604	470	904	767	308
5	业务员D	657	302	649	273	922	316
6	业务员E	780	337	261	576	199	235
7	业务员F	172	752	639	909	441	561
8							
9	业务员	月份	销售额				
10			VLOOKUP	INDEX	OFFSET	INDIRECT	
11	业务员D	6月	316	316	316	316	
12							
13		1-6月		3119	3119	3119	

图 5.31 多种公式查找业务员销售额

这个问题可以使用 VLOOKUP、INDEX、OFFSET 和 INDIRECT 四种函数组合，公式分别如下：

```
=VLOOKUP(A11,A2:G7,MATCH(B11&"月",A1:G1,))  
=INDEX(B2:G7,MATCH(A11,A2:A7,),MATCH(B11&"月",B1:G1,))  
=OFFSET(A1,MATCH(A11,A2:A7,),MATCH(B11&"月",B1:G1,))  
=INDIRECT("R"&MATCH(A11,A1:A7,)&"C"&MATCH(B11&"月",A1:G1,))
```

除了最后一个公式使用了 **INDIRECT** 函数的 R1C1 引用样式外，其他的函数组合并不陌生，返回结果也一样。但如需要利用这个公式继续计算业务员 D 在 1-6 月的销售额，查找和引用函数的效果就不一样了。

以 B5 单元格使用区域运算符（冒号）连接上述公式，再外套 **SUM** 函数求和，仅在使用引用函数时能返回正确的结果，查找函数甚至无法计算，公式直接报错。正确的公式如下：

```
=SUM(B5:INDEX(B2:G7,MATCH(A11,A2:A7,),MATCH(B11&"月",B1:G1,)))  
=SUM(B5:OFFSET(A1,MATCH(A11,A2:A7,),MATCH(B11&"月",B1:G1,)))  
=SUM(B5:INDIRECT("R"&MATCH(A11,A1:A7,)&"C"&MATCH(B11&"月",A1:G1,)))
```

从这个例子可以看出查找函数和引用函数还是有差异的，这里建议读者了解以下几点：

- ☐ 在返回值为单个单元格时，使用查找函数和引用函数的结果没有区别；
- ☐ 查找函数能做的工作，使用引用函数也能完成；
- ☐ 反之，一部分引用函数的工作就不能用查找函数来实现。

关于查找函数与引用函数的差异，在后续章节中还会举例讲解。本节示例文件：《5.7 查找返回指定区域.xlsx》

## 5.8 查找结果报错的处理

使用函数出错是难免的，关于各种错误返回值的形成原因已在第 4 章中讲述，这里再介绍一下具体的处理方法。

### 5.8.1 使用 IFERROR 函数消错

如图 5.32 所示的例子，查找结果中出现了“#N/A”，原因在于某些货号在本期没有销售，在“销售汇总”中查找不到就返回了错误值。

D2      fx      =VLOOKUP(A2,\$G\$1:\$H\$13,2,0)								
	A	B	C	D	E	F	G	H
1	货号	期初结存	本期进货	本期销售	期末结存		货号	销售汇总
2	E091	557	184	315	426		E049	263
3	A038	524	125	285	364		C012	287
4	A045	665	158	#N/A	#N/A		B080	319
5	A057	527	172	394	305		A057	394
6	B030	724	139	297	566		E091	315
7	B077	628	194	320	502		E001	286
8	B080	577	162	319	420		F087	324
9	C012	841	112	287	666		D032	382
10	E049	917	185	263	839		A038	285
11	C050	980	109	#N/A	#N/A		B030	297
12	C054	518	183	#N/A	#N/A		B077	320
13	C074	502	198	280	420		C074	280
14	D005	901	191	#N/A	#N/A			
15	D032	931	103	382	652			

图 5.32 VLOOKUP 查找未消错

对于查找函数返回的错误，不加处理不仅表格不够美观，而且会影响到之后的计算结果（如上例的“期末结存”列）。

从 2007 版开始，Excel 提供了一个全新的消错函数 IFERROR，使用这个函数可以快速屏蔽函数计算结果中的错误。

在上例中，在 D2 单元格中录入如下公式，并向下复制，对于之前显示错误值的地方，全部返回 0 值，也不会影响到 E 列“期末结存”的计算结果。

```
=IFERROR(VLOOKUP(A2,$G$1:$H$13,2,0),0)
```

IFERROR 函数有 2 个参数，第一个参数是需要消错的公式，第二个参数是第一参数计算结果出错时返回的内容。运算结果可以理解为第一参数的结果出错则返回第二参数，否则就返回第一参数的运算结果。

## 5.8.2 2003 版中的消错方法

如果用户尚未升级到 2007 版以上，消错会麻烦很多，可以先用 IS 类函数判断是否存在错误，然后再外套 IF 函数处理，上述公式需更改为如下：

```
=IF(ISNA(VLOOKUP(A2,$G$1:$H$13,2,0)),0,VLOOKUP(A2,$G$1:$H$13,2,0))
```

其中错误的公式如 VLOOKUP 函数要出现 2 次，明显不如使用 IFERROR 函数消错简洁。

职场点睛：学会对函数公式运算结果消错很重要，比较一下消错前后的表格，消错后的表格不仅美观，还会给人留下表格制作者很专业的印象。

本节示例文件：《5.8 查找结果报错的处理.xlsx》

## 5.9 查找引用函数之实战应用

Excel 学习贵在实战应用，本节将举一些查找引用函数的常见应用，帮助加深理解。

### 5.9.1 计算分类百分比

如图 5.33，要求在 D2 单元格中录入一个公式，向下复制到 D13 单元格，完成各品类水果在各产地中的数量占比。

	A	B	C	D	E	F	G
1	品类	产地	数量	占比		公式分析	
2	苹果	山东	1	17%		C2/C5	
3		日本	2	33%		C3/C5	
4		美国	3	50%		C4/C5	
5	苹果小计		6	100%		C5/C5	
6	梨子	本地	4	18%		C6/C10	
7		日本	5	23%		C7/C10	
8		广西	6	27%		C8/C10	
9		云南	7	32%		C9/C10	
10	梨子小计		22	100%		C10/C10	
11	桃子	广西	8	47%		C11/C13	
12		云南	9	53%		C12/C13	
13	桃子小计		17	100%		C13/C13	
14							
15	在D2单元格写一个公式，下拉至D13，完成占比计算。						

图 5.33 计算分类百分比

#### 1. 问题分析

数据透视表中有分类百分比的功能，使用函数计算也不难。分析公式就是一个简单的除法，被除数依次为 C 列的单元格引用，从 C2 单元格到 C13 单元格，但除数不一样，如图 5.33 中 F 列分析。

如果能使用一个函数查找引用到 C5、C10、C13 单元格，问题就简单了。总结数据区域的规律如下：

- ❑ 除数引用的单元格区域是分段变化的，第一段是 C5、第二段是 C10、最后一段是 C13；
- ❑ 除数引用单元格在 A 列的文字都有“小计”两个字。

对于返回单元格引用分段变化，可以考虑改变公式中的单元格引用类型，在公式向下复制时引用不同的单元格区域。

对于查找值如有共性内容可以考虑在查找中使用通配符。

#### 2. 解决方案

在 D2 单元格中录入如下公式，并向下复制完成。

```
=C2/VLOOKUP("*小计",A2:C$13,3,)
```

#### 3. 公式解析

公式中的除数是一个 VLOOKUP 函数，第 1 参数使用“\*小计”。这里的“\*”是通配符，代表查找以任意字符开头，“小计”结束的字符串。

VLOOKUP 函数第 2 参数使用了“A2:C\$13”区域，上限没有锁定，随着公式向下复制，引用区域会依次变为“A3:C\$13”、“A4:C\$13”、“A5:C\$13”、...“A12:C\$13”、“A13:C\$13”，实现查找返回值的分段变化。



## 5.9.2 查找客户首次还款月份

如图 5.34，需要在 J 列填列各个客户当年第一笔还款的月份。

	A	B	C	D	E	F	G	H	I	J
1	客户	1月	2月	3月	4月	5月	6月	7月	8月	首次还款月份
2	客户A			4500				800		3月
3	客户B						9000		800	6月
4	客户C	344			6000					1月
5	客户D							5000		7月
6	客户E		500							2月

图 5.34 首次还款月份

### 1. 问题分析

这是一个查找问题，一般查找第一个（或者是唯一的）可以想到精确查找，查找最后一个可以想到模糊查找。这里可以使用 MATCH 函数返回第一笔还款所在的列数，然后外套 INDEX 函数；也可以在查找到第一笔还款所在的列数后连接上字符“月”。

### 2. 解决方案

在 J2 单元格录入如下公式，按三键结束，向下复制完成。

```
=MATCH(0/B2:I2,)&"月"
```

### 3. 公式解析

数据区域中有很多空格，这里使用“0/B2:I2”将空格变成了错误值#DIV/0!，其他数值返回 0。选中 J2 单元格，在编辑栏中选取这一段公式“0/B2:I2”，按 F9 后返回结果为：

```
{#DIV/0!,#DIV/0!,0,#DIV/0!,#DIV/0!,#DIV/0!,0,#DIV/0!}。
```

将这一段公式放到 B8：I8 单元格中，看起来会更加清楚，如图 5.35 所示。

	A	B	C	D	E	F	G	H	I	J
1	客户	1月	2月	3月	4月	5月	6月	7月	8月	首次还款月份
2	客户A			4500				800		3月
3	客户B						9000		800	6月
4	客户C	344			6000					1月
5	客户D							5000		7月
6	客户E		500							2月
7										
8		#DIV/0!	#DIV/0!	0	#DIV/0!	#DIV/0!	#DIV/0!	0	#DIV/0!	
9										
10	公式：	{=0/B2:I2}								

图 5.35 公式分解运算结果

这里使用的 MATCH 函数是精确查找，第一个参数在逗号之前省略，相当于查找 0，即返回第一个 0 所在的位置 3，然后用连接符&连接上“月”，即得到首次还款所在月份。

使用连接符连接“月”属于取巧，如果 B1：I1 单元格区域中的月份不是从 1 月份开始，上述公式会返回错误的结果。标准的公式如下：

```
=INDEX(B$1:I$1,MATCH(0/B2:I2,))
```

这也是一个数组公式，需要按三键结束。MATCH 部分理解同上，外套 INDEX 函数在 B\$1:I\$1 单元格区域中返回 MATCH 函数查找得到位置的月份。使用 B\$1:I\$1 的写法是考虑到公式要向下复制，需要锁定所在的行。

5.9.3 另类的多条件查找

世界之大真可谓无奇不有！有些用户做的表格非常奇怪，Excel 中很多技巧就是来源于不规范的表格。如图 5.36 所示，A：D 是数据源，一共有 1,044 行，G3：I13 单元格中是各个供应商的进货数量，在 J 列中对于有进货的供应商查找对应物料代码的进货价格。

	A	B	C	D	E	F	G	H	I	J
1	日期	供应商	物料代码	单价		代码	供应商			单价
2	2007/8/7	WZ	0409.0122	14.4			WZ	DY	GL	
3	2007/8/7	WZ	0409.0123	9.36		0409.0297	1			60.35
4	2007/8/7	WZ	0409.0124	7.37		0409.1305			1	4075
5	2007/8/7	WZ	0409.0086	91.88		0409.0343	1			444.02
6	2007-08-08	WZ	0409.0125	207.73		0409.0547	29			10.99
7	2007-08-08	WZ	0409.0126	77.58		0409.0356	1			12.58
8	2007-08-08	WZ	0409.0127	6.35		0409.0136	2			67.33
9	2007-08-08	WZ	0409.0128	109.28		0409.0440	2			139.23
10	2007-08-08	WZ	0409.0129	41.34		0409.1021		2		893
11	2007-08-08	WZ	0409.0130	239.24		0409.0571	1			84.59
12	2007-08-08	WZ	0409.0131	88.8		0409.1020		1		686
13	2007-08-08	WZ	0409.0132	336.71		0409.1069		3		2048
1043	2012-05-16	GL	0409.1574	52235						
1044	2012-05-16	GL	0409.1575	7585						

图 5.36 另类的多条件查找

1. 问题分析

在查找区域中如果把供应商单独设置成一列，原来 3 列的进货数量可以仅使用一列，问题就变成了一个典型的多条件查找，根本没有悬念，如图 5.37 所示。

	A	B	C	D	E	F	G	H	I
1	日期	供应商	物料代码	单价		代码	供应商	数量	单价
2	2007/8/7	WZ	0409.0122	14.4		0409.0297	WZ	1	
3	2007/8/7	WZ	0409.0123	9.36		0409.1305	SL	1	
4	2007/8/7	WZ	0409.0124	7.37		0409.0343	WZ	1	
5	2007/8/7	WZ	0409.0086	91.88		0409.0547	WZ	29	
6	2007-08-08	WZ	0409.0125	207.73		0409.0356	WZ	1	
7	2007-08-08	WZ	0409.0126	77.58		0409.0136	WZ	2	
8	2007-08-08	WZ	0409.0127	6.35		0409.0440	WZ	2	
9	2007-08-08	WZ	0409.0128	109.28		0409.1021	DY	2	
10	2007-08-08	WZ	0409.0129	41.34		0409.0571	WZ	1	
11	2007-08-08	WZ	0409.0130	239.24		0409.1020	DY	1	
12	2007-08-08	WZ	0409.0131	88.8		0409.1069	DY	3	
13	2007-08-08	WZ	0409.0132	336.71					
1043	2012-05-16	GL	0409.1574	52235					
1044	2012-05-16	GL	0409.1575	7585					

图 5.37 规范的表格

要查找的表格虽然不符合制表规则，但还是有规律可循的。如果能在 J3 单元格公式向下复制时引用的供应商名称会随之变动，还将回到多条件查找的套路上。

再次引用一下函数帮助文件中的经典语句——“如果需要更改公式中对单元格的引用，而不更改公式本身，请使用函数 INDIRECT”。

2. 解决方案

在 J3 单元格中录入如下公式，按三键结束，向下复制完成。

```
=INDEX(D$2:D$1044,MATCH(INDIRECT("R2C"&SUM((G3:I3<>"")*{7,8,9})),&F3,B$2:B$1044&C$2:C$1044,))
```

### 3. 公式解析

这个公式外层是 INDEX+MATCH 函数组合,关键在于 MATCH 函数的第 1 个参数使用了 INDIRECT 函数的 R1C1 引用。

公式的核心部分是“SUM((G3:I3<>"")\*{7,8,9})”，这一段的含义是判断 G3: I3 单元格区域是否非空,如非空则乘上其所在列的序号,然后外套 SUM 函数求和。常量数组{7,8,9}代表 G3: I3 单元格所在的列号, SUM 求和的结果就是需要查找引用的供应商名称所在单元格的列号。

在 J3 单元格中进入编辑栏,选取该段公式,按 F9 的返回结果是 7, INDIRECT 函数的引用为“R2C7”,即引用供应商名称所在的 G2 单元格。这个引用会随着公式向下复制而变化,比如在 J4 单元格中, INDIRECT 函数引用的是“R2C9”,即 I2 单元格。

后面的问题就简单了,再连接物料代码所在的 F3 单元格,一个典型的 INDEX+MATCH 多条件查找。

## 5.9.4 在合并单元格中查找

相对更普遍的是使用合并单元格。在数据源表中,合并单元格是绝对禁止的,因正常情况下合并单元格仅左上角单元格中有数值,其他都是空值。使用合并单元格会给后续的工作带来很多麻烦。

如图 5.38 所示, A: C 列为数据源,其中 A2: A10 单元格区域有合并单元格。E2 和 F2 单元格设置了数据有效性下拉列表,需要在 G2 单元格中查询对应月份、品牌的销售金额。

	A	B	C	D	E	F	G
1	月份	品牌	金额		月份	品牌	金额
2	3月	东日	64,650		3月	中天	69,753
3		中天	69,753				
4		富鑫	55,798				
5	4月	东日	78,494				
6		中天	31,552				
7		富鑫	13,133				
8	5月	东日	33,828				
9		中天	49,654				
10		富鑫	52,002				

图 5.38 在合并单元格中查找

### 1. 问题分析

如果 A 列没有合并单元格,即 A2: A10 单元格区域中每个单元格都是有值的,这就是一个简单的多条件查找问题,使用类似“=LOOKUP(1,0/((A2:A10=E2)\*(B2:B10=F2)),C2:C10)”之类的公式即可解决。但在有合并单元格的情况下,查找各个月份第 2 个品牌以下都会出错(如查找品牌“中天”、“富鑫”时公式会出错)。

如果能用函数公式构建一个内存数组,里面是月份数据,填充完整 A2: A10 单元格区域中的空值,问题还是能回到多条件查找上去。

### 2. 解决方案

在 G2 单元格中录入如下公式,按三键结束。

```
=LOOKUP(1,0/((LOOKUP(ROW(2:10),IF(A2:A10<>"",ROW(2:10)),A2:A10)=E2)*(B2:B10=F2)),C2:C10)
```

### 3. 公式解析

这个公式的外层是一个 LOOKUP (1,0) 函数组合，之前已多次介绍，关键在于 LOOKUP 函数的第 2 参数的写法，最难理解的是“LOOKUP(ROW(2:10),IF(A2:A10<>"",ROW(2:10)),A2:A10)”这一段。使用 F9 键查看的结果是{"3月";"3月";"3月";"4月";"4月";"4月";"5月";"5月";"5月"}，相当于将 A2: A10 单元格区域中的空格全部用月份来填充。公式分解如图 5.39 所示。

	I	J	K	L	M	N	O	P	Q	R
1	①	②	③		④		①	{=ROW(2:10)}		
2	2	2	3月		3月		②	{=IF(A2:A10<>"",ROW(2:10))}		
3	3	FALSE	0		3月		③	{=A2:A10}		
4	4	FALSE	0		3月					
5	5	5	4月		4月		④	=lookup(①, ②, ③)		
6	6	FALSE	0		4月					
7	7	FALSE	0		4月					
8	8	8	5月		5月					
9	9	FALSE	0		5月					
10	10	FALSE	0		5月					

图 5.39 LOOKUP 函数填充合并单元格中空值

公式“IF(A2:A10<>"",ROW(2:10))”先对合并单元格区域判断是否非空，如非空值返回其所在的行号，空值返回 FALSE，然后将返回的内存数组{2;FALSE;FALSE;5;FALSE;FALSE;8;FALSE;FALSE}作为外层 LOOKUP 函数的第 2 个参数。

外层 LOOKUP 函数的第 1 个参数使用一个顺序排列的行数组“ROW(2:10)”，依次在第 2 个参数中查找，如查找到则返回第 3 参数中对应位置的值；对于查找不到的数值，如查找不到 3，则找到比 3 小的最大值 2，在第 3 参数中返回对应位置的值“3 月”。其他查找结果的理解以此类推。

理解了这一段公式，后面的就简单了，回到 LOOKUP (1,0) 用模糊查找做精确查找工作的套路上。

### 5.9.5 制作工资条

很多中小型企业没有上薪资系统，工资表通过 Excel 编制。如需给员工发放工资条，就需要对原有的工资表做一些加工。方法有很多，使用查找引用函数就是一种值得推荐的方法，优点在于一次设置完成之后，以后月份通过公式复制即可完成。

如已有图 5.40 所示的工资表，每月需生成如图 5.41 的工资条，一个公式向右、向下复制完成。

	A	B	C	D	E	F	G	H	I
1	工号	姓名	基本工资	岗位薪资	业绩提成	代扣社保	应纳税额	个人所得税	实发薪资
2	A001	越俊	1,000	6,000	400	1,332	6,068	151.80	5,916.20
3	A002	崔亨	1,000	6,000	1,500	1,530	6,970	242.00	6,728.00
4	A003	于成中	1,000	5,000	400	1,152	5,248	69.80	5,178.20
5	A004	狄影淑	1,000	6,000	1,500	1,530	6,970	242.00	6,728.00
6	A005	糜彩青	1,000	4,000	2,000	1,260	5,740	119.00	5,621.00
7	A006	秋彩伊	1,000	5,000	1,500	1,350	6,150	160.00	5,990.00
8	A007	裘健栋	1,000	5,000	400	1,152	5,248	69.80	5,178.20
9	A008	邵育华	1,000	4,000	2,000	1,260	5,740	119.00	5,621.00
10	A009	彭嘉玲	1,000	6,000	800	1,404	6,396	184.60	6,211.40

图 5.40 工资表

	A	B	C	D	E	F	G	H	I
1	工号	姓名	基本工资	岗位薪资	业绩提成	代扣社保	应纳税额	个人所得税	实发薪资
2	A001	越俊	1,000	6,000	400	1,332	6,068	151.80	5,916.20
3	工号	姓名	基本工资	岗位薪资	业绩提成	代扣社保	应纳税额	个人所得税	实发薪资
4	A002	崔亨	1,000	6,000	1,500	1,530	6,970	242.00	6,728.00
5	工号	姓名	基本工资	岗位薪资	业绩提成	代扣社保	应纳税额	个人所得税	实发薪资
6	A003	于成中	1,000	5,000	400	1,152	5,248	69.80	5,178.20
7	工号	姓名	基本工资	岗位薪资	业绩提成	代扣社保	应纳税额	个人所得税	实发薪资
8	A004	狄影淑	1,000	6,000	1,500	1,530	6,970	242.00	6,728.00

图 5.41 工资条 1

### 1. 问题分析

这是一个引用问题，工资条中内容全部来自工资表，奇数行引用工资表中的第 1 行，偶数行依次引用工资表中第 2 行及以下。因在 A1 单元格的公式向下复制时改变的是当前行的行号，可以通过行号的变化来做判断。

### 2. 解决方案

在“工资条”工作表的 A1 单元格录入如下公式，并向右、向下复制完成。

```
=IF(MOD(ROW(),2),工资表!A$1,OFFSET(工资表!A$1,ROW()/2,))
```

### 3. 公式解析

公式外层是一个 IF 函数，IF 函数第 1 参数使用了 MOD 函数“MOD(ROW(),2)”。MOD 函数会返回两个数相除的余数，在第 1 参数是行号（自然数），第 2 参数为 2 的情况下，返回值非 0 即 1。

在奇数行时，MOD 函数返回值为 1，IF 函数返回第 2 参数的值，即引用“工资表”工作表的第 1 行；在偶数行时，MOD 函数返回值为 0，IF 函数返回第 3 参数的值。

这里 IF 函数的第 3 参数是一个 OFFSET 函数，表示自“工资表”工作表 A\$1 单元格起向下偏移的返回值，“ROW()/2”代表如公式在“工资条”工作表的 2、4、6 行中，OFFSET 函数分别向下偏移 1、2、3 行，返回第 2、3、4 行的值。因公式中对于 A1 单元格是行绝对列相对的混合引用，在公式向右复制时会改变起点单元格的列。

备注：如在公式复制时需要带着格式，可以在复制完前 2 行后完成格式的设置，然后全选这两行再向下复制。

如制作的工资条需要间隔一个空行，可以使用如下公式，效果如图 5.42 所示。

```
=CHOOSE(MOD(ROW(),3)+1,"",工资表!A$1,OFFSET(工资表!A$1,(ROW()+1)/3,))
```

	A	B	C	D	E	F	G	H	I
1	工号	姓名	基本工资	岗位薪资	业绩提成	代扣社保	应纳税额	个人所得税	实发薪资
2	A001	越俊	1,000	6,000	400	1,332	6,068	151.80	5,916.20
3									
4	工号	姓名	基本工资	岗位薪资	业绩提成	代扣社保	应纳税额	个人所得税	实发薪资
5	A002	崔亨	1,000	6,000	1,500	1,530	6,970	242.00	6,728.00
6									

图 5.42 工资条 2

这个公式外层是 CHOOSE 函数，效果类似于 IF 函数，相对于 IF 函数处理多条件判断只能嵌套，CHOOSE 可通过第 1 参数的返回值变化来实现多条件判断取值。

这里 CHOOSE 函数第 1 参数还是使用了 MOD 函数，一个自然数被 3 除，余数只有三种可能，即 0、1、2，加上 1 为 1、2、3；第 1 参数为 1 取 CHOOSE 函数的第 2 参数，为 2 取 CHOOSE 函数的第 3 参数，...

---

职场点睛：类似的方法还可以应用于隔行插入，学会这一招，无疑会大幅度提升工作效率。  
本节示例文件：《5.9 查找引用函数之实战应用.xlsx》

## 5.10 小结

本章从 VLOOKUP 函数开始，依次介绍了 INDEX、MATCH、LOOKUP、INDIRECT、OFFSET 等常用的查找与引用函数以及最经典的 VLOOKUP{1,0}、INDEX+MATCH 及 LOOKUP(1,0)函数套路。

函数的魅力在于解决问题方案的多样性，通过本章的示例可以发现很多问题都能用多个方案解决。学习函数的乐趣在于能利用函数解决实际工作中的问题，但有时候也要想办法让复杂的函数公式无用武之地，比如规范数据源之后，一些复杂的问题就能用最常见的套路解决。应尽可能从源头上把复杂的问题简单化。

因篇幅所限，函数公式举例仅是浅尝即止，有点意犹未尽。比如在第 1 节中谈到了 VLOOKUP 函数的局限性，其中前 3 条都已经通过其他函数组合解决了，唯有最后一条——“不能用一个公式返回所有满足条件的查找值”的问题尚未解决，将在第 8 章中重点讲解。