

## 第 6 章 掀起你的盖头来——理解日期时间函数

日常工作中难免会接触日期与时间，如不理解日期与时间的实质，涉及到的计算会非常头痛。本章从日期与时间的实质谈起，讲解一些最常用的日期时间函数及其应用，帮助读者提高处理日期时间数据的工作效率。

### 6.1 认识日期与时间

不要把日期与时间想得太复杂，其实就是一个数字。日期是一个整数，时间是一个小数，当然还有取值的范围。

#### 6.1.1 日期与时间的实质

很多用户制表的习惯是在录入完成后设置数字显示的格式。一个错误的操作习惯是先全选工作表，然后设置格式，看看这样做会发生什么。

##### 1. 数字与日期时间的显示可以相互转换

图 6.1 的例子是在录入数据完成之后先全选工作表，然后设置了会计专用格式。

	A	B	C	D	E	F	G
1	资产大类	品名	购置或使用日期 (×年×月×日)	原值或评 估净值	每月折旧额	净值	C列日期
2	厂房及机器	0.5吨燃油锅炉	40,543.00	82,000.00	615.00	70,930.00	2010/12/31
3	厂房及机器	1.5T站驾前移式电动叉	40,543.00	690,333.44	5,177.50	597,138.44	2010/12/31
4	家具装置及设备	1.5匹华宝空调	39,323.00	2,300.00	34.50	299.00	2007/8/29
5	家具装置及设备	15楼办公室CISCO无线网络设备	40,633.00	32,521.37	903.37	18,067.44	2011/3/31
6	租赁改良	16、17、19楼装修工程款	40,896.00	196,833.45	3,280.56	173,869.54	2011/12/19
7	家具装置及设备	17F机房新增监控系统	40,906.00	67,155.10	1,865.42	54,097.17	2011/12/29

图 6.1 显示为数字的日期

C 列原先录入的是日期，通过设置会计专用格式后转化成了数字。这是一个正向的例子，如果把常规的数字格式改变成日期格式，结果又会如何呢？

在本书第 3 章中介绍“数字 72 变”时举过一个例子，一个数字设置成 Excel 内置的 10 多种不同的单元格格式，显示结果各不相同，这些格式中就有日期格式和时间格式，如图 6.2 所示。

	A	B	C
1	数字	显示	格式
2	12345	1933/10/18	日期
3	12345.6	1933/10/18	日期
4	12345	0:00:00	时间
5	12346.6	14:24:00	时间

图 6.2 数字显示为日期及时间格式

图 6.2 中 A2 和 A3 单元格中数字不同，但显示为日期格式时返回结果相同；而 A4 单元格中一个整数显示为时间格式时返回结果为 0。

通过上述 2 个例子，可以发现日期与时间和数字有着密切的联系。日期像是一个数字的整数部分，而时间则是数字的小数部分。

以上仅是推测，看一下微软帮助文件中的讲解。

## 2. Excel 中的日期与时间规则

在 Excel 中默认使用 1900 日期系统，以 1900 年 1 月 1 日为系统默认的第 1 天，这一天的日期序列号为 1，每增加一天日期的序列号加 1，如 1900 年 1 月 2 日的日期的序列号为 2，……其他依此类推。系统中最大的日期是 9999 年 12 月 31 日（序列号为 2,958,466），在起止期间的日期都能与数字相互转换，反之不然。

	A	B	C	D	E	F
1	数字	日期	文本长度	C列公式		
2	0	1900/1/0	1	=LEN(B2)		
3	1	1900/1/1	1	=LEN(B3)		
4	2	1900/1/2	1	=LEN(B4)		
5	10000	1927/5/18	5	=LEN(B5)		
6	40000	2009/7/6	5	=LEN(B6)		
7	41000	2012/4/1	5	=LEN(B7)		
8	2958465	9999/12/31	7	=LEN(B8)		
9	2958466	#####	7	=LEN(B9)		
10	-1	#####	2	=LEN(B10)		

超过日期范围时转换成数字会出错

图 6.3 将数字转换成日期格式

图 6.3 是一个将数字转换成日期的例子，在 0-2,958,466 之间的数字都能转换成日期格式，如超出这个范围 Excel 返回值就会报错。这个例子验证了日期是有范围的，不仅是一个整数，而且还是一个不超过 2,958,466 的正整数。

理解了日期之后，时间就简单了。每增加一天，日期的序列号增加 1，一天有 24 个小时，小时折算到天是 1/24 天，一小时有 60 分钟，一分钟有 60 秒，折算方法是继续除以 60。

## 3. 判断日期时间与文本

日期的特点是每个月的天数不一样，可能有 28、29、30、31 这 4 种情况。如果搞错了当月的天数，可能会写出一个错误的日期值。

	A	B	C	D	E	F
1	日期	数值	长度	公式		
2	2012/2/26	40965	5	=LEN(A2)		
3	2012/2/27	40966	5	=LEN(A3)		
4	2012/2/28	40967	5	=LEN(A4)		
5	2012/2/29	40968	5	=LEN(A5)		
6	2012/2/30	2012/2/30	9	=LEN(A6)		
7	2012/2/31	2012/2/31	9	=LEN(A7)		
8						
9						

数值右对齐，而文本是左对齐的。

图 6.4 判断日期与文本

图 6.4 中最后 2 个日期是错误的，2012 年 2 月份只有 29 天。Excel 对于超出正常范围的日期仍会以原录入格式显示，区别于规范的日期，超出范围的日期仅是一个文本，在单元格中靠左对齐显示。

职场点睛：在实际工作中如使用了错误的日期格式会给后续工作带来麻烦，在未设置格式对齐的情况下，观察日期的对齐方式是最简单的判断真假日期的方法。

还有一种方法是使用 LEN 函数求出字符的长度，目前日期序列号都是 5 位数的数字（比如 2012 年 6 月 30 日的日期序列号为 41,090），使用 LEN 函数返回值为 5。例如，LEN 函数返回值不是 5，说明单元格中的日期实际上是一个文本，如上图中最后 2 个“日期”的字符长度都是 9 位，就不属于数字范畴了。

#### 4. 转换日期时间与数值

如是标准的日期时间格式，其本身就是一个数字，可以直接参与数值运算。如是文本状态的日期或者时间，转换成数字序列号的方法有如下几种：

- ❑ 通过分列的方法转换成数值（第 3 章第 2 节中有过介绍）；
- ❑ 参与一次数值运算的方式转换成日期时间序列号，比如原数值\*1；
- ❑ 还可以通过 Excel 内置的函数转换。

	A	B	C	D
1	文本日期时间	数值	公式	备注
2	2012-6-18	41078	=DATEVALUE(A2)	
3	6-18	41078	=DATEVALUE(A3)	可以省略系统年份
4	12-6-18	41078	=DATEVALUE(A4)	可以使用2位数表示年份
5		41078	=DATEVALUE("2012-6-18")	可以用文本字符串表示日期
6		41078	=DATEVALUE("6-18")	
7		41078	=DATEVALUE("12-6-18")	
8	10000	#VALUE!	=DATEVALUE(A8)	参数必须是文本日期
9	9999-12-32	#VALUE!	=DATEVALUE(A9)	超出系统日期范围
10	6-18 9:18:36	41078	=DATEVALUE(A10)	仅返回日期部分
11	9:18:36	0	=DATEVALUE(A11)	对文本时间返回0
12				
13	6-18 9:18:36	0.387916667	=TIMEVALUE(A13)	仅返回时间部分
14	9:18:36	0.387916667	=TIMEVALUE(A14)	
15		0.387916667	=TIMEVALUE("9:18:36")	可以用文本字符串表示时间

图 6.5 使用函数转换文本日期时间为数值序列号

图 6.5 使用了函数将文本状态的日期与时间转换成了数值序列号，其中 DATEVALUE 函数可以转换文本日期，TIMEVALUE 函数可以转换文本时间。

通过上述举例可以发现如下规律：

- ❑ 在 Excel 中表示日期时如省略年份，会自动以系统年份代替；
- ❑ 函数的参数值必须是一个文本的日期时间值，不能是文本的数字，也不能超出日期时间的范围；
- ❑ 可以在函数参数中直接使用以一对引号括起来的日期或者时间。

尤其是最后一条规律非常重要，不仅是这两个函数的参数中可以使用一对引号括起来的文本状态的日期和时间，其他提取日期时间信息的函数中也可以，而且还可以省略或者简写系统当前年份。

6.1.2 解读日期与时间格式代码

本书第 3 章中介绍过数字的自定义格式代码，而日期与时间本身也是数字，自然适用之前介绍的自定义格式代码。除此之外，日期与时间还有一些特殊的格式代码。

表 6.1 日期常用自定义数字格式代码注释

代码	代码注释
yy	将年份显示为 00 - 99
yyyy	将年份显示为 1900 - 9999
m	将月份显示为 1 - 12
mm	将月份显示为 01 - 12
mmm	将月份显示为 Jan - Dec
mmmm	将月份显示为 January - December
mmmmm	将月份显示为该月份的第一个字母
d	将日期显示为 1 - 31
dd	将日期显示为 01 - 31
ddd	将日期显示为 Sun - Sat
dddd	将日期显示为 Sunday - Saturday
aaaa	使用中文来显示星期几（星期一一星期日），常用于TEXT函数
aaa	使用中文显示星期几（一一日），不显示“星期”两个字
e	同代码“yyyy”，将年份显示为 1900 - 9999

表 6.1 显示了与日期相关的自定义数字代码格式，下面以一个时间“2012-6-18 9:04:08”来看一下具体的显示变化。

	A	B	C	D
1	日期	更改格式后显示内容	自定义数字格式	备注
2	2012/6/18 09:04:08	2012	yyyy	
3	2012/6/18 09:04:08	12	yy	
4	2012/6/18 09:04:08	6	m	
5	2012/6/18 09:04:08	06	mm	
6	2012/6/18 09:04:08	Jun	mmm	
7	2012/6/18 09:04:08	June	mmmm	
8	2012/6/18 09:04:08	J	mmmmm	
9	2012/6/18 09:04:08	18	d	
10	2012/6/18 09:04:08	18	dd	
11	2012/6/18 09:04:08	Mon	ddd	周一
12	2012/6/18 09:04:08	Monday	dddd	
13	2012/6/18 09:04:08	星期一	aaaa	
14	2012/6/18 09:04:08	一	aaa	
15	2012/6/18 09:04:08	2012	e	
16	2012/6/18 09:04:08	二〇一二年〇六月十八日	[DBNum1]yyyy年mm月dd日	中文小写日期
17	2012/6/18 09:04:08	贰零壹贰年零陆月壹拾捌日	[DBNum1]yyyy年mm月dd日	中文大写日期
18	2012/6/18 09:04:08	2012年 June 18 星期一	e年 mmmm dd aaaa	各种格式的排列组合

图 6.6 日期的不同显示

图 6.6 中 2-15 行展示了对于同一个日期，使用不同的自定义格式代码后显示的内容，而 16-18 行显示了一些代码组合，可见对于日期，在单元格中的变化也是无穷的。

表 6.2 时间常用自定义数字格式代码注释

代码	代码注释
h	将小时显示为 0 - 23
hh	将小时显示为 00 - 23

m	将分钟显示为 0 - 59 ， 必须与h、s或[]配合使用
mm	将分钟显示为 00 - 59 ， 必须与h、s或[]配合使用
s	将秒显示为 0 - 59
ss	将秒显示为 00 - 59
[]	显示超出进制的时间（如大于24的小时数或者大于60的分与秒）
h AM/PM	使小时显示类似于 4 AM
h:mm AM/PM	使时间显示类似于 4:36 PM
h:mm:ss A/P	使时间显示类似于 4:36:03 P
[h]:mm	按小时计算的一段时间，如 25.02
[mm]:ss	按分钟计算的一段时间，如 63:46
[ss]	按秒计算的一段时间
h:mm:ss.00	百分之几秒

相对于日期的自定义数字格式代码，时间同样很精彩。表 6.2 显示了与时间相关的自定义数字格式代码，下面以一个时间“2012-6-18 9:04:08”来看一下具体的显示变化。

	A	B	C	D
1	日期	更改格式后显示内容	自定义数字格式	备注
2	2012/6/18 09:04:08	9	h	
3	2012/6/18 09:04:08	09	hh	
4	2012/6/18 09:04:08	8	s	
5	2012/6/18 09:04:08	08	ss	
6	2012/6/18 09:04:08	9 AM	h AM/PM	
7	2012/6/18 09:04:08	9:04 AM	h:mm AM/PM	
8	2012/6/18 09:04:08	9:04:08 A	h:mm:ss A/P	
9	2012/6/18 09:04:08	985881:04	[h]:mm	计算了从1900年1月1日开始经历的小时数
10	2012/6/18 09:04:08	59152864:08	[mm]:ss	计算了从1900年1月1日开始经历的分钟数
11	2012/6/18 09:04:08	3549171848	[ss]	计算了从1900年1月1日开始经历的秒数

图 6.7 时间的不同显示

图 6.7 中，一个日期也会有不同的显示变化，相对于时间，在一般情况下日期仅根据数字的小数部分显示。在代码使用了[]时，Excel 才会从 1900 年 1 月 1 日开始计算相关的小时、分钟和秒。

职场点睛：如需要在 Excel 中展示一些特殊格式的日期或者时间，可以尝试在录入标准格式的日期和时间之后使用单元格自定义格式，这样得到的还是一个数值，便于后续计算分析。

6.1.3 Excel 中的易容术与整容术

使用单元格自定义格式代码实现数字的所见非所得在本书中称为 Excel 中的易容术，相对于易容术，还有一种彻底改变单元格中返回值的方法，即 Excel 中的整容术。

1. 易容术的局限性

介绍 Excel 中的整容术，自然和易容术不够完美有关。易容术虽然很奇妙，但其并未改变一个数字的实质，看到的仅是表层化妆易容之后的内容。比如把一个美女化妆成一个帅哥，看上去是帅哥，实质上还是美女。Excel 中也是如此，请看如下例子。

一个时间“2012-6-18 09:04:08”，通过自定义格式转换成数字“2012”，如需引用已转换后的数字并连接上一个文字“年”，结果如图 6.8 所示。


B3			=B2&"年"	
	A	B	C	D
1	日期	更改格式后显示内容	自定义数字格式	
2	2012/6/18 09:04:08	2012	yyyy	
3		41078.3778703704年		
4				

图 6.8 易容术的局限性

图 6.8 中 B3 单元格中的公式是“=B2&"年"”，就是在 B2 单元格内容后面连接一个“年”字，但显示结果并未如愿。原因在于 B2 单元格中运用了 Excel 的易容术，仅是改变了显示，日期还是日期，并未真正变成“2012”。在使用连接符&时，相当执行了一次计算，使用了真正的数字序列号，这样才会得到 B3 单元格中的结果。

类似的问题不仅存在于公式运用中，如要把 B2 单元格中的显示值“2012”直接复制出来，还是有一定难度的，如直接使用复制，因会带出格式，显示值是 2012，实际还是一个数值，如选择性复制，则会直接显示日期时间序列号，如图 6.9 所示。

B5		fx 2012/6/18 9:04:08		
	A	B	C	D
1	日期	更改格式后显示内容	自定义数字格式	
2	2012/6/18 09:04:08	2012	yyyy	
3		41078.3778703704年		
4				
5	直接复制B2单元格	2012		
6				
7	复制B2单元格的数值	41078.37787		

图 6.9 使用易容术后无法简单复制得到单元格显示的内容

## 2. Excel 中的整容术——TEXT 函数

解决的方案是使用 Excel 中的整容术——TEXT 函数，如图 6.10 所示。


C2			<div>=TEXT(A2,"yyyy")</div>	
	A	B	C	D
1	日期	更改格式后显示内容	使用TEXT函数	
2	2012/6/18 09:04:08	2012	2012	
3		41078.3778703704年	2012年	
4				

图 6.10 使用 Excel 中的整容术——TEXT 函数

TEXT 函数语法很简单，仅 2 个参数，如表 6.3 所示：

表 6.3 TEXT 函数 2 个参数的理解

参数顺序	参数含义	上例中的取值	注意事项
第1参数	需要改变的数值	A2单元格	可以是数值、公式及单元格的引用
第2参数	需要改变成的数字格式	"yyyy"	必须使用双引号括起来，可参考自定义格式代码

TEXT 函数第 2 参数的写法可以先参考单元格自定义格式代码，使用 TEXT 函数得到的就是显示在



单元格中的内容，我们称之为对第 1 参数做了整容。

### 3. 易容术与整容术的差异

两者还是有所差别的，使用易容术可以把一个美女化妆成一个帅哥，而使用整容术是无法把美女整容成帅哥的。

在 Excel 中，单元格自定义格式和 TEXT 函数最大的差异是无法根据条件改变返回单元格中的颜色显示，就是 TEXT 函数返回的结果就是一个文本，是不带格式的。

在图 6.6 和图 6.7 的举例中，都可以在 E2 单元格录入一个公式，向下复制得到和 B 列使用自定义格式相同的返回结果。

=TEXT(A2,C2)

如果要设置不同的颜色，使用 TEXT 函数是无效的，如图 6.11 所示。

D2		fx		=TEXT(A2,C2)
	A	B	C	D
1	数值	自定义格式	自定义格式代码	TEXT 函数
2	6	6	[红色][<10]0;G/通用格式	6
3	10	10	[红色][<10]0;G/通用格式	10
4	2	2	[红色][<10]0;G/通用格式	2
5	9	9	[红色][<10]0;G/通用格式	9
6	9	9	[红色][<10]0;G/通用格式	9
7	14	14	[红色][<10]0;G/通用格式	14
8	7	7	[红色][<10]0;G/通用格式	7

图 6.11 TEXT 函数无法显示颜色

这里 C 列的自定义格式代码的含义是如单元格中的值小于 10，则显示为红色，其他返回默认的格式。使用 TEXT 函数时会发现，根据值的大小改变颜色的代码对于函数返回结果无效。

本节示例文件：《6.1 认识日期与时间.xlsx》

## 6.2 使用函数提取日期时间信息

Excel 中的日期与时间是一些数字，如何取得这些数字？这些数字背后还隐藏着哪些信息？本节介绍提取日期时间信息相关的函数，从系统日期时间开始，年、月、日、时、分、秒、星期等等。

### 6.2.1 提取当前系统日期与时间

如果需要在 Excel 表格中录入当前系统的日期与时间，不需要在单元格中一个一个数字地录入。有使用函数与快捷键两种方法。但这两种方法录入的都是系统当前的日期与时间，和用户的电脑设置有关，并非所谓的英国格林威治标准时间。

#### 1. 两种录入方法比较

比如当前系统日期为 2012 年 7 月 25 日，时间为下午 14 点 30 分，在单元格中按“Ctrl+;”快捷键及函数“=TODAY()”都能录入当前系统日期；按“Ctrl+Shift+;”快捷键及函数“=NOW()”都能录入当前系统时间，如图 6.12 所示。

	A	B	C	D	E
1	日期		时间		
2	Ctrl+;	TODAY()	Ctrl+Shift+;	NOW()	
3	2012/7/25	2012/7/25	14:30	2012/7/25 14:31	
4					

图 6.12 录入系统当前日期与时间

使用快捷键录入时间和使用 NOW 函数显示的时间有差异，原因在于录入 NOW 函数时，系统时间已经是下午 14 点 31 分了，而使用“Ctrl+Shift+;”快捷键时还是 14 点 30 分。

通过这个例子可以发现，使用快捷键录入的日期与时间在录入后是不会随着系统变化的，而使用 TODAY 函数和 NOW 函数录入的系统日期与时间，会随着系统日期时间变化而变化。打开工作簿、按一次 F9 键或者在 Excel 中执行一次运算，都会改变使用函数录入的系统日期与时间。

## 2. 使用函数提取日期时间的优点

使用函数录入系统日期和时间，相对于快捷键有其优越性，在于返回值会自动更新。有时候自动更新会减少很多工作量，请看下面的例子。

某企业信用部门根据财务部门提供的客户开票应收款资料来决定是否需要办理催款手续。确定催收的逻辑是客户欠款且已过期，如下图 6.13。

	A	B	C	D	E
1	发票号码	到期日期	应收金额	已收金额	是否需催收
2	A12023	2012-6-8	2000	1000	是
3	A12024	2012-7-9	97000	100000	否
4	A12025	2012-9-24	64000	2000	否
5	A12026	2012-5-21	42000	5000	是
6	A12027	2012-9-30	33000	0	否

图 6.13 客户应收账款催收表

### 1) 问题分析

根据确定是否催收的逻辑，需要做两个判断，而欠款是否到期和当前的日期有关。

### 2) 解决方案

在 E2 单元格中录入如下公式，向下复制完成。

```
=IF(AND(TODAY()>B2,C2>D2),"是","否")
```

### 3) 公式解析

公式外层是一个 IF 函数，根据第 1 参数即 AND 函数运算的结果返回是否需要催收的结果。内层 AND 函数有 2 个参数，在所有参数返回值都为 TRUE 的时候，AND 函数返回 TRUE，否则返回 FALSE。

职场点睛：这里使用 TODAY 函数提取系统日期与 B 列的日期做比较，在文件打开时公式会自动重新计算。在实际工作中使用日期函数能避免手工改变当前日期的麻烦。

再看一个例子，计算今天是当年的第几天。某企业预计每天的销售额是 10 万元，系统中已有当年的销售总金额，需要测算一下销售预算是否能达成，如图 6.14 所示。

	A	B	C	D
1	日均销售预算	天数	系统中销售额	业绩达成率
2	100,000	209	21,200,000	101%

图 6.14 计算当天是当年的第几天



#### 4) 问题分析

计算当天是当年的第几天，可以通过当天的日期序列号和当年第一天的日期序列号相减得到。系统当天的日期序列号可以通过 TODAY 函数得到，难点在于取得当年第一天的日期序列号。如需保持公式今后的适用性，是不能直接使用 2012 年 1 月 1 日这样的日期，否则在 2013 年公式就不通用。

#### 5) 解决方案

在 B2 单元格录入如下公式，按回车键结束即可。

```
=TODAY()-"1-1"+1
```

#### 6) 公式解析

在 Excel 中录入日期时如省略了年度，Excel 默认加上系统当前时间的年度。这里输入的“1-1”就相当于“2012-1-1”（目前系统时间为 2012 年）。

当年第 1 天与 TODAY 函数返回的日期序列号相减再加上 1，就能得到当天在本年中的时间序列号。

### 3. 使用 VBA 改版系统自带的 NOW 函数

系统自带的 TODAY 函数和 NOW 函数都属于易失性函数，简单地说就是使用这两个函数得到的结果是靠不住的，会随着系统的日期时间改变而改变。在工作中，按一次 F9，执行一次其他运算，甚至在自动重算开启时在空白单元格中录入都有可能改变这两个函数的返回结果。

自动重算的好处已经举例介绍过，但有时候并不需要 Excel 自动重算的结果，请看下面的例子。

比如某企业售后服务部门需要记录所有的产品质量投诉电话的时间，如图 6.15 所示，在 A 列记录投诉内容，B 列使用公式自动生成当前系统时间。在录入若干记录后会发现，所有的时间都是一样的。

```
=IF(A2<>"" ,NOW(),"")
```

B2		=IF(A2<>"" ,NOW(),"")		
	A	B	C	D
1	投诉内容	时间		
2	系统不稳定	2012-07-25 3:30:22 PM		
3	SD卡读不出	2012-07-25 3:30:22 PM		
4	经常自动关机	2012-07-25 3:30:22 PM		
5				

图 6.15 易失性函数运行结果

原因在于 NOW 函数的易失性，每次录入都会自动重算，最后得到的已不是 A 列数据录入时的时间。要解决这个问题，可以使用 VBA 自定义一个函数，改变 Excel 自带的 NOW 函数的易失性。

按“ALT+F11”快捷键打开 VBE 编辑器，插入一个模块，录入如下代码：

```
Function Nows()  
Dim Tim As String  
    Tim = Format(Now, "yyyy-mm-dd hh:mm:ss")  
    Nows = Tim  
End Function
```

这是一个自定义函数，函数名为 Nows，没有参数，在使用时，在单元格中录入“=Nows()”就能返回当前的系统时间和日期。

这个函数相对于系统自带的 NOW 函数，具有不会自动重算的特点，请看图 6.16 中数据录入结果。

C3		=IF(A3<>"",nows(),"")	
	A	B	C
1	投诉内容	时间	
2		系统自带的NOW函数	自定义函数NOWS
3	系统不稳定	2012-07-25 3:48:17 PM	2012-07-25 15:46:15
4	SD卡读不出	2012-07-25 3:48:17 PM	2012-07-25 15:47:08
5	经常自动关机	2012-07-25 3:48:17 PM	2012-07-25 15:48:16

图 6.16 自定义函数 NOWS 与系统自带函数 NOW 比较

以上解决方案并不完美，在文件打开时还是会自动重算，这里仅提供一种思路。如需实现用函数公式录入当前时间且永远不会改变，可以结合使用 VBA 的工作表事件。在“自动录入时间”工作表的代码窗口中录入如下代码：

```
Private Sub Worksheet_Change(ByVal Target As Range)
    If Target.Column = 1 And Target.Row > 2 Then
        Target.Offset(0, 2) = Target.Offset(0, 2).Value
    End If
End Sub
```

此时在 A 列第 3 行起录入内容，Excel 会自动在 C 列生成当前系统时间，并将公式运算结果数值化，如图 6.17。

	A	B	C
1	投诉内容	时间	
2		系统自带的NOW函数	自定义函数NOWS
3	系统不稳定	2012-07-25 4:23:34 PM	2012-7-25 16:16
4	SD卡读不出	2012-07-25 4:23:34 PM	2012-7-25 16:20
5	经常自动关机	2012-07-25 4:23:34 PM	2012-7-25 16:22
6	电池容量虚标	2012-07-25 4:23:34 PM	2012-7-25 16:23

图 6.17 使用工作表事件后输入时的返回结果

职场点睛：在 Excel 内置的功能无法满足工作要求时，可以考虑尝试一下 VBA，先从简单的几句代码开始。

备注：也可以不使用自定义函数，使用内置的 NOW 函数和工作表事件，代码需要改动。

## 6.2.2 提取常规的日期与时间信息

除了上面介绍的 TODAY 函数和 NOW 函数外，Excel 还提供了 6 个内置的函数，可以得到其他日期与时间信息，如图 6.18 所示。

	A	B	C	D
1	函数	返回值	公式	说明
2	YEAR	2012	=YEAR(NOW())	返回某日期对应的年份。返回值为 1900 到 9999 之间的整数。
3	MONTH	7	=MONTH(NOW())	返回以序列号表示的日期中的月份。月份是介于 1 到 12 之间的整数。
4	DAY	25	=DAY(NOW())	返回以序列号表示的某日期的天数，用整数 1 到 31 表示。
5	HOUR	16	=HOUR(NOW())	返回时间值的小时数。即一个介于 0 (12:00 A.M.) 到 23 (11:00 P.M.) 之间的整数。
6	MINUTE	31	=MINUTE(NOW())	返回时间值中的分钟，为一个介于 0 到 59 之间的整数。
7	SECOND	53	=SECOND(NOW())	返回时间值的秒数。返回的秒数为 0 到 59 之间的整数。

图 6.18 使用函数提取系统中的其他日期时间信息

这 6 个函数都只有 1 个参数，参数是一个日期或者时间值，可以是一个满足日期或者时间条件的数值，也可以是一个可作为日期或者时间参与运算的文本。这几个日期函数都是比较简单的，这里举一些例子来加深理解。

1. 计算上月天数

某企业隔月计算商场促销员薪资。促销员每天工作 4 个小时，按合同规定月度全勤（没有双休日休假）才能拿到全额底薪，如有缺勤按天扣除底薪的 5%，扣完为止。促销员的岗位薪资按出勤天数计发，每天 100 元。

	A	B	C	D	E	F	G
1	月度天数	30					
2							
3	促销员	基础薪资	岗位薪资/天	缺勤天数	应发薪资		
4	促销员A	500	100	3	3125		
5	促销员B	500	100	2	3250		
6	促销员C	600	100		3600		
7	促销员D	600	100	1	3470		
8	促销员E	600	100	4	3080		
9							
10	E4公式：	=MAX(B4*(1-5%*D4),0)+C4*(\$B\$1-D4)					

图 6.19 计算上月天数

图 6.19 中是一个计算促销员薪资的表格模板，D 列数据由人事部门提供，E 列已设置好公式，要求 B2 单元格设置公式，自动取得上个月的天数（薪资隔月发放）。

1) 问题分析

计算上个月有多少天得方法有多种，一种思路是取得上个月最后一天的日期序列号，然后外套 DAY 函数求得这一天的天数序号。每个月最后一天的天数序号即是这个月的天数。

2) 解决方案

在 B2 单元格中录入如下公式，按回车键结束。

=DAY(TODAY()-DAY(TODAY()))

3) 公式解析

公式外层是一个 DAY 函数，内层“TODAY()-DAY(TODAY())”作为外层 DAY 函数的参数。

内层中“DAY(TODAY())”得到的是系统当天的天数序号，再和“TODAY()”相减，返回的是上个月最后一天的序列号。

比如当前时间是 2012 年 7 月 26 日，“DAY(TODAY())”返回值是 26，“TODAY()-DAY(TODAY())”返回值是 41090，即 2012 年 6 月 30 日的日期序列号，再外套 DAY 函数就能得到结果。

2. 根据身份证号码判断生肖

生肖也称属相，是中国和东亚地区的一些民族用来代表年份和人的出生年的十二种动物。生肖的周期为 12 年。每一人在其出生年都有一种动物作为生肖。已知身份证号码，可以算出所属的生肖，当然仅是根据公历年度。

	A	B	C
1	身份证号	出生年份	生肖
2	310109197812281210	1978	马
3	310109731228121	1973	牛
4	310108200107141420	2001	蛇

图 6.20 根据身份证号码计算生肖

#### 1) 问题分析

生肖以 12 年为一个周期，需要推算生肖必须先知道出生年份。身份证号码上有出生年月日信息，根据身份证号码位数不同年份分成 2 位和 4 位两种，公式需要通用于 2 种长度的身份证号码。

#### 2) 解决方案

先在 B2 单元格录入如下公式并向下复制，作为计算生肖的辅助列；

```
=YEAR(TEXT(MID(A2,7,6+(LEN(A2)=18)*2),"0-00-00"))
```

然后在 C2 单元格录入公式，向下复制完成生肖的推算。

```
=MID("猴鸡狗猪鼠牛虎兔龙蛇马羊",MOD(B2,12)+1,1)
```

#### 3) 公式解析

C 列的公式仅是根据生肖的先后顺序规律取值，这里不做讲解，关键要理解 B 列的公式。

B 列公式最内层是一个 MID 函数，取出身份证号码中出生年月日信息，使用 “+(LEN(A2)=18)\*2” 能自适应不同的身份证号码长度。

外套 TEXT 函数，将 MID 函数取出的出生年月日字符串转换成日期格式。最后外套 YEAR 函数求得出生的年份。

### 6.2.3 提取与周相关的信息

Excel 中还有 2 个与周相关的函数，在涉及到分星期的计算时非常有用。中国人的习惯是每周的第一天是周一，最后一天是周日，符合先工作后休息的原则；西方人不一样，惯例是一周的第一天是周日。

考虑到东西方不同的习俗，Excel 中与周计算相关的函数都有第 2 参数，读者可以根据自己的习惯灵活选用。

#### 1. 标注双休日销售额

图 6.21 的表格中记载了某企业 2012 年 5 月份的销售额，需要分析一下双休日的销售情况，在 C 列加辅助列判断是否是双休日。

	A	B	C
1	日期	销售额	是否双休日
26	2012-5-25	75,301	
27	2012-5-26	91,949	是
28	2012-5-27	99,058	是
29	2012-5-28	61,185	
30	2012-5-29	43,969	
31	2012-5-30	69,273	
32	2012-5-31	71,882	

图 6.21 判断是否是双休日

#### 1) 解决方案

在 C2 单元格录入如下公式，向下复制完成。

```
=IF(WEEKDAY(A2,2)>5,"是","")
```

## 2) 公式解析

这个公式外层使用过了 IF 函数，IF 函数第 1 参数用了 WEEKDAY 函数判断返回的星期序号是否大于 5，如大于 5，则 IF 函数返回“是”，否则返回空值。

在 2010 版中，WEEKDAY 函数第 2 参数有 10 种选择，这里选择 2，表示 1 周从周一开始周日结束，周一返回 1，周日返回 7，WEEKDAY 函数返回值大于 5 的就是周六和周日。

## 2. 计算今天是本年的第几周

想起微博上流传的一个段子，某男士下班后外出应酬，对家人谎称在公司加班。其妻子电话查岗，回答正在电脑前埋头苦干。妻子问了个问题，算一下今天是当年的第几周，限 30 秒内回答。

### 1) 问题分析

这个问题如不在电脑前的确不好回答，查年历也要计算一下，还很容易出错。但在 Excel 中有一个函数 WEEKNUM 就能直接解决。

### 2) 解决方案

在任意单元格中录入如下公式，按回车键结束即可。

```
=WEEKNUM(TODAY(),2)
```

### 3) 公式解析

以 TODAY 函数返回的系统当前日期序列号作为 WEEKNUM 函数的第 1 参数，第 2 参数使用 2，表示 1 周是从周一开始。以日期 2012 年 7 月 27 日为例，使用这个公式返回的星期序号是 31。

使用 WEEKNUM 函数时需要注意第 2 参数的取值，不同的选择返回结果可能会不一样，请看图 6.22。

	A	B	C
1	日期	周序号	公式
2	2012-1-2	1	=WEEKNUM(A2,1)
3	2012-1-2	2	=WEEKNUM(A3,2)
4	2012-7-27	30	=WEEKNUM(A4,1)
5	2012-7-27	31	=WEEKNUM(A5,2)

图 6.22 WEEKNUM 函数第 2 参数不同的取值比较

这里 WEEKNUM 函数第 2 参数使用 1 和 2 返回的结果对于有些日期是不一样的，原因在于 2012 年 1 月 1 日这一天是周日。如 WEEKNUM 函数的第 2 参数使用 1，则 1 月 1 日是第 1 周的第 1 天，如第 2 参数使用 2，则 1 月 1 日是第 1 周的最后 1 天，1 月 2 日就是第 2 周了。之后 2012 年所有的日期在使用 WEEKNUM 函数计算周数时用不同的参数 1 或者 2 都会相差 1 周。

本节示例文件：《6.2 使用函数提取日期时间信息.xlsm》

## 6.3 使用函数构建日期与时间

除了通过快捷键及 NOW、TODAY 函数录入系统当前日期时间外，还可以使用函数来构建日期时间。使用函数构建日期时间时参数可以是单元格引用或者是其他公式运算的结果，应用会更加灵活。

6.3.1 使用 DATE 函数构建日期

DATE 函数返回表示特定日期的连续序列号，在通过公式或单元格引用提供年月日时，DATE 函数是最为有用的，推荐理解掌握并灵活应用。

DATE 函数有 3 个参数，都是必须的，公式理解如表 6.4 所示。

表 6.4 DATE函数 3 个参数的理解

参数顺序	参数含义	注意事项
第1参数	Year必需。Year参数的值可以包含一到四位数字	Excel将根据计算机所使用的日期系统来解释Year参数
第2参数	Month必需。一个正数或负数，表示一年中从1-12月的各个月	
第3参数	Day必需。一个正数或负数，表示一月中从1-31日的各天	

比较难理解的是月和日参数还可以是负数，这里通过举例来加强一下理解，请看图 6.23。

	A	B	C	D	E	F
1	年	月	日	返回值	公式	返回值*1
2	2012	2	29	2012-2-29	=DATE(A2,B2,C2)	40968
3	2012	2	30	2012-3-1	=DATE(A3,B3,C3)	40969
4	2012	3	0	2012-2-29	=DATE(A4,B4,C4)	40968
5	2012	3	-1	2012-2-28	=DATE(A5,B5,C5)	40967
6	2012	3	-1.5	2012-2-27	=DATE(A6,B6,C6)	40966
7	2012	3	-1.2	2012-2-27	=DATE(A7,B7,C7)	40966
8	2012	3.3	1	2012-3-1	=DATE(A8,B8,C8)	40969
9	9999	1	1	9999-1-1	=DATE(A9,B9,C9)	2958101
10	10000	1	1	#NUM!	=DATE(A10,B10,C10)	#NUM!

图 6.23 DATE 函数 3 个参数取值举例

DATE 函数属于比较智能的函数，有自动容错机制，允许用户犯错。如参数超出正常的取值范围，函数并不会简单地报错，而是通过类似于进位（或退位）的方式，通过调整上一个参数的取值来容错。

比如 2012 年的 2 月份只有 29 天，如直接在单元格中录入“2012-2-30”，Excel 会将其作为文本处理，不能参与计算，再使用 DATEVALUE 函数转换也不行。而录入公式“=DATE(2012,2,30)”，Excel 会自动在第 3 参数上进位，因 2 月份仅有 29 天，则满 29 进 1，第 2 参数在原取值 2 的基础上加 1，第 3 参数在原取值 30 的基础上减 29，得到结果为 2012 年 3 月 1 日。

自动容错也要在允许的范围之内，如超出了 Excel 允许的日期范围，比如大于 9999 年 12 月 31 日的日期，DATE 函数还是会报错的。

在本章第 2 节中举过一个计算上月天数的例子，当时使用的是 DAY 函数和 TODAY 函数的嵌套，公式略有些难理解。

```
=DAY(TODAY()-DAY(TODAY()))
```

在学习了 DATE 函数后，可以通过其他思路来解决这个问题。

1. 问题分析

可以通过 DATE 函数构建上月最后一天的日期，然后外套 DAY 函数求得这一天的天数序号，即当月的天数。

2. 解决方案

使用如下公式，按回车键结束。



```
=DAY(DATE(YEAR(NOW()),MONTH(NOW()),0))
```

### 3. 公式解析

这个公式核心部分是内层的 DATE 函数，前两个参数都是通过 NOW 函数获得当前系统时间的年月信息，第 3 参数使用 0。因天数序列号没有 0，Excel 会从第 2 参数退一位给第 3 参数，这样第 2 参数返回的是系统日期的上一个月的序号，第 3 参数返回的是系统日期上一个月的天数。

	A	B
1	上月天数	公式
2		30 =DAY(TODAY()-DAY(TODAY()))
3		30 =DAY(DATE(YEAR(NOW()),MONTH(NOW()),0))

图 6.24 使用 DATE 函数构建上月最后一天的日期

再举一个判断闰年与平年的例子，如图 6.25，A 列存放需要判断的年份，要求在 B 列一个公式完成判断。

	A	B	C
1	年份	DAY+DATE	公式
2	2000	闰年	=IF(DAY(DATE(A2,2,29))=29,"闰年","平年")
3	2004	闰年	=IF(DAY(DATE(A3,2,29))=29,"闰年","平年")
4	2007	平年	=IF(DAY(DATE(A4,2,29))=29,"闰年","平年")
5	2008	闰年	=IF(DAY(DATE(A5,2,29))=29,"闰年","平年")
6	2012	闰年	=IF(DAY(DATE(A6,2,29))=29,"闰年","平年")
7	2100	平年	=IF(DAY(DATE(A7,2,29))=29,"闰年","平年")

图 6.25 判断闰年与平年

### 4. 问题分析

判断闰年与平年有多种思路，其中最简单的方案是判断这一年 2 月份是否有 29 日。因 DATE 函数参数具有容错性，如超过取值范围会自动进位，可以先用 DATE 函数构建\*年 2 月 29 日的日期，然后用 DAY 函数判断这一天的天数序号是否是 29。

### 5. 解决方案

在单元格中录入如下公式，按回车键结束。

```
=IF(DAY(DATE(YEAR(NOW()),2,29))=29,"闰年","平年")
```

### 6. 公式解析

公式中“YEAR(NOW())”取得系统当前时间的年份，用 DATE 函数强制构建这一年的 2 月 29 日。如不存在这一天，DATE 函数的第 3 参数会自动进位上去，那样返回的日期就的天数序号就不是 29 了。

然后外套 DAY 函数取返回日期的天数序号，判断是否等于 29，就能得到闰年与平年信息。

职场点睛：通过这两个例子可以发现 DATE 函数的确非常管用，在实际工作中涉及到日期计算问题时，使用 DATE 函数构建日期是一种比较通用的解决方案。

## 6.3.2 使用 TIME 函数构建时间

TIME 函数用法与 DATE 函数类似，也是 3 个参数，也有容错功能。

	A	B	C	D	E	F
1	小时	分	秒	返回值	公式	返回值*1
2	21	10	30	21:10:30	=TIME(A2,B2,C2)	0.882291667
3	21	10	70	21:11:10	=TIME(A3,B3,C3)	0.88275463
4	21	10	0	21:10:00	=TIME(A4,B4,C4)	0.881944444
5	21	10	-10	21:09:50	=TIME(A5,B5,C5)	0.881828704

图 6.26 TIME 函数 3 个参数取值举例

相对于 DATE 函数，TIME 函数的应用会简单很多，原因在于 1 天就是 24 个小时，1 个小时 60 分钟，1 分钟 60 秒都是永恒不变的。不同于 DATE 函数，一年中各个月份的天数是不同的，从 28 天到 31 天都有，还有平年与闰年之分。

### 6.3.3 使用函数返回特定要求的日期

Excel 中还有一些函数能够返回特定的日期，比如间隔若干个月的日期，比如间隔若干个工作日后的日期等。

#### 1. 取“隔月对日”的函数

DATE 函数的容错功能比较智能，但其终究不能完全理解用户的需求，有时候容错仅是自作聪明，返回的结果并非用户的要求，请看以下案例。

某境外上市公司的境内子公司引入了一套总账本地化系统，在导入该系统时需要对固定资产的“开始折旧日期”根据原系统中的“开始使用日期”调后一个月。因是间隔一个月，很容易想到使用 DATE 函数构建新的日期，第 2 参数月份使用原来的月份加 1，公式如下，完成效果如图 6.27 所示。

=DATE(YEAR(C2),MONTH(C2)+1,DAY(C2))

D2		=DATE(YEAR(C2),MONTH(C2)+1,DAY(C2))				
	A	B	C	D	E	F
1	固定资产编码	固定资产描述	开始使用日期	开始折旧日期		
2	020103489	新概念店家具	2012-1-30	2012-3-1		
3	020103490	投影机	2012-1-31	2012-3-2		
4	021300009	虹膜	2012-2-5	2012-3-5		
5	022300012	新概念店家具	2012-2-29	2012-3-29		
6	H0201003203	笔记本电脑	2012-3-8	2012-4-8		
7	H020103098	手持条码设备	2012-3-31	2012-5-1		
8	H020103108	IBM硬件	2012-4-15	2012-5-15		

图 6.27 使用 DATE 函数求隔月对日的日期

#### 1) 问题分析

公式粗看没有问题，但图中 D 列使用黄颜色标注的单元格中的返回结果都是错误的。原因在于一年中各个月份的天数是不一样的，比如 2012 年的 2 月份只有 29 天，对于 DATE 函数构建的日期“2012-2-30”，Excel 会自动将第 3 参数满 29 升级，“智能”地变成了“2012-3-1”。

显然这种智能并不是这个问题中所需要的，DATE 函数无法直接处理此类问题，但 Excel 中有以字母 E 开头的日期函数，就是为此量身定做的。

#### 2) 解决方案

在 E3 单元格中录入如下公式，向下复制完成，效果如图 6.28 所示。

=EDATE(C3,1)

E3 $\text{fx}$ =EDATE(C3,1)					
	A	B	C	D	E
1				开始折旧日期	
2	固定资产编码	固定资产描述	开始使用日期	DATE	EDATE
3	020103489	新概念店家具	2012-1-30	2012-3-1	2012-2-29
4	020103490	投影机	2012-1-31	2012-3-2	2012-2-29
5	021300009	虹膜	2012-2-5	2012-3-5	2012-3-5
6	022300012	新概念店家具	2012-2-29	2012-3-29	2012-3-29
7	H0201003203	笔记本电脑	2012-3-8	2012-4-8	2012-4-8
8	H020103098	手持条码设备	2012-3-31	2012-5-1	2012-4-30
9	H020103108	IBM硬件	2012-4-15	2012-5-15	2012-5-15

图 6.28 使用 EDATE 函数计算隔月对日

### 3) 公式解析

使用了 EDATE 函数，这个函数有两个参数，以第一个参数中的日期为基础，返回间隔若干个月（第 2 个参数）后的日期。这个函数习惯称之为取“隔月对日”。

EDATE 函数的智能在于如果第一个参数是某个月的最后一天，则返回的是间隔若干月后的某个月的最后一天，这样就避免了使用 DATE 函数构建日期时自动进位的问题。

### 2. 取“隔月末日”的函数

除了 EDATE 函数之外，Excel 中还有一个以字母 E 开头的函数 EOMONTH，这个函数能够取到间隔若干月的月份的最后一天，习惯称之为取“隔月末日”。

还是看一下之前提到的计算上个月天数的例子，可以考虑 DAY+EOMONTH 函数组合，公式如下：

=DAY(EOMONTH(NOW(),-1))

#### 1) 公式解析

这个公式的思路仍是取得上个月最后一天的日期序列号，差异在于使用了 EOMONTH 函数。

EOMONTH 函数参数理解同 EDATE 函数，不同之处在于返回的是间隔若干个月后某个月份最后一天的序列号，具体是哪一个月份要看函数的第 2 参数。

这里 EOMONTH 函数的第 2 参数使用 -1，相当于返回第 1 参数日期所在月份的前一个月，如是 0 则是当月，如是 1 则是后一个月，其他依此类推。

目前已有三种思路计算上个月的天数，如图 6.29 所示，如不使用 DAY 函数外套的思路，还会有更多解决方案。

	A	B
1	上月天数	公式
2	30	=DAY(TODAY()-DAY(TODAY()))
3	30	=DAY(DATE(YEAR(NOW()),MONTH(NOW()),0))
4	30	=DAY(EOMONTH(NOW(),-1))

图 6.29 计算上个月天数的 3 种解决方案

EOMONTH 函数能取到某个月最后一天的日期，这一点比用 DATE 函数直接构建月末日期更加直接。回过头来再看判断闰年的例子，使用 DAY+EOMONTH 函数组合也是一种思路，公式如下：

=IF(DAY(EOMONTH(A2&"-2-1",0))=29,"闰年","平年")

D2		fx		=IF(DAY(EOMONTH(A2&"-2-1",0))=29,"闰年","平年")
A	B	D	E	
1	年份	DAY+DATE	DAY+EOMONTH	公式
2	2000	闰年	闰年	=IF(DAY(EOMONTH(A2&"-2-1",0))=29,"闰年","平年")
3	2004	闰年	闰年	=IF(DAY(EOMONTH(A3&"-2-1",0))=29,"闰年","平年")
4	2007	平年	平年	=IF(DAY(EOMONTH(A4&"-2-1",0))=29,"闰年","平年")
5	2008	闰年	闰年	=IF(DAY(EOMONTH(A5&"-2-1",0))=29,"闰年","平年")
6	2012	闰年	闰年	=IF(DAY(EOMONTH(A6&"-2-1",0))=29,"闰年","平年")
7	2100	平年	平年	=IF(DAY(EOMONTH(A7&"-2-1",0))=29,"闰年","平年")

图 6.30 使用 DAY+EOMONTH 函数判断闰年

## 2) 公式解析

EOMONTH 函数的第 1 参数使用需要判断的年份来连接字符串“-2-1”，返回需要判断年份的 2 月 1 日，第 2 参数使用 0，EOMONTH 函数返回需要判断年份的 2 月份的最后一天。之后外套 DAY 函数及 IF 函数的判断同前述公式。

## 3. 取间隔若干工作日的函数

在日常工作中经常会遇到和工作日（不包括周末和专门指定的假日）有关的问题，比如办理护照，受理之后会收到回执，需要 10 个工作日完成。返回间隔若干个工作日后的日期，在 Excel 中有一个函数 WORKDAY 能够实现。

看一个计算工作日的例子，某员工应在 2012 年 4 月 30 日离职，但因未有合适的交接人员，企业挽留其再工作 5 天，计算其最后一个工作日的日期（2012 年 5 月 1 日是国定假日，2-3 日属于换休）。

	A	B	C
1	今天日期	再工作5天后的离职日期	公式
2	2012-4-30	2012-5-7	=WORKDAY(A2,5)
3	2012-4-30	2012-5-10	=WORKDAY(A2,5,{"2012-5-1","2012-5-2","2012-5-3"})
4	2012-4-30	2012-5-10	=WORKDAY(A3,5,{"2012-5-1","2012-5-2","2012-5-3","2012-5-5"})

图 6.31 使用 WORKDAY 函数计算离职日期

从这个例子可以看出，WORKDAY 函数对于工作日仅能通过双休日来判断，如有双休日之外的节假日，需要在 WORKDAY 函数的第 3 参数中列出，否则 Excel 不知道这些天为假日，即使是国定假日也无法判断。

如果假日本身就是双休日，则是无需列在 WORKDAY 函数的第 3 参数中的，所以第 2 个公式和第 3 个公式不同，但返回结果一样。

本节示例文件：《6.3 使用函数构建日期和时间.xlsx》

## 6.4 使用日期函数执行计算

日期与时间本身就是一个数字，数字就可以执行计算，比如两个日期相减，日期时间和数值的四则运算等。Excel 中还有一些特殊的日期函数，可以根据需要计算的日期返回一些特定的值，比如间隔的年、月、日、工作日等。

6.4.1 微软也会藏一手——认识隐藏函数 DATEDIF

从事人力资源工作的 Excel 用户可能接触过一个奇特的函数——DATEDIF。说这个函数奇特，原因在于其不存在于函数列表框中，帮助文件中也找不到。

有点奇怪，看来微软也留了一手，这个函数的存在最初是为了兼容 Lotus 1-2-3 软件。虽然找不到帮助文件，但还是可以通过网络搜索找到的相关资料，DATEDIF 函数有 3 个参数，请看表 6.5。

表 6.5 DATEDIF 函数 3 个参数理解

参数顺序	参数含义	注意事项
第1参数	代表时间段内第一个日期或起始日期	可以是带引号的文本串、系列号或其他公式或函数的结果等。 (还可以是TEXT函数格式化输出的结果)
第2参数	代表时间段内最后一个日期或结束日期	
第3参数	所需信息的返回时间单位代码	

前 2 个参数不难理解，就是 2 个日期，只是第 2 个参数的日期序列号要大于第 1 个参数的日期序列号，如输入以下公式，Excel 返回结果就是错误值#NUM!。

=DATEDIF("2012-6-30","2012-6-1","y")

这个函数难在第 3 个参数上，第 3 参数有 6 种不同的写法，结合实例看一下测试结果。

	A	B	C	D	E	F
1	起始日期	结束日期	差	公式	参数	说明
2	2010-8-1	2012-2-4	1	=DATEDIF(A2,B2,"Y")	Y	相差年数
3	2010-8-1	2012-2-4	18	=DATEDIF(A3,B3,"M")	M	相差总月数
4	2010-8-1	2012-2-4	6	=DATEDIF(A4,B4,"YM")	YM	一年内相差月数
5	2010-8-1	2012-2-4	552	=DATEDIF(A5,B5,"D")	D	相差总天数
6	2010-8-1	2012-2-4	187	=DATEDIF(A6,B6,"YD")	YD	一年内相差天数
7	2010-8-1	2012-2-4	3	=DATEDIF(A7,B7,"MD")	MD	一月内相差天数

图 6.32 DATEDIF 函数第 3 参数举例

从图 6.32 的举例可以看出，对于年份的差异，只有 1 种参数的用法，返回间隔的整年数；而对于月份的差异，有 2 种参数写法，分别表示“相差的总月数”和“一年内相差的月数”；而到了涉及天数的差异，写法变成了 3 种，分别为“相差总天数”、“一年内相差天数”和“一月内相差天数”。

差数写法选择多，变化也多；功能强大，相对也容易出错。比如需要根据身份证号码计算员工的实际年龄，对于 DATEDIF 函数第 3 参数的选择就不能大意，如图 6.33 所示。

	A	B	C	D	E	F
1	员工	身份证号	年龄			
2			年	月	天	年纪
3	张三	511025198501019611	27	6	30	27年6月30天
4	李四	432503198812304352	23	7	1	23年7月1天
5	王五	511025770316628	35	4	15	35年4月15天
6	赵六	130301198308090514	28	11	22	28年11月22天
7	钱七	130502870529316	25	2	2	25年2月2天

图 6.33 根据身份证号码计算实际年龄

1) 问题分析

身份证号码中有出生年月日信息，提取时需要注意身份证号码可能存在的位数差异；如果需要公式



计算结果能随时根据系统时间更新，就需要使用 TODAY 函数产生当前的系统日期；而计算年月日，使用 DATEDIF 函数最简单。

#### 2) 解决方案

在 C3、D3、E3 单元格分别录入如下公式，向下填充完成。

```
=DATEDIF(TEXT(MID(B3,7,6+(LEN(B3)-18)*2),"0-00-00"),TODAY(),"Y")  
=DATEDIF(TEXT(MID(B3,7,6+(LEN(B3)-18)*2),"0-00-00"),TODAY(),"YM")  
=DATEDIF(TEXT(MID(B3,7,6+(LEN(B3)-18)*2),"0-00-00"),TODAY(),"MD")
```

最后在 F3 单元格使用连接符&，将前 3 列计算结果连接，并加上文字年月日，公式如下：

```
=C3&"年"&D3&"月"&E3&"天"
```

#### 3) 公式解析

前 3 列中公式都是 DATEDIF 函数，且第 1 参数和第 2 参数完全一样，第 1 参数提取身份证号码中的出生年月日信息，第 2 参数用 DAY 函数返回系统当前日期序列号。

其中公式“TEXT(MID(B3,7,6+(LEN(B3)-18)\*2),"0-00-00)”之前的案例中已经介绍过，表示截取身份证号码中的年月日信息段，并用 TEXT 函数转换成日期格式的文本。

DATEDIF 函数的第 3 参数用“YM”，表示取一年内的相隔月份，“MD”表示一个月内的相隔天数，三列计算结果连接得到的就是员工的即时年龄。

这个例子中已经计算了年，相关月份计算就不需要再考虑年，已经计算了年月，涉及到天的计算就不需要考虑年月，但并非所有情况下都是如此，下面举一个需要计算累计月份的例子。

某企业在发放员工薪资时内部规定每个月扣除 50 元作为押金，待员工离职时手续完备才全部返还。如图 6.34 所示，B 列和 C 列分别存放员工入职时间和离职时间，需要在 D 列计算员工累计工作月份。

	A	B	C	D	E
1	姓名	入职时间	离职时间	累计工作月份	押金返还
2	李明	2005年7月12日	2012年6月30日	83	4,150
3	张诚	2006年9月13日	2012年6月30日	69	3,450
4	何佳	2006年4月15日	2012年6月30日	74	3,700
5	李明诚	2010年12月13日	2012年6月30日	18	900
6	李涛	2010年8月30日	2012年6月30日	22	1,100
7	张晓群	2011年8月22日	2012年6月30日	10	500
8	邓捷	2011年6月9日	2012年6月30日	12	600

图 6.34 计算员工累计工作月份

#### 4) 问题分析

这里计算月份时没有先计算年，从题意也能看出需要计算的是一个累计的月份而非一年中间隔的月份，DATEDIF 函数第 3 参数使用“M”。

#### 5) 解决方案

在 D2 和 E2 单元格中分别录入如下公式，向下复制完成。

```
=DATEDIF(B2,C2,"M")  
=D2*50
```

#### 6) 公式解析

就是一个简单的 DATEDIF 函数，第 3 参数使用“M”计算两个时点之间间隔的累计月份。从这个例子更能看出 DATEDIF 函数在计算日期间隔上的优势，其他函数不可能一个公式完成相关的计算。



6.4.2 与工作日相关的计算

在第 3 节中介绍过一个返回间隔若干个工作日后日期的函数，与之相关，Excel 中还有一个函数可以计算两个时点之间的净工作日的函数 NETWORKDAYS。

Excel 中的函数其实并不难学，很多函数只要有点英文基础的用户都能猜出函数的用途，比如这个 NETWORKDAYS 函数就是。这个函数有 3 个参数，请看表 6.6。

表 6.6 NETWORKDAYS函数 3 个参数的理解

参数顺序	参数含义	注意事项
第1参数	一个代表开始日期的日期	和之前介绍的日期函数一样，可以是带引号的文本串、系列号或其他公式或函数的结果等（还可以是TEXT函数格式化输出的结果）
第2参数	一个代表终止日期的日期	
第3参数	可选，不在工作日历中的一个或多个日期所构成的可选区域	该列表可以是包含日期的单元格区域，或是表示日期的序列号的数组常量

这里所指的工作日是不包含双休日的，如有其它假日，需要体现在函数的第 3 参数中，否则 Excel 是无法自动判断的，这点和 WORKDAY 函数完全一样。

某企业在每年的第 3 季度（7~9 月）发放高温津贴，标准为每个工作日 10 元。该企业除执行双休日制度之外，每个月的最后一天企业内部放假。需要计算 2012 年第 3 季度应发的高温津贴，如图 6.35 所示。

	A	B	C	D	E	F
1	员工	三季度工作日	缺勤	高温津贴		内部假日
2	张三	63	6	570		2012-7-31
3	李四	63	0	630		2012-8-31
4	王五	63	5	580		2012-9-30
5	赵六	63	4	590		
6	钱七	63	2	610		

图 6.35 计算第三季度工作日

1) 问题分析

计算净工作日最简单的方法是使用 NETWORKDAYS 函数，这里企业内部有休假的特别规定，需要体现在函数的第 3 参数中。

2) 解决方案

在 B 单元格中录入如下公式，向下复制完成。

=NETWORKDAYS("2012-7-1","2012-9-30",{ "2012-7-31","2012-8-31","2012-9-30"})

3) 公式解析

NETWORKDAYS 函数的第 1 参数是需要计算期间的第 1 天，第 2 参数是最后 1 天，第 3 参数使用常量数组列举了内部的节假日。

如果把内部节假日放到单元格区域 F2: F4 中，上述公式可修改成如下形式：

=NETWORKDAYS("2012-7-1","2012-9-30",F2:F4)

如果省略第 3 参数，计算结果是 65，和正确的答案相差 2 天，而企业内部节假日是 3 天。原因在于 2012 年 9 月 30 日本身就是双休日，列在第 3 参数中不影响函数的计算结果。

如果不是用 NETWORKDAYS 函数计算工作日，其他方案公式会比较难理解，比如如下的数组公式只能计算工作日，无法直接减掉月末的内部节假日。

=SUM(N(WEEKDAY(ROW(INDIRECT(DATE(2012,7,1)&" "&DATE(2012,9,30))),2)<6))

如要减去内部假日，还要再加上一段公式判断，明显不如使用 NETWORKDAYS 函数简单。

=SUM(N(WEEKDAY(ROW(INDIRECT(DATE(2012,7,1)&"."&DATE(2012,9,30))),2)<6))-SUM(N(WEEKDAY(F2:F4,2)<6))

职场点睛：日期时间函数的特点是数量比较多，不同的函数有不同的用途，灵活应用必将大幅度提高工作效率。建议对于从事财务、人事等工作的用户，先对日期时间函数的功能有一个大致的了解，混个脸熟，遇到相关应用时再仔细研究函数的帮助文件。

本节示例文件：《6.4 使用日期函数执行计算.xlsx》

## 6.5 日期时间函数之实战应用

函数学习的特点是单个函数不难理解，但涉及到函数嵌套就不那么容易了。这点在日期时间函数中非常明显，实际工作中很多应用涉及到其他类型的函数，有时候还要和 Excel 的内置功能结合起来。

### 6.5.1 合同到期续签提醒

某企业行政部门负责管理公司除员工劳动合同之外的所有合同，在合同到期前一个月提醒相关部门进入续签流程，合同续签管理表如图 6.36 所示。

	A	B	C	D	E	F	G
1	当前日期	2012-7-5		即将到期		已过期	
2							
3	合同序列号	合同种类	签约单位	对方单位	开始日期	终止日期	是否续签
4	A2011001	租赁	三联	四季酒店	2011-7-1	2012-6-30	
5	A2011002	租赁	三联	万达广场	2011-7-10	2012-7-9	
6	A2012001	租赁	三联	徐汇房管局	2011-9-1	2012-8-31	
7	B2010001	劳务	富丽	上海外服	2010-8-1	2012-7-31	
8	C2011001	装修	自然美	荣欣装潢	2011-2-1	2012-7-31	
9	D2011001	咨询	富丽	泰悦咨询	2011-1-1	2012-12-31	
10	D2011002	咨询	富丽	新高信事务所	2011-8-1	2012-7-31	

图 6.36 合同续签管理表

图中 B2 单元格在实际工作中应使用函数“=TODAY()”，这样在每次打开工作簿时会自动计算，根据系统日期返回结果。

#### 1) 问题分析

合同续签提醒实质上就是比较当前日期与合同终止日期，如两者相差小于 30 天就应进入续签流程（除非企业管理层有特定的要求，否则用 30 天做比较即可），如当前日期大于合同终止日期，合同属于已过期。

#### 2) 解决方案 1

为了让判断结果清晰可见，可以使用条件格式，分别对“即将到期”的合同用黄色标注，“已过期”的合同用红色标注，操作步骤如下：

（1）选中需要设置条件格式的 A4: G10 单元格区域，单击“开始”|“样式”|“条件格式”|“新建规则”命令，弹出“新建格式规则”对话框，在“选择规则类型”框中选取“使用公式确定需要设置格式的单元格”；

(2) 在“为符合此公式的值设置格式”框中输入如下公式，并设置格式为黄色填充；

=AND(\$G4="", \$F4>\$B\$1, \$F4-\$B\$1<30)

(3) 继续为已过期合同设置条件格式，操作同上述步骤，公式如下，设置格式为红色填充，操作完成效果如图 6.37 所示。

=AND(G4="", \$B\$1>\$F4)

	A	B	C	D	E	F	G
1	当前日期	2012-7-5		即将到期		已过期	
2							
3	合同序列号	合同种类	签约单位	对方单位	开始日期	终止日期	是否续签
4	A2011001	租赁	三联	四季酒店	2011-7-1	2012-6-30	
5	A2011002	租赁	三联	万达广场	2011-7-10	2012-7-9	
6	A2012001	租赁	三联	徐汇房管局	2011-9-1	2012-8-31	
7	B2010001	劳务	富丽	上海外服	2010-8-1	2012-7-31	
8	C2011001	装修	自然美	荣欣装潢	2011-2-1	2012-7-31	
9	D2011001	咨询	富丽	泰悦咨询	2011-1-1	2012-12-31	
10	D2011002	咨询	富丽	新高信事务所	2011-8-1	2012-7-31	

图 6.37 对“即将到期”和“已到期”合同设置颜色

### 3) 公式解析

条件格式中的公式都是 AND 函数，只有在 AND 函数中所有的条件都满足的情况下，AND 函数才返回 TRUE，对于设置的格式起作用。

第 1 个公式 AND 函数有 3 个条件，“\$G4=""”表示尚未有续签结果，“\$F4>\$B\$1”代表合同尚未到期，“\$F4-\$B\$1<30”代表当前时间距离合同到期日已不到 30 天，在 3 个条件全部满足，即已距离到期日不到 30 天且未有续签安排的情况下，条件格式才起作用，相关合同显示为黄色。

第 2 个公式的 AND 函数有 2 个参数，理解同上，表示在合同已到期且未有续签安排时，相关合同显示为红色。

如果在已得到续签结果的情况下，在 G 列选择“已续约”或者“不续约”，条件格式设置的颜色会自动变化。比如对于第 8 行的装修合同，到期后无需再续签，在 G8 单元格中选择“不续约”，条件格式的返回结果会自动更新如图 6.38 所示。

	A	B	C	D	E	F	G
1	当前日期	2012-7-5		即将到期	3	已过期	1
2							
3	合同序列号	合同种类	签约单位	对方单位	开始日期	终止日期	是否续签
4	A2011001	租赁	三联	四季酒店	2011-7-1	2012-6-30	
5	A2011002	租赁	三联	万达广场	2011-7-10	2012-7-9	
6	A2012001	租赁	三联	徐汇房管局	2011-9-1	2012-8-31	
7	B2010001	劳务	富丽	上海外服	2010-8-1	2012-7-31	
8	C2011001	装修	自然美	荣欣装潢	2011-2-1	2012-7-31	不续约
9	D2011001	咨询	富丽	泰悦咨询	2011-1-1	2012-12-31	
10	D2011002	咨询	富丽	新高信事务所	2011-8-1	2012-7-31	

图 6.38 变更续约状况后的结果

### 4) 解决方案 2

除了设置条件格式用醒目的颜色标注“即将到期”和“已过期”合同之外，还可以使用函数公式计

算具体的天数。分别在 E1 和 G1 单元格中录入如下公式，并按“Ctrl+Shift+Enter”三键结束，返回结果如图 6.38 中第 1 行。

```
=SUM((F4:F10>B1)*(F4:F10-B1<30)*(G4:G10=""))  
=SUM((B1>F4:F10)*(G4:G10=""))
```

5) 公式解析

这两个公式外层都是 SUM 函数，虽然还没有讲解过，但用过电脑的用户应该都知道这是一个求和的函数；内层分别是 3 段和 2 段逻辑判断的结果相乘，具体的写法类似于之前条件格式中的设置。

公式属于典型的 SUM 多条件计数公式，具体将在下一章中重点讲解，这里先保留悬念。有兴趣的用户可以通过 F9 看一下运算结果，猜测一下公式的运算原理。

职场点睛：合同管理属于企业日常管理中的常见应用，学会这一招，在工作簿打开时就会自动提示已过期和需要续签的合同，大幅度提升了工作效率。

6.5.2 计算三八妇女节应发福利费

某企业为女性员工发放三八妇女节福利费，发放资格为 2012 年 3 月 1 日在册且年满 20 周岁的女性员工，发放标准为每人 200 元。人事部门根据已有的员工登记表计算应发福利费金额，相关表格如图 6.39 所示。

	A	B	C	D	E
1	工号	身份证号	入职时间	福利费	备注
2	A00001	341122199205161621	2012-1-8	0	不满20周岁
3	A00002	342322197902240023	2012-3-4	0	3月1日后入职
4	A00003	342322810101002	2006-1-24	200	
5	A00004	342322198303080026	2012-1-9	200	
6	A00005	342322198407240020	2007-2-14	200	
7	A00006	34232219851012006X	2012-1-9	200	
8	A00007	342322197306140018	2012-1-8	0	男性

图 6.39 计算三八节福利费

1) 问题分析

根据福利费的发放标准，需要满足 3 个条件，女性、年满 20 周岁，2012 年 3 月 1 日前在册。其中性别、年龄可以通过身份证号码中的信息提取，3 月 1 日前是否在册可以通过入职时间判断。

2) 解决方案

在 D2 单元格中录入如下公式，向下复制完成。

```
=(MOD(RIGHT(LEFT(B2,17)),2)=0)*(DATEDIF(TEXT(MID(B2,7,6+(LEN(B2)=18)*2),"0-00-00"),"2012-3-1","y")>=20)*(C2<DATE(2012,3,1))*200
```

3) 公式解析

公式有点长，但并不难理解。对于公式分段用 F9 求值，可以发现实际上是 3 段运算结果与 200 的乘积，以 D2 公式为例，F9 分解后的结果是“=TRUE\*FALSE\*TRUE\*200”。

这个公式也可以写成 AND 函数的形式，代表 3 个条件同时满足时取值 200，否则取值 0，公式如下：

```
=AND(MOD(RIGHT(LEFT(B2,17)),2)=0,DATEDIF(TEXT(MID(B2,7,6+(LEN(B2)=18)*2),"0-00-00"),"2012-3-1","y")>=20,C2<DATE(2012,3,1))*200
```

第 1 段条件“MOD(RIGHT(LEFT(B2,17)),2)=0”判断是否是女性，公式兼顾了 15 位和 18 位的身份证，其中“RIGHT(LEFT(B2,17))”取到含有性别信息的字符，“MOD(字符,2)=0”判断是否是偶数，

如为偶数是女性，奇数是男性；

第 2 段条件 “DATEDIF(TEXT(MID(B2,7,6+(LEN(B2)=18)\*2),"0-00-00"),"2012-3-1","y")>=20” 判断年龄是否已满 20 周岁，其中 “TEXT(MID(B2,7,6+(LEN(B2)=18)\*2),"0-00-00)” 的思路取得身份证号码中的出生年月日，使用 DATEDIF 函数计算截止 2012 年 3 月 1 日的年龄（周岁）；

第 3 段条件 “C2<DATE(2012,3,1)” 判断入职时间是否早于 2012 年 3 月 1 日；

在 3 段条件全部满足的情况下，AND 函数返回 TRUE，然后与 200 相乘，就能得到发放福利费的金  
额。

图 6.39 的 E 列中列出福利费为 0 的原因，读者可以使用 F9 测试一下各段条件返回的结果，加深对于公式的理解。

职场点睛：如果不会使用函数公式，一个个判断既麻烦还容易出错。建议在日常工作中养成使用函数公式的习惯。

### 6.5.3 计算迟到、早退和加班时间

目前很多企业已经使用了打卡考勤系统，员工上下班打卡的信息通过打卡机记录后自动生成一个数据库，不需要人事部门手工汇总打卡信息。名为考勤系统，但实际上功能参差不齐，有些系统能根据打卡信息自动计算出员工迟到、早退和加班时间，有的仅是提供一些数据而已。

某企业上班时间为 8:00——17:00，晚上加班从 17:30 算起。考勤系统中导出的资料如图 6.40 所示，需要 F-H 列分别计算员工迟到、早退和加班时间。

	A	B	C	D	E	F	G	H
1	部门	姓名编号	日期	时间	比对方式	迟到	早退	加班
346	总公司	8	2012-6-7	22:01:27	指纹			4:31:27
347	总公司	38	2012-6-7	22:01:32	指纹			4:31:32
348	总公司	35	2012-6-7	22:01:36	指纹			4:31:36
349	总公司	24	2012-6-7	22:01:54	指纹			4:31:54
350	总公司	23	2012-6-7	22:02:05	指纹			4:32:05
351	总公司	17	2012-6-7	22:02:19	指纹			4:32:19

图 6.40 计算员工迟到、早退和加班时间

#### 1) 问题分析

使用函数计算员工考勤情况必须基于一定的假设，这里假设每个员工一天中都会在上班和下班时打卡（每次可能打卡不止一次），如有加班也不会超过当日零点。这样可以根据一天中最早的一次打卡时间来判断是否迟到，以最后一次打卡时间来判断是否是早退及有否加班。

可能存在上下班时都不止一次打卡，公式需要在第一次打卡时间所在的行列出迟到时间，在最后一次打卡时间所在的行列出早退和加班时间。

#### 2) 解决方案

在 F2: H2 单元格中分别录入如下公式，按 “Ctrl+Shift+Enter” 三键结束，向下复制完成。

```
=IF(D2=MIN(IF((C2=C$2:C$351)*(B2=B$2:B$351),D$2:D$351)),TEXT(D2-8/24,"h:mm:ss;"),"  
=IF(D2=MAX(IF((C2=C$2:C$351)*(B2=B$2:B$351),D$2:D$351)),TEXT(17/24-D2,"h:mm:ss;"),"  
=IF(D2=MAX(IF((C2=C$2:C$351)*(B2=B$2:B$351),D$2:D$351)),TEXT(D2-17.5/24,"h:mm:ss;"),")
```

#### 3) 公式解析

这 3 个公式思路类似，这里仅解释 F 列的公式，即第 1 个公式。

公式的外层是一个 IF 函数，第 1 参数使用了多条件判断取值公式，返回值和 D2 单元格中的时间信息比较。其中 “IF((C2=C\$2:C\$351)\*(B2=B\$2:B\$351),D\$2:D\$351)” 是根据 B 列的姓名编号和 C 列的日



期返回对应员工和日期的考勤时间记录，外套 MIN 函数取得最小值，即对应员工和日期的当天第一条考勤记录。

外层 IF 函数第一个参数计算结果为 TRUE 时，返回 IF 函数第 2 个参数，即 TEXT 函数运算结果。TEXT 函数的第 2 参数中使用了分号 “;” 分隔，分号后面为空，表示将第 1 参数计算结果为正数的按指定格式返回，如第 1 参数计算结果非正数，则返回为空值。

“8/24” 是将上午上班时间 8 点转换成数字，如 “D2-8/24” 运算结果大于 0，即存在迟到则返回迟到的时间信息，返回格式为 “h:mm:ss”，如 “D2-8/24” 运算结果不大于 0，使用 TEXT 函数将返回结果强制转换成空值。

后面 2 个公式的差异在于用 MAX 函数取得各个员工当日最后一笔考勤时间，早退判断的标准是早于 17 点，即 “17/24”，加班判断标注是大于 17 点 30 分，即 “17.5/24”。

因函数的计算是基于一定的假设，如考勤打卡存在异常，可能计算得出的结果会很奇怪。比如正常的迟到或者早退一般都不会超过一个小时，对于一个小时以上的迟到或者早退，设置条件格式标注，如图 6.41 所示。




图 6.41 对 F-G 列设置条件格式

通过 I 列辅助公式筛选出 F-G 列非空的行，可以看到 275-276 行存在异常，如图 6.42 所示。

	A	B	C	D	E	F	G	H	I
1	部门	姓名编	日期	时间	比对方式	迟到	早退	加班	
32	总公司	43	2012-6-1	16:54:32	指纹		0:05:28		TRUE
159	总公司	36	2012-6-4	8:00:11	指纹	0:00:11			TRUE
160	总公司	21	2012-6-4	8:00:13	指纹	0:00:13			TRUE
161	总公司	37	2012-6-4	8:00:20	指纹	0:00:20			TRUE
162	总公司	43	2012-6-4	8:00:23	指纹	0:00:23			TRUE
163	总公司	13	2012-6-4	16:52:25	指纹		0:07:35		TRUE
164	总公司	17	2012-6-4	16:56:40	指纹		0:03:20		TRUE
206	总公司	30	2012-6-5	8:01:51	指纹	0:01:51			TRUE
207	总公司	29	2012-6-5	8:01:53	指纹	0:01:53			TRUE
208	总公司	37	2012-6-5	8:02:41	指纹	0:02:41			TRUE
209	总公司	36	2012-6-5	8:03:38	指纹	0:03:38			TRUE
273	总公司	37	2012-6-6	8:00:40	指纹	0:00:40			TRUE
274	总公司	43	2012-6-6	8:00:42	指纹	0:00:42			TRUE
275	总公司	6	2012-6-6	14:38:46	指纹		2:21:14		TRUE
276	总公司	3	2012-6-6	15:00:55	指纹	7:00:55	1:59:05		TRUE
318	总公司	30	2012-6-7	8:01:33	指纹	0:01:33			TRUE
319	总公司	37	2012-6-7	8:18:24	指纹	0:18:24			TRUE
320	总公司	36	2012-6-7	8:19:24	指纹	0:19:24			TRUE

图 6.42 存在异常的迟到、早退时间

其中 275 行是有请假，276 行姓名编号为 3 的员工在 2012 年 6 月 6 日仅打卡一次。对于此类情况，



需要人事部门手工甄别。

职场点睛：对于实际工作中遇到的复杂的计算问题，初学者不求一步到位，能想到使用函数公式就是一种进步。根据数据源的规律一步步优化公式，工作效率自然会逐步提高。

## 6.5.4 计算应收账款账期

核算应收账款账期是财务人员的常规工作之一。财务软件和 ERP 系统在这方面并不完善，更不用说还在使用手工帐的企业了。

如图 6.43 列示了应收账款账期计算表，A 列为应收款金额，B 列为正常回款期，需要在 D-K 列中把应收款金额放在对应的位置上。

	A	B	C	D	E	F	G	H	I	J	K
1	核算日期	2012年7月5日	账期	-1000	0	30	60	90	180	365	730
2	应回款金额	正常回款期	天数	未到期	30天以内	31-60天	61-90天	91-180天	181天-1年	1年至2年	2年以上
3	29,986.73	2012年3月25日	102					29,986.73			
4	19,433.66	2012年4月25日	71				19,433.66				
5	16,428.02	2012年5月25日	41			16,428.02					
6	334,552.04	2012年7月30日	-25	334,552.04							
7	214,376.98	2011年6月30日	371							214,376.98	
8	55,934.36	2012年6月30日	5		55,934.36						
9	10,671.00	2010年6月30日	736								10,671.00
10	51,511.57	2011年12月25日	193						51,511.57		

图 6.43 应收账款账期计算表

### 1) 问题分析

如需要账期随系统日期自动更新，就需要使用日期函数 TODAY()，这里为演示方便，使用了 B1 单元格中的时间“2012 年 7 月 5 日”。

D1: K1 是辅助区域，对应到 D2: K2 中的账期，为使用模糊查找公式做好准备。

### 2) 解决方案

先在 C3 单元格录入公式“=B\$1-B3”，下拉完成辅助列的设置；然后在 D3 单元格中录入如下公式，向下向右复制完成。

=IF(COLUMN()-3=MATCH(\$C3,\$D\$1:\$K\$1),\$A3,"")

### 3) 公式解析

C 列公式计算账期天数，因 B 列的正常回款时间可能迟于核算日期，C 列计算结果会存在负数。

D3 公式外层是 IF 函数，第 1 参数用 MATCH 函数查找的结果与当前单元格列号减去 3 比较。MATCH 函数省略了第 3 参数，代表模糊查找（详见第 5 章中的讲解），减去 3 是因为在图 6.43 的表格中 D 列前面还有 3 列。

公式的含义为如果 MATCH 函数在 D1:K1 辅助区域中查找账期返回值与当前单元格列号减 3 相等，则在这个单元格中返回应收款金额，否则返回空值。

使用辅助区域能起到简化公式的作用，在计算完成后可以隐藏 C 列和第 1 行。

职场点睛：这个案例给我们一个启示，实际工作中类似于应收账款账龄之类的计算也是可以使用函数公式自动生成的，Excel 的功能很强大。

本节示例文件：《6.5 日期时间函数之实战应用.xlsx》

## 6.6 小结

本章从日期时间的实质谈起，介绍了一些常用的日期时间函数及其在日常工作中的应用。使用日期

时间函数，思路是多样的，这正是 Excel 函数魅力之所在。第 5 节中的举例有一定的难度，公式开始长起来了，有了些函数综合应用的味道。其实局限于目前学到的函数，这些还是比较简单的，实战中的应用更加复杂，相关内容将在之后的章节中讲述。

如果读者还在使用 2003 版的 Office 软件，在打开本章的示例文件时可能会出错，原因在于日期函数中有一些在 2003 版中需要加载“分析工具库”后才能使用，涉及到的函数有 EDATE、EOMONTH、WEEKNUM、WORKDAY、NETWORKDAYS 等。

在 2003 版中单击“工具”菜单下的“加载宏”命令，在弹出的“加载宏”对话框中勾选“分析工具库”，单击“确定”按钮即可完成加载，如图 6.44 所示。




图 6.44 在 2003 版中加载“分析工具库”