

第4章 函数公式之内功心法

在 Excel 工作表中分析数据必然要用到函数公式。什么是函数？什么是公式？函数和公式是不是一回事？

比较官方的定义：函数公式就是由用户自行设计并结合常量数据、单元格引用、运算符等元素进行数据处理和计算的算式。说得形象一点，函数公式就是 26 个英文字母、10 个阿拉伯数字、若干运算符的排列组合及其恩怨情仇纠缠一生。

有人把 Excel 中内置的操作技巧（比如单元格自定义格式、数据有效性、条件格式等）比作武术中的外功，把函数公式比作内功。相对而言外家功夫学起来更快，但如果没有内家功夫支持，练到一定的程度就会遭遇瓶颈，杀伤力有限也难以继续突破。大家都看过功夫片，应该知道什么叫做花拳绣腿，没有函数功底支持的 Excel 内置功能就好比是花拳绣腿。

读者如有进一步提升 Excel 操作水平的需求，函数运用能力的提高无疑是无法回避的问题。但要提升函数水平，有必要先搞清楚一些基本的概念，实际工作中很多问题都源自基本概念的模糊。本章将从最基础的知识点讲起，引领读者一起来修炼 Excel 函数公式的内功心法。

本章概念性讲述较多，如有一定函数基础的用户，可以跳过直接阅读后续章节。

4.1 函数学习的若干误区

函数公式是 Excel 内置功能中最精彩的部分，但对于函数公式的理解很多用户尚存在误区，有些已经影响到 Excel 应用水平的进一步提升。

4.1.1 函数很多很难学

曾经见到有人排过一份学习日程表，一天学习 10 个函数，双休日努力一下加倍学习，在一个月之内学会全部的 300 多个函数。有点震惊，不知道这位朋友最后有没有完成函数的学习大业。如果有人认识掌握 Excel 中全部函数的高人，不妨介绍认识一下。

Excel 中的函数虽多，但常用的并不多，不会超过 30 个。没有必要学会全部函数，相信也没有人会掌握所有的函数。再退一步，即使真的有人会全部函数那也未必能算是函数高手，真正的高手强在灵活应用。

学习函数可以从最基础最常用的函数着手，然后结合自己的行业与工作内容进一步学习。Excel 中已对内置函数有一个分类，如图 4.1 所示。




图 4.1 常用函数的分类

用户可以结合自己的工作内容有选择地学习，毕竟学习的目的在于提高工作效率！

4.1.2 函数无用论

这个观点应该无需驳斥了，通过本书前3章的学习，读者可以发现即使是一些内置的 Excel 功能（如数据有效性、条件格式等）的应用也离不开函数的支持。

抛开函数对于 Excel 内置功能的扩展，函数最大的功效在于数据能够自动更新。如图 4.2 所示，C5 单元格用函数计算了 A1:A5 单元格数值的合计，公式为“=SUM(A1:A5)”。




图 4.2 公式的计算结果会自动更新

如不使用函数公式，直接手工在 C5 单元格录入结算的结果 15，这样在 A1:A5 单元格区域数值变动后，还需要手工改动 C5 的值。如使用公式即会自动更新。

4.1.3 函数万能论

相对于函数无用论，函数万能论的破坏力更强。

1. 函数不能解决的问题

函数的确很精彩，但函数绝对不是万能的，很多问题用函数公式会很困难，甚至是无

法解决。这里举几个简单的例子：

- ❑ 跨工作簿、跨工作表汇总、计算；
- ❑ 提取工作簿中的信息，如当前工作表名等；
- ❑ 提取工作表中非数据层的信息，如设置了单元格自定义格式后实际显示的内容。

2. 解决问题的思路

函数并非万能，但可以与 Excel 中的其他功能结合运用，遇到问题建议的解决思路如下：

- ❑ 首先是规范制表，如第 1 章中提到的打造自己的数据源表；
- ❑ 其次在数据源规范的情况下，优先使用 Excel 的内置功能；
- ❑ 再次是使用函数计算及扩展 Excel 的内置功能；
- ❑ 最后是使用 VBA 功能自定义一些解决方案。

4.2 录入函数公式的两种方法

了解了函数学习的若干误区之后，从本节开始正式接触函数公式。首先讲讲如何在单元格中录入函数公式。

4.2.1 直接录入函数公式

用户输入函数公式时，通常以“=”开始，否则 Excel 会自动将其识别为文本。

当单元格中首先输入“=”时，Excel 就会识别其为公式的开始。例如，要在工作表的 G1 单元格输入一个函数公式“=SUM(B2:B4)”，在输入“=”后再输入一个英文字母 S（英文半角状态，大小写皆可），此时 Excel 会自动带出所有以字母 S 开头的函数，如图 4.3 所示。




图 4.3 直接录入函数公式

此时手工录入完整的函数公式或者通过滚动条选择 SUM 函数后再选择参数的数据区域皆可，最后按 Enter 键结束公式的输入。

由操作过程可见，只要输入函数的第一个字母，Excel 会智能地把由这个字母开头的函数全部带出来，如输入字母 S 后再输入字母 U，则会带出以“SU”开头的函数。


这个功能方便了用户记忆函数公式的字母拼写。

如果读者在使用时没有这项功能，可能和版本有关，在 2007 版以下是没有这项功能的。如在 2007 版以上未显示这项功能，读者可以检查一下 Excel 选项中的设置，以 2010 版为例，单击“文件”|“选项”|“公式”命令，勾选“公式记忆式键入”，如图 4.4 所示。




图 4.4 公式记忆式键入

4.2.2 插入函数公式

用户也可以通过插入的方式来录入函数公式，单击“公式”|“函数库”|“插入函数”命令，或者直接单击编辑栏左边的“插入函数”按钮 ，此时会显示如图 4.5 所示的对话框。

这种操作方法的特点是可以快速在类别中选择需要的函数，有时候用户并不了解函数的具体功能，也可以通过预览的函数功能帮助来选择满足自己需要的函数。

在“选择函数”框中选取 SUM 函数，单击“确定”按钮后，出现如图 4.6 所示的对话框。

此时在 Number1 参数框中录入函数的参数“B2: B5”，或者单击参数框选取按钮 ，选取需要求和运算的单元格区域，返回“函数参数”对话框后单击“确定”按钮也能返回

正确的结果。




图 4.5 “插入函数”对话框




图 4.6 “函数参数”对话框

4.2.3 两种输入函数公式方法的比较

第二种方法更加适合初学者，但不适合输入复杂的嵌套函数公式。对于函数有兴趣的读者，要逐步适应第 1 种方法。

4.3 Excel 中的引用样式和引用类型

Excel 中有两种引用样式和 4 种引用类型，理解这些概念是进一步学习函数的公式的基础。

4.3.1 Excel 中的两种引用样式

关于引用样式这个概念很多人都很模糊，不熟悉函数的用户可能没接触过这方面的概念。

1. 了解Excel中的两种引用样式

引用样式分两种，默认的是“A1 引用样式”，说得通俗一点，就是单元格所在的行号用数字表示（2003 版是 1~65 536，2007 版以上是 1~1 048 576），单元格所在的列标用英文大写字母表示（2003 版是 A~IV，2007 版以上是 A~XFD），表示单元格为 A1、IV65536、XFD1048576 等。

另一种引用样式为“R1C1 引用样式”，使用字母 R 与数字来表示第几行，使用字母 C 与数字来表示第几列。表 4.1 为两种引用样式的比较。

表 4.1 两种引用样式的比较

A1 引用样式	R1C1 引用样式
\$A\$1	R1C1
D5	R[?][C[?]]
\$A\$1:\$D\$5	R1C1:R5C4

2. 两种引用样式的切换

默认的是“A1 引用样式”，用户可以单击“文件”|“选项”命令，在弹出的“Excel 选项”对话框中的“公式”选项卡下勾选“R1C1 引用样式”来改变默认的引用样式，如图 4.7 所示。



图 4.7 切换 Excel 中的引用样式

如改变默认的“A1 引用样式”为“R1C1 引用样式”之后，Excel 界面的列标题将会全部变成数字，如图 4.8 所示。




图 4.8 R1C1 引用样式的外观

这个改动足以让那些对 Excel 不熟悉的小 MM 们不知所措，接下来就可以上演英雄救美的戏了。

4.3.2 从一个案例来理解单元格的引用类型

如在公式中使用单元格引用，需要考虑目标单元格与引用单元格之间的引用类型，不同的引用类型涉及到的写法是不一样的。在 Excel 中，公式所在单元格与公式所引用单元格的位置关系称为单元格引用的类型。下面就通过一个案例来理解一下单元格的引用类型。

案例：计算业务员销售额提成，如图 4.9 所示。案例工作表中 A 列是业务员姓名，B 列是相关业务员的销售额，在 F1 单元格中存放业务员的销售额提成率，需要在 C 列计算业务员的销售提成。

	A	B	C	D	E	F	G
1	业务员	销售额	销售提成		提成率	0.01%	
2	业务员1	6,571,654	657.17				
3	业务员2	3,282,680					
4	业务员3	6,684,780					
5	业务员4	1,793,355					
6	业务员5	7,908,358					
7	业务员6	5,986,309					
8	业务员7	5,945,807					
9	业务员8	3,855,689					
10	业务员9	2,289,668					
11							

图 4.9 计算业务员销售提成

备注：

(1) 销售提成根据各个业务员的销售额和提成率来计算；

(2) 销售提成 = 销售额 * 提成率。

已经在 C2 单元格录入公式“=B2*F1”，如果直接把公式向下复制（拖动），C3 以后的单元格得到的结果全部为 0。

原因在于随着公式向下复制，公式中引用的单元格会自动改变，比如 C3 单元格中是“=B3*F2”，C4 单元格中是“=B4*F3”，公式中“*”号右边引用的单元格并非一直是需要的 F1 单元格，如图 4.10 所示。

	A	B	C	D	E	F
1	业务员	销售额	销售提成		提成率	0.01%
2	业务员1	6,571,654	657.17	=B2*F1		
3	业务员2	3,282,680	-	=B3*F2		
4	业务员3	6,684,780	-	=B4*F3		
5	业务员4	1,793,355	-	=B5*F4		
6	业务员5	7,908,358	-	=B6*F5		
7	业务员6	5,986,309	-	=B7*F6		
8	业务员7	5,945,807	-	=B8*F7		
9	业务员8	3,855,689	-	=B9*F8		
10	业务员9	2,289,668	-	=B10*F9		

图 4.10 错误的业务员销售提成公式

正确的方法是改变 C2 单元格中公式引用单元格的类型，将 C2 单元格中的公式改为“=B2*\$F\$1”，再向下复制公式即能得到正确的结果。这时候我们再检查 C3 单元格的公式，已经是“=B3*\$F\$1”，C3 之后的单元格也全部引用的是 F1 单元格，如图 4.11 所示。

	A	B	C	D	E	F
1	业务员	销售额	销售提成		提成率	0.01%
2	业务员1	6,571,654	657.17	=B2*\$F\$1		
3	业务员2	3,282,680	328.27	=B3*\$F\$1		
4	业务员3	6,684,780	668.48	=B4*\$F\$1		
5	业务员4	1,793,355	179.34	=B5*\$F\$1		
6	业务员5	7,908,358	790.84	=B6*\$F\$1		
7	业务员6	5,986,309	598.63	=B7*\$F\$1		
8	业务员7	5,945,807	594.58	=B8*\$F\$1		
9	业务员8	3,855,689	385.57	=B9*\$F\$1		
10	业务员9	2,289,668	228.97	=B10*\$F\$1		

图 4.11 正确的业务员销售提成公式

比较前后 2 个公式，差异仅在于 F1 与 \$F\$1，这两者有什么区别呢？这里给读者介绍一下 Excel 中单元格的引用类型。

在我们使用的默认 A1 引用样式下，引用单元格的行号和列标之前都没有“\$”符号，如“F1”，我们称之为相对引用；如引用单元格的行号和列标之前都有“\$”符号，如“\$F\$1”，我们称之为绝对引用；如果仅行前有“\$”符号或者仅列前有“\$”符号，如“\$F1”或者“\$F1”，我们称之为混合引用。

混合引用又可以分为“行绝对列相对混合引用”和“行相对列绝对混合引用”两种，Excel 中的单元格引用类型共分 4 种，如表 4.2 所示。

表 4.2 单元格的 4 种引用类型

A1 样式	R1C1 样式	特 点
\$A\$1	R1C1	绝对引用，复制公式时不改变引用单元格地址
A1	R[?]C[?]	相对引用，复制公式时会改变引用单元格地址
A\$1	R1C[?]	行绝对列相对，纵向复制公式时不改变引用单元格地址，横向复制公式时会改变
\$A1	R[?]C1	行相对列绝对，纵向复制公式时会改变引用单元格地址，横向复制公式时不改变

备注：上表中的“?”代表数字，表示在相对引用时，公式所在单元格与公式引用单元格之间的行、列差异数字。

这4种单元格引用类型比较难理解，初学者不必刻意地记忆到底是属于哪一种引用类型，可以在实践中慢慢理解，错过几次后就能豁然开朗融会贯通了。

这里介绍一下选择判断的原则：

- ❑ 如在单一单元格中，不存在公式复制的需求，此时不需要考虑单元格引用的类型；
- ❑ 如公式需要纵向复制，需要考虑是否要改变引用单元格的行引用类型；
- ❑ 如公式需要横向复制，需要考虑是否要改变引用单元格的列引用类型。

4.3.3 快速改变单元格引用类型

单元格的4种引用类型可以通过添加及删除“\$”符号来改变，Excel提供了一个快捷键F4，可以快速在4种引用类型之间切换。

使用F4键来改变单元格的引用类型，必须先选中公式所在的单元格，然后在编辑栏中选中需要改变引用类型的单元格地址，如不选择直接按F4键，此时Excel执行的是“重复上一次操作”。

如要改变C2单元格中引用F1单元格的引用类型，需要在编辑栏中选中“F1”，如图4.12所示。




图 4.12 改变单元格的引用类型

此时连续按F4键，可以发现单元格的引用类型依次在相对引用、绝对引用、行绝对列相对引用、行相对列绝对引用之间切换。用户可以通过按F4键快速改变单元格的引用类型。

备注：这种方法需要进入编辑栏操作，因此在批量改变某区域的单元格引用类型时有一定的局限性，Excel的内置功能无法完美地做到批量改变区域的单元格引用类型，VBA可以提供解决方案。

仍使用上例数据，如在I列和J列中引用业务员和销售提成资料，如需要快速地把相对引用改成绝对引用，可以使用以下的VBA代码：

```
Sub 相对引用转换成绝对引用 ()
Dim rng As Range, N As Long
Set rng = Application.InputBox("请输入需要转换的区域", "输入区域", Type:=8)
```

```

rng.Select
For N = 1 To rng.Count
    Application.SendKeys "{F2}"
    Application.SendKeys "+{HOME}"
    Application.SendKeys "{F4}"
    Application.SendKeys "~"
Next N
End Sub

```

单击工作表中已添加的 **批量改变相对引用到绝对引用** 按钮，在弹出的“输入区域”对话框中选择要转换引用类型的区域，单击“确定”按钮后即可自动更改选择区域的引用类型。如图 4.13 所示。



图 4.13 “输入区域”对话框

如读者打开示例文件，单击工作表中的 **批量改变相对引用到绝对引用** 按钮后出现类似如图 4.14 所示的提示框，请检查一下“宏设置”。



图 4.14 无法运行宏提示框

如果读者未改变 Excel 默认的“宏设置”，在打开启用宏的工作簿时，Excel 会自动禁用宏，并弹出安全警告信息，如图 4.15 所示，单击“启用内容”按钮后，安全警告信息消失，这时可以执行打开工作簿中的所有宏。




图 4.15 未启用宏时的安全警告

用户如需避免频繁地弹出宏安全警告信息，可以单击“文件”|“选项”|“信任中心”|“信任中心设置”按钮，进入“信任中心”后选择“宏设置”选项卡，更改宏设置。默认的设置是“禁用所有宏，并发出通知”，如图 4.16 所示，不建议读者改变默认的设置。

备注：

(1) 关于 VBA 编程非本书重点，这里仅介绍如何启用宏、使用宏；

(2) 所有的代码都有自身的局限性，并无所谓完全通用的，以上代码仅是将选择区域的相对引用全部转换为绝对引用，如有其他需求，有兴趣的读者可以自行研究一下。




图 4.16 信任中心——宏设置

4.3.4 引用类型的常见应用——编制九九乘法表

关于单元格的引用类型，需要反复练习才能真正掌握。这里举一个典型的例子来帮助读者理解。

如图 4.17 所示，以第 1 行和第 1 列数字为辅助单元格，在 B2: J10 单元格区域中编制九九乘法表，要求在 B2 单元格中录入一个公式，然后通过向下向右复制公式完成。

	A	B	C	D	E	F	G	H	I	J
1		1	2	3	4	5	6	7	8	9
2	1	1*1=1								
3	2	1*2=2	2*2=4							
4	3	1*3=3	2*3=6	3*3=9						
5	4	1*4=4	2*4=8	3*4=12	4*4=16					
6	5	1*5=5	2*5=10	3*5=15	4*5=20	5*5=25				
7	6	1*6=6	2*6=12	3*6=18	4*6=24	5*6=30	6*6=36			
8	7	1*7=7	2*7=14	3*7=21	4*7=28	5*7=35	6*7=42	7*7=49		
9	8	1*8=8	2*8=16	3*8=24	4*8=32	5*8=40	6*8=48	7*8=56	8*8=64	
10	9	1*9=9	2*9=18	3*9=27	4*9=36	5*9=45	6*9=54	7*9=63	8*9=72	9*9=81

图 4.17 编制九九乘法表

问题分析：观察图 4.17 中 B2: J10 单元格区域的显示结果，九九乘法表显示在整个区域的左下方，区域右上方为空。内容非全部显示，可以先使用 IF 函数屏蔽部分返回结果。

了解这个规律之后，还需将其转换成函数能够读懂的语言，继续分析有显示内容的单元格：

- ❑ I9 单元格左侧辅助单元格数据 8 等于上方辅助单元格数据 8；
- ❑ D5 单元格左侧辅助单元格数据 4 大于上方辅助单元格数据 3。

其他单元格规律以此类推，如图 4.18 所示。

	A	B	C	D	E	F	G	H	I	J
1		1	2	3	4	5	6	7	8	9
2	1	1*1=1								
3	2	1*2=2	2*2=4							
4	3	1*3=3	2*3=6	3*3=9						
5	4	1*4=4	2*4=8	3*4=12	4*4=16					
6	5	1*5=5	2*5=10	3*5=15	4*5=20	5*5=25				
7	6	1*6=6	2*6=12	3*6=18	4*6=24	5*6=30	6*6=36			
8	7	1*7=7	2*7=14	3*7=21	4*7=28	5*7=35	6*7=42	7*7=49		
9	8	1*8=8	2*8=16	3*8=24	4*8=32	5*8=40	6*8=48	7*8=56	8*8=64	
10	9	1*9=9	2*9=18	3*9=27	4*9=36	5*9=45	6*9=54	7*9=63	8*9=72	9*9=81

图 4.18 九九乘法表中显示数据分析

由此可以找到数据显示的规律，即左侧引用的辅助单元格的值大于等于上方引用的辅助单元格的值的情况下显示九九乘法表，否则显示为空。

解决方案：在 B2 单元格录入如下公式，并向下向右复制，填充整个 B2: J10 单元格区域。

```
=IF($A2>=B$1,B$1&"*"&$A2&"="&B$1*$A2,"")
```

公式解析：这里使用了一个 IF 函数，IF 函数的语法结构如下：

```
IF（逻辑值,逻辑值为 TRUE 时的返回结果,逻辑值为 FALSE 时的返回结果）
```

第 1 个参数使用了“\$A2>=B\$1”，即判断左侧引用的辅助单元格的值是否大于等于上方引用的辅助单元格的值；

如判断结果为 TRUE 则返回第 2 个参数，判断结果为 FALSE 则返回第 3 个参数，这里两个半角状态下的双引号“”代表返回空值。

这个例子的关键是在 B2 单元格中引用辅助单元格时的引用类型，分析如下：

比如引用 A2 单元格，因公式向下复制时需引用到 A 列的其他单元格，公式向右复制时仍需要引用 A 列，所以列是绝对引用，行是相对引用，应更改为 \$A2；

对于 B1 的引用正好相反，属于行绝对引用，列相对引用，应更改为 B\$1。

公式中还出现了一个连接符“&”，其作用是将“&”两边的内容连接起来，如连接的内容是文本，还需在外面加上一对半角状态下的双引号，如连接“*”和“=”时的写法。

4.4 定义名称的妙用

单元格引用类型是一个很基础的概念，读者应该能够完全理解应用。不是很难但频繁更改单元格的引用类型的确有点麻烦，有时候可以通过定义名称的方法来简化公式。

4.4.1 通过定义名称简化公式

沿用上一节中计算提成的例子，在 D1 单元格中输入公式“=B2*提成率”，并向下复制，得到的结果和 C 列完全一样。但我们观察 D 列的公式，“*”号右边都是“提成率”，未涉及到引用单元格 F1 及其引用类型的改变，如图 4.19 所示。

	A	B	C	D	E	F
1	业务员	销售额	销售提成		提成率	0.01%
2	业务员1	6,571,654	657.17	657.17	=B2*提成率	
3	业务员2	3,282,680	328.27	328.27	=B3*提成率	
4	业务员3	6,684,780	668.48	668.48	=B4*提成率	
5	业务员4	1,793,355	179.34	179.34	=B5*提成率	
6	业务员5	7,908,358	790.84	790.84	=B6*提成率	
7	业务员6	5,986,309	598.63	598.63	=B7*提成率	
8	业务员7	5,945,807	594.58	594.58	=B8*提成率	
9	业务员8	3,855,689	385.57	385.57	=B9*提成率	
10	业务员9	2,289,668	228.97	228.97	=B10*提成率	

图 4.19 通过定义名称简化公式

职场点睛：这里的“提成率”实际上是一个已定义的“名称”。建议用户在实际工作中可以将公式中涉及到的常量定义为名称，这样公式不仅易于理解而且还起到了简化录入的效果。

4.4.2 创建名称的 3 种方法

定义名称的方法有多种，最常用的方法有 3 种，这里做一些简单介绍。

1. 使用功能区命令

单击“公式”|“定义的名称”|“定义名称”，在下拉菜单中选择“定义名称”命令，弹出“新建名称”对话框，如图 4.20 所示。例如，定义一个名称“VAT”，数值为 0.17，在“名称”框中输入名称为“VAT”，引用位置为“=0.17”，如图 4.21 所示，单击“确定”按钮完成名称的新建。




图 4.20 “新建名称”对话框




图 4.21 新建名称“VAT”

2. 使用“名称框”

其次可以通过“名称框”来快速定义名称，在示例文件中选取“定义名称”工作表中的“B2:E5”单元格区域，在名称框中输入需要定义的名称 data，如图 4.22 所示，按回车键结束。

3. 根据所选内容创建

还可以通过工作表中已有内容来创建名称，在示例文件中选取“定义名称”工作表中

的“A1:E5”区域，单击“公式”|“定义的名称”|“根据所选内容创建”命令，弹出“以选定区域创建名称”对话框，如图 4.23 所示，单击“确定”按钮即可创建以选定区域首行及最左列内容命名的名称。




图 4.22 使用“名称框”创建名称




图 4.23 根据所选内容创建名称

4.4.3 管理已定义的名称

Excel 中提供了名称管理器功能，可以快速地管理工作簿中已定义的名称。

单击“公式”|“定义的名称”|“名称管理器”命令，弹出“名称管理器”对话框，如图 4.24 所示。也可以使用 Ctrl+F3 快捷键调出名称框。






图 4.24 名称管理器

在“名称管理器”中列出了本工作簿中所有已定义的名称，读者可以自行尝试一下上文中所述的 3 种方法，观察一下“名称管理器”中显示内容的变化。

“名称管理器”中也可以“新建”名称，但多数情况下用于“编辑”和“删除”名称，比如要修改名称 data 的引用范围，可以在选中名称 data 后单击“编辑”按钮，弹出“编辑名称”对话框，如图 4.25 所示。

如需将引用范围改为单元格“A1:E5”，可直接在引用位置框“=定义名称!\$B\$2:\$E\$2”中修改引用范围，或者单击引用位置框右边的选取按钮，

弹出“编辑名称-引用位置”对话框，如图 4.26 所示，直接选取需要更新的引用区域后再单击选取按钮回到图 4.25 所示的“编辑名称”对话框中，最后单击“确定”按钮完成名称 data 的编辑。







图 4.25 “编辑名称”对话框



	A	B	C	D	E
1		一季度	二季度	三季度	四季度
2	A公司	228	289	100	297
3	B公司	287	155	236	134
4	C公司	174	270	148	166
5	D公司	300	142	124	127

图 4.26 “编辑名称-引用位置”对话框

备注：有时候名称“引用位置”框中内容比较多，读者如按习惯的操作方式在“引用位置”框中使用左右方向键移动光标所在位置，往往会出现一些出乎意料的结果。解决方案是在进入“引用位置”框后先按 F2 键，切换到“编辑”状态后才能在“引用位置”框中正常移动光标。

切换之前左下角“状态栏”中显示的是“点”，切换后显示的“编辑”。

删除名称的操作更简单，比如要删除名称 VAT，选取名称 VAT 后，单击“删除”按钮，弹出删除名称确认框，如图 4.27 所示，单击“确定”按钮即可删除已定义的名称 VAT。




图 4.27 删除名称确认框

4.4.4 快速查看工作簿中已定义的名称

一个工作簿中可能已经定义了很多名称，我们虽然可以通过“名称管理器”来管理这些名称，但如果一个一个去看名称的“引用位置”很不方便，Excel 提供了“粘贴名称”的功能。

先选中需要粘贴名称的起始单元格，如 G1 单元格，单击“公式”|“定义的名称”|“用于公式”|“粘贴名称”命令，弹出“粘贴名称”对话框，如图 4.28 所示（也可直接使用 F3 键），单击“粘贴列表”按钮，就在选中的单元格区域粘贴了工作簿中所有非隐藏

状态的名称，如图 4.29 所示。



图 4.28 “粘贴名称”对话框

G	H	I	J
A公司	=定义名称!\$B\$2:\$E\$2		
B公司	=定义名称!\$B\$3:\$E\$3		
C公司	=定义名称!\$B\$4:\$E\$4		
data	=定义名称!\$A\$1:\$E\$5		
D公司	=定义名称!\$B\$5:\$E\$5		
VAT	=0.17		
二季度	=定义名称!\$C\$2:\$C\$5		
三季度	=定义名称!\$D\$2:\$D\$5		
四季度	=定义名称!\$E\$2:\$E\$5		
提成率	=相对引用与绝对引用!\$F\$1		
一季度	=定义名称!\$B\$2:\$B\$5		

图 4.29 已粘贴的名称列表

4.4.5 名称的命名规则及使用范围

Excel 对名称的命名有其限制，并不能随心所欲地命名名称。

如拟在工作表中插入一个名称“4 季度”，在“新建名称”对话框的“名称”框中输入“4 季度”，“引用位置”框中选取范围后，单击“确定”按钮，则会弹出“输入的名称无效”提示框，如图 4.30 所示。




图 4.30 “输入的名称无效”提示框

提示框中列出的这 3 条，读者需要理解透彻，这已经包含了命名名称规则的绝大部分，这里再补充容易混淆的几条：

- ❑ Excel 识别名称时，不区分字母大小写，如在单元格中输入“=VAT”和“=vat”，返回的结果是一样的；
- ❑ 当名称引用单元格区域时，需要考虑单元格引用的类型，在多数情况下需要使用单元格的绝对引用类型，相关内容会在以后章节中展开；
- ❑ 在使用“根据所选内容创建”名称功能时，如设置的默认名称不符合 Excel 的命名规则，Excel 会自动加以修正。修正的规则是在数字前面加上“下划线”，空格用“下划线”代替，无效字符直接删除。

如“定义名称”工作表中“A14: D14”单元格中的内容，如果作为要定义的名称，直接以插入名称的方式新增，Excel 会报错，提示如图 4.30 右边所示。但如果选择“A14: D18”区域，以选中的区域创建名称，Excel 能执行命令，但会自动修正一下名称。

读者可以自行测试一下根据选取区域“A14: D18”创建名称的结果。

4.5 公式调试及运行结果分析

使用函数公式，难免会遇到出错的情况。公式很长，到底是哪一段出错了？有时候公式运行的结果并没有报错，但其返回值明显不符合最初编写公式的要求。面对如此种种情况，有必要深入了解一下函数公式运行的机制，找到出错原因。


4.5.1 错误的类型及其形成原因

在 Excel 中使用函数公式常见的错误有 8 种，如表 4.3 所示。

表 4.3 Excel 中函数公式常见的错误类型

返回值	错误名称	错误原因分析
#DIV/0!	被“零”除错误	当一个数除以零 (0) 或不包含任何值的单元格时，Excel 将显示此错误
#N/A	“值不可用”错误	常见于查找函数未能返回结果
#NAME?	“无效名称”错误	如函数名拼写错误，函数中使用的文本参数未加引号等
#NULL!	“空值”错误	如果在工作表上指定两个并不相交的区域的交集，则将出现此错误
#NUM!	数字错误	可能在需要数字参数的函数中提供了错误的数据类型
#REF!	“无效的单元格引用”错误	常出现于公式中引用的原单元格被删除或返回的区域不存在
#VALUE!	值错误	常见于公式中使用了错误的数据类型，无法正确计算出结果
#####	值显示错误	可能是列宽不足以显示所有内容，或者在单元格中使用了负日期或时间

4.5.2 使用“错误检查”标志分析错误形成原因

了解了各种错误产生的原因，即可在出现错误的时候采取适当的措施。即使忘记了这 8 种错误产生的原因也无需紧张，如在 Excel 中启用了错误检查（默认设置），则单击显示错误的单元格旁边的错误检查标志，Excel 会有智能的提示。

如图 4.31 中 A3 单元格中的公式为“=8/0”，运算结果出错，单击错误检查标志，在弹出的下拉菜单中即能显示出错的原因。

关于 Excel 中的错误检查规则，可以在 Excel 选项中更改。单击“文件”|“选项”命令，在弹出的“Excel 选项”对话框中选择“公式”选项卡，最下方即是与错误检查有关的设置，如图 4.32 所示，建议不更改默认的设置。

4.5.3 使用“公式求值”工具分析公式运算过程

在了解了函数公式运行结果报错的原因之后，再看看那些没有报错的错误。相对于报错的错误，不报错的错误更加隐蔽，很多是属于公式编写中的逻辑问题。这里通过一个

学生成绩分级的例子来介绍一下常用的错误分析查找工具。




图 4.31 使用错误检查标志分析错误



图 4.32 Excel 选项中与错误检查有关的设置

在“运行结果分析”工作表中，A 列是学生姓名，B 列是考试成绩，需要在 C 列返回成绩的等级，相关成绩分级规则在“G1: H5”单元格区域。

解决方案有很多，如果使用 IF 公式，正确的公式如下：

```
=IF(B2>=90,"优秀",IF(B2>=80,"良好",IF(B2>=60,"及格","不及格")))
```

但在实际工作中，很多人不理解 IF 函数的运行机制，写出一些五花八门的错误公式，然后面对错误的结果束手无策，以下列举一个错误的公式：

```
=IF(B2>=90,"优秀",IF(B2>=0,"不及格",IF(B2>=60,"及格","良好")))
```

C 列返回的是正确的等级，D 列是错误的等级，如图 4.33 所示，错误的地方已用黄色单元格标示。

	A	B	C	D	E	F	G	H
1	学生	成绩	正确的等级	错误的等级			分数	等级
2	A	91	优秀	优秀			[0,60)	不及格
3	B	61	及格	不及格			[60,80)	及格
4	C	58	不及格	不及格			[80,90)	良好
5	D	65	及格	不及格			[90,100]	优秀
6	E	84	良好	不及格				
7								
8	正确的公式	=IF(B2>=90,"优秀",IF(B2>=80,"良好",IF(B2>=60,"及格","不及格")))						
9	错误的公式	=IF(B2>=90,"优秀",IF(B2>=0,"不及格",IF(B2>=60,"及格","良好")))						

图 4.33 计算学生成绩等级

对于有一定函数使用经验的用户，发现此类错误并不困难，但对于初学者就有一定的难度了。这里使用 Excel 的“公式求值”工具，查看公式分步运算结果。

因学生 B 的等级出错，对 D3 单元格的公式加以分析，选取 D3 单元格，单击“公式”|“公式审核”|“公式求值”命令，弹出“公式求值”对话框，如图 4.34 所示。




图 4.34 “公式求值”对话框

单击“求值”按钮，观察“运行结果分析”框中的变化，在单击 4 次“求值”按钮后，得到的结果如图 4.35 所示。




图 4.35 4 次“求值”后的运算结果

通过观察“运行结果分析”框中的变化，可以发现 IF 函数公式是从左往右执行的，如第一个判断不满足，再执行第 2 个判断，以此类推。

以 D3 单元格公式为例，第 1 个条件“ ≥ 90 ”不满足，执行第 2 个条件判断；第 2 个条件“ ≥ 0 ”是满足的，不会再执行下一个判断，直接返回值“不及格”。

D4: D6 单元格的情况一样，都是满足了第 2 个条件就返回值，之后的条件不会被判断执行。

由此可以明白这个 IF 函数公式出错的原因在于公式编写时的逻辑错误，一般使用 IF 函数多层嵌套时，条件都是从大到小或者是从小到大的，跳跃的条件值很容易出错。

备注：并非所有的嵌套函数公式都是从左往右执行的，更多的是先执行最内层的公式，从内往外执行。

4.5.4 使用 F9 键查看公式运算结果

使用“公式求值”工具可以看到公式运行的结果及运算的步骤，对于有经验的用户，更加关注的是公式运行的结果。相对于“公式求值”工具，使用 F9 键查看公式运算结果会更加灵活且直观。

如提取邮箱地址中的名字，邮箱地址的命名规则为“姓.名字@服务器域名”，如图 4.36 所示。

	A	B
1	邮箱	名字
2	Zhang.Xiaoning@app.com.cn	Xiaoning
3	Huang.Zhaojun@app.com.cn	Zhaojun
4	Deng.Hua@app.com.cn	Hua
5	Inter.Jon@app.com.cn	Jon
6	Nana.Wilms@app.com.cn	Wilms
7	Wang.Linsheng@app.com.cn	Linsheng
8	Fu.Guojiang@app.com.cn	Guojiang

图 4.36 提取邮箱地址中的名字

解决这个问题，在 B2 单元格使用如下公式，并向下复制，得到正确的结果。

```
=MID(LEFT(A2,FIND("@",A2)-1),FIND(".",A2)+1,99)
```

这是一个函数嵌套公式，虽然涉及到的 3 个函数尚未讲解过，但并不影响分析公式运算过程。

要了解这个公式的整体架构，可以进入编辑栏，将鼠标插入到公式的不同位置中，观察编辑栏下方的“函数屏幕提示”（如未显示“函数屏幕提示”，可进入“Excel 选项”对话框，在高级选项卡下的“显示”组中勾选“显示函数屏幕提示”）。

如将鼠标插入 LEFT 函数的函数名中，“函数屏幕提示”如图 4.37 所示，显示了一个 MID 函数，且第一个参数“text”以粗体显示，含义是这个 LEFT 函数是 MID 函数的第一个参数。

=MID(LEFT(A2,FIND("@",A2)-1),FIND(".",A2)+1,99)
MID(text, start_num, num_chars)

图 4.37 “函数屏幕提示” 1

再将鼠标插入第 1 个 FIND 函数的函数名中，此时“函数屏幕提示”如图 4.38 所示，显示了一个 LEFT 函数，且第二个参数“[num_chars]”以粗体显示且外面有一对半角状态的“[]”，含义是这个 FIND 函数是 LEFT 函数的第二个参数，且对于 LEFT 函数第二个参数不是必须的。

=MID(LEFT(A2,FIND("@",A2)-1),FIND(".",A2)+1,99)
LEFT(text, [num_chars])

图 4.38 “函数屏幕提示” 2

公式的其他部分显示以此类推，使用这种方法不管嵌套公式有多复杂，都能搞清楚其整体架构。了解了公式架构之后，就是分段查看运算结果了。

在编辑栏中选取公式的某一部分，如选取“FIND("@",A2)”，如图4.39所示。注意选取公式的一部分时需要选取完整，如有遗漏部分，如括号没有成对出现、参数不全等，Excel会返回错误的运算结果或者跳出错误提示框。




图 4.39 选取公式的一部分

此时按下 F9 键，在编辑栏中，公式“FIND("@",A2)”返回运算结果 15，如图 4.40 所示。




图 4.40 按 F9 键后的返回结果 1

如果此时按下 Enter 键，Excel 将永久地把“FIND("@",A2)”更改为 15，如仅是调试公式，可按 Esc 键退出。通过这种方法，可以得到函数公式分段计算的结果，再结合函数帮助，就可以理解复杂的函数嵌套公式了。

公式解析：

- ❑ 用 FIND 函数查找“@”在 A2 单元格字符中的位置，减去 1 后作为 LEFT 函数的第 2 个参数，然后用 LEFT 函数返回“@”左侧的字符，并将返回结果作为 MID 函数的第 1 个参数；
- ❑ 用 FIND 函数查找“.”在 A2 单元格字符中的位置，然后加 1 作为 MID 函数的第 2 个参数；
- ❑ MID 函数的第 3 参数使用 99，如返回的字符不足 99 个，则按实际返回。

多选取几段公式按 F9 键显示计算结果，如图 4.41 所示，MID 函数的第 1 个参数取到了 A2 单元格中“@”符号左侧的字符，第 2 个参数是用“.”符号在 A2 单元格中的位置 6 加上 1，表示从“.”之后开始取字符，第 3 个参数 99 在这个公式中已足够大，可取到“.”后所有的字符，即邮箱地址中的名字。




图 4.41 按 F9 键后的返回结果 2

以上展示了 F9 键在分段查看公式运算结果中的作用。F9 键也不是万能的，比如在公式引用整列的情况下，使用 F9 键查看部分公式运算结果时会报错。如 D1 单元格公式为“=FIND("A",A:A)”，选取公式引用的 A 列“A:A”，按下 F9 键，结果如图 4.42 所示。

这个例子说明公式的长度是有限制的，不能超过 8192 个字符。这个错误和引用整列并没有直接的联系，有时候在一些引用区域较大的公式中使用 F9 功能键也会出现类似的

错误。在调试公式时，可以先适当缩小公式的引用范围，或者把部分公式的运算结果先放到单元格中，这样显示会比用 F9 键更加直观，相关内容会在之后的章节中详细讲解。




图 4.42 使用 F9 键时有时也会出错

职场点睛：F9 键属于必须掌握的功能键，在实际工作中应用的意义在于能提升函数公式的阅读能力，其功能类似于阅读外文资料时的翻译工具。只有在具备一定的函数公式阅读理解能力之后，才能将函数高手好的思路运用到自己的工作中去。

4.5.5 了解运算符及其优先顺序

在学习了使用“公式求值”工具及 F9 键分析公式运行结果之后，可能读者还会存在如下疑问，即一个复杂的函数公式到底是如何运算的？先运算哪一段，再运行哪一段？使用“公式求值”工具能得到运算的过程，但 Excel 并没有告知其中的原理。其实道理很简单，和运算符有关。

1. 了解Excel中的4大类运算符

没必要去记忆运算符的概念，只需了解 Excel 中有 4 种类型的运算符：算术运算符、比较运算符、文本运算符和引用运算符。

- ❑ 算术运算符：主要包括加+、减-、乘*、除/、负号-、百分号%、乘幂^；
- ❑ 比较运算符：主要有等号=、大于>、小于<、大于等于>=、小于等于<=、不等于<>，可以使用比较运算符比较两个值，返回结果是一个逻辑值，不是 TRUE 就是 FALSE；
- ❑ 文本运算符：就是连接符&，主要用于数字、字符及字符串的连接；
- ❑ 引用运算符：主要有区域运算符（冒号）、联合运算符（逗号）、交叉运算符（空格），多用于函数公式中引用参数区域。

除了引用运算符比较陌生之外，其他在小学阶段应该都已经接触过了，仅是写法略有些差异，这里不再赘述。

2. 理解运算符的先后顺序

与常规的算术四则运算类似，Excel 中的运算符也有其运算的优先级，其总体规律如下：

- ❑ 4 大类运算符的优先顺序是：引用运算符→算术运算符→文本运算符→比较运

算符；

- ❑ 同一大类运算符中仅算术运算符还有不同的优先级，其他 3 大类根据出现的先后顺序；
- ❑ 算术运算符的优先顺序是：负号-→百分比%→乘幂^→乘*和除/→加+和减-；
- ❑ 括号可以改变运算的优先级别。

综合 Excel 运算符的优先级规律，可得到如表 4.5 所示的运算符优先顺序表。

表 4.4 Excel运算符的优先顺序

顺序	符号	说 明
1	括号()	如有多组括号，其计算顺序是由最内层逐级向外计算
2	引用运算符	包括冒号、空格和逗号，同一类中按照出现的先后次序
3	算术运算符：负号-	并不是减号，其作用是取得与原值正负号相反的值
4	算术运算符：百分比%	相当于进行了除以 100 的运算，可连续使用，如输入“=4%”等于 0.0004
5	算术运算符：乘幂^	例如 10^2 等于 100，2^3 等于 8
6	算术运算符：乘*和除/	不同于数学运算中的写法
7	算术运算符：加+和减-	同数学运算中的符号及含义
8	文本运算符：连接符&	连接文本时，需在文本外面加上一对半角状态先的双引号””
9	比较运算符	包括=、>、<、>=、<=、<>6 种，后 3 种不同于数学运算中的写法

3. 运算符在函数公式中的应用

学习了上述概念之后，进入实战阶段，看看函数公式到底是如何运算的，运算符是如何改变运算顺序的。

【例 1】一对括号的差异

一个多条件查找的例子，用到了 LOOKUP 函数，正确的公式和错误的公式仅差一对括号，如图 4.43 所示。

	A	B	C	D	E	F	G	H	I	J
1	品名	销货单号	销货数量		品名	销货单号	销货数量			
2	A01	B001	100				正确	错误		
3	A02	B004	200		A02	B004	200	400		
4	A03	B003	300		A04	B004	700	700		
5	A02	B002	400							
6	A04	B005	500		正确的公式：					
7	A06	B006	600		=LOOKUP(1,0/((A\$2:A\$10=E3)*(B\$2:B\$10=F3)),C\$2:C\$10)					
8	A04	B004	700		错误的公式：					
9	A08	B008	800		=LOOKUP(1,0/(A\$2:A\$10=E3)*(B\$2:B\$10=F3),C\$2:C\$10)					
10	A09	B009	900							

图 4.43 一对括号的差异

两个公式粗看差不多，差异仅是正确的公式在“0/”后面多了一对括号。就是这对括号，改变了公式运算的顺序，最终影响到了运算结果。

在这两个公式中分别选取 LOOKUP 函数的第 2 个参数，使用 F9 键观察运算结果，可以发现两个公式的返回值是不同的。如果将这段按 F9 键运算的结果放到单元格中，观察将更加直观，如图 4.44 所示。

如没有这对括号，LOOKUP 函数的第 2 参数中因运算符是乘*和除/，属于同一级，按照出现的先后顺序运算；加了括号之后，先运算括号中的乘*，然后再是“0/”。

备注：关于 LOOKUP 函数的应用，将在第 5 章中详细讲解，这里仅需了解公式运算的先后顺序。

	A	B	C	D	E	F
1	品名	销货单号	销货数量		正确	错误
2	A01	B001	100		#DIV/0!	#DIV/0!
3	A02	B004	200		0	0
4	A03	B003	300		#DIV/0!	#DIV/0!
5	A02	B002	400		#DIV/0!	0
6	A04	B005	500		#DIV/0!	#DIV/0!
7	A06	B006	600		#DIV/0!	#DIV/0!
8	A04	B004	700		#DIV/0!	#DIV/0!
9	A08	B008	800		#DIV/0!	#DIV/0!
10	A09	B009	900		#DIV/0!	#DIV/0!

图 4.44 两个公式部分计算结果

4.5.6 理解逻辑值与数值之间的关系

在函数公式中有一类特殊的值——逻辑值。逻辑值只有两个：TRUE 和 FALSE。虽然只有两个，但函数公式因有逻辑值而更加精彩！

谈到逻辑值，首先要说说 IF 函数。IF 函数据说人人都会，属于不需要学的，不就有个条件，符合条件执行参数 2，不符合条件执行参数 3，公式结构如下：

IF (逻辑值, 逻辑值为 TRUE 时的返回结果, 逻辑值为 FALSE 时的返回结果)

这里出个题目考考大家，看看函数“=IF(3,6,9)”返回什么结果？不要在电脑上运算，告知结果及原因。

答案是 6，如果读者能够正确回答，恭喜您，您的函数已经具备一定的基础了；如果答错了也不用沮丧，还是要恭喜您，阅读本书是您的正确选择。

查阅 Excel 自带的关于 IF 函数的帮助文件是找不到答案的，IF 函数的第 1 个参数是一个逻辑测试值，这个公式中出现了数字 3。Excel 非但没有报错，还能返回结果，原因在于这里的数值参与了逻辑判断。

关于逻辑值与数值的关系，可以先从以下两个方面来理解：

- ❑ 逻辑值参与数值运算时，TRUE 相当于 1，FALSE 相当于 0；
- ❑ 数值参与逻辑判断时，0 相当于 FALSE，非 0 的其他数值相当于 TRUE。

本题中 IF 函数的第 1 个参数使用了 3，相当于数值 3 参与了逻辑判断，非 0 值相当于 TRUE，所以返回了 IF 函数的第 2 个参数 6。关于 IF 函数第 1 参数中使用不同值后的返回情况如图 4.45 所示。

	A	B	C
1	返回值	A列公式	备注
2	6	=IF(TRUE,6,9)	
3	6	=IF(1,6,9)	
4	6	=IF(-3.8,6,9)	第1参数非0值相当于TRUE
5	9	=IF(FALSE,6,9)	
6	9	=IF(0,6,9)	
7	#VALUE!	=IF("AA",6,9)	第1参数如使用文本则会报错

图 4.45 IF 函数第 1 参数不同取值的返回情况

了解了这个规律之后，可以利用逻辑值做一些公式的简化，请看如下示例：

【例2】身份证号码中提取出生年月日

因身份证号码还有 15 位和 18 位之分，在提取身份证号码中信息时常规思路是先做一次判断，看看是 15 位的还是 18 位的，然后分别使用不同的公式，如图 4.46 所示。

简化版的公式把“(LEN(A2)=18)”运算结果的逻辑值再乘以 2，根据不同的逻辑判断结果返回 0 或者 2，然后再加 6，以 6 或者 8 作为 MID 函数的第 3 参数，这样就简化了公式。

	A	B	C
1	身份证号码	常规	简化
2	310109197812051280	19781205	19781205
3	310109781205128	781205	781205
4			
5	常规公式：		
6	=IF(LEN(A2)=18,MID(A2,7,8),MID(A2,7,6))		
7	简化公式：		
8	=MID(A2,7,6+(LEN(A2)=18)*2)		

图 4.46 取出生年月日的两个公式

4.6 三键后面的奥妙

之前讲解的函数公式都是以“=”开头，按 Enter 键结束的，叫做普通公式。还有一类公式如果直接按 Enter 键结束，可能会报错，也可能出现一个错误的返回结果（非错误值）。此类公式需要按 Ctrl+Shift+Enter 三键结束，有个特定的称谓叫做数组公式。

4.6.1 认识第一个数组公式

数组公式属于函数公式中最精彩的部分，正如使用 Excel 离不开函数一样，使用函数也无法回避数组公式。数组公式并不神秘，就算零基础的用户也能轻松学会，下面就看一个简单的例子。

已知产品销售数量和单价，要求总销售额。常规方法是加个辅助列，得出每个产品的销售额，然后再求和。使用数组公式可以不用辅助列，一个公式完成，如图 4.47 所示。

F7		fx		{=SUM(B2:B6*C2:C6)}				
	A	B	C	D	E	F	G	H
1	产品编码	数量	单价		金额			
2	BD0004	16	111		1,776			
3	BD0005	27	193		5,211			
4	BD0006	16	171		2,736			
5	BD0007	27	124		3,348			
6	BD0008	13	184		2,392			
7				总金额	15,463	15,463		
8	常规公式：	=SUM(E2:E6)						
9	数组公式：	{=SUM(B2:B6*C2:C6)}						
10								

外面的那一组花括号不是手工输入的，是按三键的返回结果。

图 4.47 普通公式与数组公式比较

选中数组公式所在的 F7 单元格，在编辑栏中可见在等号外面有一组花括号{}。这组花括号不是手工输入的，是按 Ctrl+Shift+Enter 三键的返回结果。

数组公式的本质是多重运算，所谓多重运算说得通俗一些就是一组数据与另一组数据之间的运算，如本例中就是两列数据相乘，然后再用 SUM 求和。如选中编辑栏中的部分公式“B2:B6*C2:C6”，按 F9 键，返回结果如图 4.48。

=SUM({1776;5211;2736;3348;2392})			
SUM(number1, [number2], ...)	G	H	I

图 4.48 数组公式中按 F9 键返回的结果

按 F9 键后，在编辑栏中返回了一串数字，正是 B2: B6 与 C2: C6 两列数据计算的结果。在不改变公式的情况下，如果不按三键结束，Excel 会报错（也可能是返回一个错误结果，这个和公式所在单元格与引用单元格的位置相关，示例文件中有举例）。这里使用三键，相当于给 Excel 发了一个通知，告知这是多重运算。如本例中在外套的 SUM 函数求和时，要把多重运算的计算结果全部加进去。

4.6.2 数组的维度与尺寸

之前介绍了数组公式的定义，数组与数组公式是两个不同的概念。所谓数组，就是元素的集合，按行和列进行排列。行和列方向称为数组的维度（讲解仅以二维为限）。单行或单列的数组是一维数组，又分为一维横向数组和一维纵向数组；多行多列（含 2 行 2 列）的数组是二维数组。

在数组每个维度上又有若干个元素，元素的数量称为数组在这个维度上的尺寸。比如有 1 个 M*N 的数组，表示其在行方向上有 M 个元素，列方向上有 N 个元素。

4.6.3 理解数组运算规则

认识了数组之后，再来理解一下数组的运算规则。中学数学课程中学习过向量运算，数组运算规则和向量运算规则很接近。以下举例讲解数组运算的规则，仅取加法运算，其他运算原理一致。

1. 单值与数组之间的运算

单值（也可是单元素数组）可以与一个数组自由运算，运算方式为单值与数组中的各个元素分别运算，最终返回与这个数组方向尺寸一致的另一个数组。如图 4.49 是一个单值与数组运算的例子。

	A	B	C	D	E	F	G
1	单值		数组		运算结果		
2	10		1		11		
3			2		12		
4			3		13		
5			4		14		
6				公式：	{=A2+C2:C5}		
7							
8							

图 4.49 单值与数组运算示意图

2. 同方向一维数组之间的运算

两个同方向一维数组的运算进行相同位置元素一一对应的运算，因此要求两个数组具有相同的尺寸。如参与运算的两个数组尺寸大小不一，则仅是两个数组都有元素的部分能

返回运算结果，其他部分返回错误值#N/A。

如图 4.50 演示了两个同方向一维数组运算，尺寸相同与尺寸不同两种情况下的返回结果。

	A	B	C	D	E	F	G	H
1	数组1		数组2		同尺寸		不同尺寸	
2	1		10		11		11	
3	2		20		22		22	
4	3		30		33		33	
5	4		40		44		44	
6	5				{=A2:A5+C2:C5}		#N/A	
7							{=A2:A6+C2:C5}	

图 4.50 同方向一维数组运算

3. 不同方向一维数组之间的运算

如是两个不同方向一维数组的运算，则没有尺寸大小的限制。如一个 M 行数组 1 与一个 N 列数组 2 的运算，则是数组 1 中的元素分别与数组 2 中的元素运算，得到 M*N 个元素，结果为 M*N 的矩阵。

如图 4.51 演示了两个不同方向的一维数组之间的运算，返回结果为 1 个数组，其行数同第 1 个数组的行、列数同第 2 数组的列。

	A	B	C	D	E	F	G
1	数组1		数组2	10	20	30	
2	1						
3	2			11	21	31	
4	3			12	22	32	
5	4			13	23	33	
6				14	24	34	
7			公式：	{=A2:A5+D1:F1}			
8							

图 4.51 不同方向一维数组之间的运算

4. 一维数组与二维数组之间的运算

当一维数组与二维数组具有某些相同尺寸的特征时，可以返回正确的运算结果。如有一个 M 行 N 列的二维数组，另有一个一维数组需与这个二维数组运算，则一维数组必须是 M 行的一维数组或者是 N 列的一维数组。

如果不满足这个运算条件，则在一维数组方向上差异部分整行或整列返回#N/A 错误。如图 4.52，分别演示了一维数组与二维数组运算时，行方向上尺寸相同与尺寸不同的返回结果。

	A	B	C	D	E	F	G	H	I	J
1	一维数组		二维数组			行相同			行不同	
2	1		10	20		11	21		11	21
3	2		20	40		22	42		22	42
4	3		30	60		33	63		33	63
5	4		40	80		44	84		44	84
6	5					{=A2:A5+C2:D5}			#N/A	#N/A
7									{=A2:A6+C2:D5}	

图 4.52 一维数组与二维数组运算

5. 二维数组之间的运算

两个二维数组运算按尺寸较小的数组所有元素的位置进行一一对应的运算，并返回两

个数组较大尺寸方向元素个数运算的数组。如果两个数组的尺寸完全相同，则全部返回正确的运算结果，否则仅两个数组较小尺寸方向元素个数运算区域可以返回正确的运算结果，超出部分均返回#N/A 错误。

如图 4.53 所示，演示了两个二维数组运算的结果，这两个二维数组的尺寸存在差异，则运算结果返回值有部分出错。

	A	B	C	D	E	F	G	H	I
1	数组1			数组2					
2	1	2		10	20	30			
3	3	4		40	50	60			
4	5	6							
5									
6	运算结果			公式：					
7	11	22	#N/A	{=A2:B4+D2:F3}					
8	43	54	#N/A						
9	#N/A	#N/A	#N/A						
10									
11									
12									

返回数组的尺寸按参与运算数组行和列的最大值，返回正确运算结果的区域按最小值。

图 4.53 两个二维数组的运算

这个例子中，返回数组的大小是 3*3，原因在于参与运算的两个数组的行和列方向最大值都是 3（数组 1 是 3 行，数组 2 是 2 行，行方向上最大值是 3；数组 1 是 2 列，数组 2 是 3 列，列方向上最大值也是 3）；其中具有正确运算结果的区域是 2*2，和参与运算的两个数组的行和列方向的最小值吻合。

4.6.4 数组运算结果与单元格中的返回值

数组运算的结果必然是一个数组，这个和数组公式运算返回的结果是两个概念。

关于数组的运算结果，可以在之前的例子中选中具有数组运算公式的单元格，进入编辑栏，全选公式后按 F9 键查看，如对于图 4.53 中的例子，选取 C7 单元格，在编辑栏中全选公式后按 F9 键，返回结果为{11,22,#N/A;43,54,#N/A;#N/A,#N/A,#N/A}，如图 4.54 所示。

VLOOKUP									
= {11,22,#N/A;43,54,#N/A;#N/A,#N/A,#N/A}									
	A	B	C	D	E	F	G	H	I
1	数组1			数组2					
2	1	2		10	20	30			
3	3	4		40	50	60			
4	5	6							
5									
6	运算结果			公式：					
7	11	22	#N/A	{=A2:B4+D2:F3}					
8	43	54	#N/A						
9	#N/A	#N/A	#N/A						
10									
11									
12									

返回结果为数组，其中逗号“,”表示同行不同列，分号“;”表示换行。

图 4.54 数组运算返回的结果

有时候在将数组运算的结果放在单元格中查看时，会发生错误或者返回了一些出乎意料的值，这类情况原因在于改变了数组运算结果返回的单元格区域大小或者方向。

如图 4.55 所示，如改变单值和一维数组运算返回的单元格区域大小，多出来的单元格区域中将会返回错误值；如改变返回的单元格区域方向仅返回一行，则只会返回单值和数

组中第一个元素运行的结果。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	单值		数组		运算结果		改变返回区域大小			返回结果在横向区域				
2	10		1		11		11			11	11	11	11	
3			2		12		12			公式:				
4			3		13		13			=A2+C2:C5				
5			4		14		14							
6				公式:	=A2+C2:C5		#N/A							
7							#N/A							
8						公式:	=A2+C2:C5							
9														
10														
11														
12														

图 4.55 改变返回的数组大小或方向

图 4.55 中 3 个公式并没有差异,但返回的结果不一样。如将测试的数组改为一维横向数组或者二维数组,都能得到类似的结果。由此可见,数组运算的结果返回到单元格中出错未必就是数组运算的结果本身有错,也有可能和选择的返回区域有关。

如需判断具体的错误原因,可以使用 F9 键。在图 4.55 的 3 个公式中,即使在返回错误值的单元格中,在编辑栏中全选公式,返回的结果都是{11;12;13;14}。如果按 F9 键的返回值中没有错误值,而返回到单元格中时出错,原因就在于选取了错误的单元格区域。

4.7 小 结

本章讲述了函数的一些基本概念如公式录入方法、引用方式、引用类型、定义名称、错误值的类型及形成原因、运算符的分类及其优先顺序、逻辑值与数值的关系、调试公式的常用方法,最后还介绍了数组的运算规则。

这些都是很基础的内容,讲解这些是为后续学习打好基础。单个函数本身不难,难在函数组合。学习函数组合好比学习外语,既要懂语法,还要有一定的单词量。

函数组合中语法就是本章已学习的内容,光会语法没有单词量肯定是不行,在以后的 3 个章节中将详细讲解一些常用的函数。